

Einführung in die Assemblerprogrammierung mit dem MIPS-Simulator SPIM

Reinhard Nitzsche

15.09.97

Letzte Änderung: 3. Juli 2001

Inhaltsverzeichnis

1	SPIM	5
1.1	Einleitung	5
1.2	Beschaffungsmöglichkeiten	5
1.3	Oberflächen	6
2	Maschinenmodell des SPIM	7
2.1	Koprozessoren	7
2.2	Register	7
3	Grundprinzipien der Assemblerprogrammierung	10
3.1	Assembler, Assemblersprache und Maschinsprache	10
3.2	Befehlsarten	10
3.3	Aufbau einer Assembler-Befehlszeile	11
3.4	Trennung von Programm und Daten	11
3.5	Tips zur übersichtlichen Gestaltung von Assemblerprogrammen .	12
3.6	Aufbau eines Assemblerprogramms	13
3.7	Notation der Befehle	13
4	Datenhaltung I: ganze Zahlen und Zeichenketten	15
4.1	Ganze Zahlen im Datensegment	15
4.1.1	Die <code>.word</code> -Direktive	15
4.1.2	Weitere Ganzzahltypen	15
4.2	Zeichenketten im Datensegment: die <code>.asciiz</code> -Direktive	16
4.3	Die Datenausrichtung im Datensegment	17
5	Transferbefehle	18
5.1	Ladebefehle und Adressierungsmodi II	18
5.1.1	Adressierungsmodi I: Register Indirekte, direkte und in- dexierte Adressierung	18
5.1.2	Weitere Ladebefehle	19
5.2	Speicherbefehle	20
5.3	Register-Transfer-Befehle	21

INHALTSVERZEICHNIS

6	Arithmetische Befehle (ganze Zahlen)	22
6.1	Adressierungsmodi II: Register direkte und unmittelbare Adressierung	22
6.2	Addition und Subtraktion	22
6.3	Multiplikation und Division	24
6.4	Sonstige arithmetische Befehle	25
7	Das SPIM-Betriebssystem (Ein- und Ausgabe)	27
8	Logische Befehle	30
8.1	Elementare logische Befehle	30
8.2	Rotations- und Schiebebefehle	30
8.3	Vergleichsbefehle	32
9	Kontrollstrukturen	35
9.1	Programmverzweigungen (einfache Fallunterscheidungen) oder: Sprungbefehle	35
9.2	Schleifen	39
9.2.1	Abweisende Schleifen	40
9.2.2	Nachprüfende Schleifen	40
9.2.3	Zählschleifen	41
9.2.4	Schleifen verlassen	41
10	Datenhaltung II: Komplexe Datentypen	43
10.1	Felder	43
10.1.1	Erstellung und Initialisierung von Feldern	43
10.1.2	Arbeiten mit Feldern	44
10.2	Verbunde	45
11	Mehrfache Fallunterscheidung (CASE-Anweisung)	46
12	Kellerspeicher (Stack)	48
12.1	Das Grundprinzip	48
12.2	Die Stackprogrammierung: Einkellern	48
12.3	Die Stackprogrammierung: Lesen vom Stack	50
12.4	Anwendungen und Beispiele	51
12.5	Vor- und Nachteile der CISC-Befehle PUSH und POP	51
13	Prozeduren	53
13.1	Einfache Unterprogrammaufrufe	53
13.2	Verschachtelte Unterprogrammaufrufe (Prozedur-Konvention)	55
13.2.1	Prolog des Callers	55
13.2.2	Prolog des Calleees	57
13.2.3	Der Callee	58
13.2.4	Epilog des Calleees	59
13.2.5	Epilog des Callers	59
13.2.6	Ein Beispiel zum Prozeduraufruf	60
13.2.7	Zusammenfassung: Prozedur-Konvention beim SPIM	62
14	Unterbrechnungen und Ausnahmen	64
14.1	Zu den Begriffen „Unterbrechung“ und „Ausnahme“	64
14.2	Die Behandlung von Unterbrechnungen und Ausnahmen	64
14.2.1	Das Statusregister	65
14.2.2	Das Cause-Register	66
14.2.3	Die Unterbrechungsbehandlungsroutine	66

15 Gleitkommazahlen	68
15.1 Datenhaltung III: Gleitkommazahlen	68
15.2 Transferbefehle	69
15.2.1 Ladebefehle	69
15.2.2 Speicherbefehle	70
15.2.3 Register-Transfer-Befehle	70
15.3 Arithmetische Befehle für Gleitkommazahlen	71
15.4 Vergleichsbefehle für Gleitkommazahlen	71
15.5 Ein Beispiel zu Gleitkommazahlen	72
A Zusammenfassende Tabellen	74
A.1 Befehlsübersicht	74
A.2 Alphabetische Befehlsübersicht	80
A.3 Direktiven-Übersicht	80
A.4 Registerübersicht	82
A.5 Betriebssystemfunktionen	82
B Literaturverzeichnis	82
C Verzeichnisse	83

Vorwort

Dieses Tutorial ist entstanden, weil im Sommersemester 1996 nur drei statt vier Tutorien für die Vorlesung Rechnerorganisation stattfanden. Professor Schweppe trat mit dem Vorschlag statt des nicht stattfindenden Tutoriums ein Tutorial zur Assemblerprogrammierung zu schreiben an mich heran. Die weitere Ausarbeitung fand dann in Abstimmung mit Professor Rojas statt. Das Ergebnis ist das vorliegende Dokument, das die Vorlesung Rechnerorganisation um die Grundlagen der Assemblerprogrammierung entlasten soll. Tatsächlich bildet der behandelte Stoff nur eine Teilmenge der Assemblerprogrammierung und der Vorlesung Rechnerorganisation.

Im Sommersemester 1997 wurde dieses Tutorial zum ersten Mal in der Vorlesung von Professor Rojas eingesetzt. Die mir zugegangenen Reaktionen der Leserinnen und Leser waren überwiegend positiv. Wertvolle Anregungen aus dem Leserkreis wurden in dieser zweiten Version des Tutorials aufgenommen.

Ich danke allen, die mir mit Korrekturlesungen und Fehlermeldungen geholfen haben.

Berlin, im September 1997

Reinhard Nitzsche

1 SPIM

1.1 Einleitung

Die Assemblerprogrammierung ist auf Simulatoren einfacher zu erlernen als direkt auf einem Rechner. Ein Simulator kann bei Programmierfehlern leicht zurückgesetzt werden, bei einem abgestürzten Rechner dauert dies ungleich länger. Der Simulator erlaubt meist eine bessere Einsicht in die wesentlichen Teile des Rechners als dies bei professionellen Assemblern der Fall ist.

Wir werden den Simulator **SPIM** verwenden, der den Prozessor **MIPS** R2000 simuliert. Der **MIPS** R2000 ist ein sehr klar strukturierter RISC-Prozessor, der im Gegensatz zu modernen RISC-Maschinen leicht zu verstehen ist. In Abschnitt 2 auf Seite 7 werden wir ein bisschen mehr über diesen Prozessor und die RISC-Technologie erfahren.

Ein weiterer Vorteil der Verwendung von Simulatoren liegt in der Möglichkeit der Simulation von fremden Prozessoren. Unser **SPIM**, den wir verwenden werden, kann z.B. auf allen UNIX-Rechnern, ab Windows 3.11 und auf Macintosh-Rechnern verwendet werden.

Wir werden uns nun Schritt für Schritt mit der Assemblerprogrammierung vertraut machen. Ich habe mich für eine synthetische Herangehensweise an die Materie entschieden. Wir werden also nicht mit einem Beispielprogramm beginnen, sondern ganz von „unten“. Dadurch gerät der Anfang etwas trocken, dieser Nachteil wird aber –so hoffe ich– durch eine bessere Verständlichkeit und gute Nachschlagsmöglichkeiten wieder wettgemacht.

Zunächst widmen wir uns in diesem Abschnitt noch dem Simulator, im Abschnitt 2 werden wir uns dann einen groben Überblick über die Technik des Originals, den **SPIM** R2000, verschaffen. (Wem jetzt der eine oder andere Begriff nicht bekannt ist, muss nicht verzweifeln, sie werden später alle noch erläutert!) Einige Grundregeln der Assemblerprogrammierung werden wir im Abschnitt 3 kennen lernen. Nachdem wir uns in den Abschnitten 4, 5 und 6 mit der Datenhaltung, den Transferbefehlen und den arithmetischen Befehlen vertraut gemacht haben, können wir schon kleinere Programme schreiben. Einige logische Befehle (Abschnitt 8) ebnen uns den Weg zu den Kontrollstrukturen in Abschnitt 9. Es folgt die Hohe Schule der Datenhaltung, nämlich die Haltung von Verbunden und Feldern in Abschnitt 10. Erst danach lernen wir eine weitere Kontrollstruktur kennen, die mehrfache Fallunterscheidung (Abschnitt 11). Schließlich wenden wir uns im Abschnitt 12 dem Kellerspeicher (Stack) zu. Auf gewissenhafte Programmierung kommt es schließlich im Abschnitt 13 an, in dem wir uns mit Konventionen zum Prozeduraufruf beschäftigen werden. Die beiden folgenden Abschnitte 15 und 14 bieten uns einen Ausblick auf weitere Aspekte der Assemblerprogrammierung, nämlich die Unterbrechungsbehandlung und mathematische Koprozessoren.

1.2 Beschaffungsmöglichkeiten

Der **SPIM** unterliegt lediglich der GNU-Lizenz, kann also frei benutzt und vervielfältigt werden. Im Internet ist er unter <ftp://ftp.cs.wisc.edu/pub/spim/> zu finden. In diesem Verzeichnis befinden sich u.a. folgende Dateien:

- `spim.tar.z` und `spim.tar.gz`: UNIX-Versionen inklusive Quelldateien
- `SPIM.sit.bin` und `SPIM.sit.Hqx.txt`: Macintosh-Programme

1. SPIM

- `spim.zip`: Windows-Programme

Die Installation sollte nicht auf allzu große Probleme stoßen. Die UNIX-Versionen müssen zunächst kompiliert werden. Die Windows-Version ist für Windows 95 geschrieben worden. Auf Windows 3.11 muss zunächst noch ein 32-Bit-Treiber installiert werden, der aber mitgeliefert wird. Der Simulator läuft auch unter Windows 3.11 einigermaßen stabil.

1.3 Oberflächen

Die Oberflächen der UNIX- und Windows-Versionen unterscheiden sich geringfügig. Die x-Oberfläche besteht im Gegensatz zu den Windows-Oberflächen aus einem einzigen Fenster, in dem Register, Programm- und Datenbereich und Ausgaben des Simulators angezeigt werden. Den Bildschirm des simulierten Rechners kann man über die Taste „Terminal“ erhalten.

Bei den Windows-Oberflächen sind Register, Programm- und Datenbereich sowie Ausgaben des Simulators („Session“) über einzelne Fenster zugänglich (Menü „Windows“). Im gleichen Menü befindet sich der SPIM-Bildschirm.

Wer mit „seinem“ Betriebssystem vertraut ist, sollte keine Probleme mit der Benutzung der jeweiligen Oberfläche haben. Zur Einarbeitung seien die jeweiligen Hilfefunktionen und die mitgelieferten Dokumentationen bzw. [Pat, S. A-36 bis A-45] empfohlen.

Zu beachten sind jedoch noch folgende Hinweise:

- Auch die letzte Programmzeile muss mit der Eingabetaste beendet werden. Andernfalls wird für die letzte Zeile eine Fehlermeldung angezeigt.
- Bei den Windows-Versionen darf die über `File/Load` geladene Datei nur in dem Verzeichnis `C:\SPIM` stehen, da andernfalls bei `File/Clear` das Programm *beendet* wird.
- Die Windows-Versionen erlauben nicht das Memory-Mapped-IO¹.

¹s. Abschnitt 14 auf Seite 64

2 Maschinenmodell des SPIM

SPIM simuliert den RISC-Prozessor **MIPS** R2000. Dieser zeichnet sich durch eine klare Architektur und einen übersichtlichen Befehlssatz aus.

Das RISC-Konzept sieht vor, statt Prozessoren mit einer großen Zahl von Befehlen auszustatten, nur vergleichsweise wenig Befehle zu implementieren. RISC steht für **Reduced Instruction Set Computer**, das Pendant heißt CISC (**Complex Instruction Set Computer**). Die wenigen implementierten RISC-Befehle können sehr viel schneller ausgeführt werden. Für viele Arbeiten müssen allerdings mehr RISC-Befehle verwendet werden als bei CISC-Rechnern.

RISC

Das Grundmodell des SPIM wurde 1984 von John Hennessy, dem Ko-Autor von [Pat] an der Universität Stanford entwickelt. Die MIPS-Prozessoren werden u.a. von DEC verwendet. Vier Jahre zuvor entwickelte David A. Patterson, der andere Autor von [Pat] an der Berkeley-Universität den Begriff RISC und den RISC-Prozessor RISC I, der für die SPARC-Prozessoren der Firma Sun Patent stand und der mit ganzen drei (!) Befehlen auskam [Tan, S. 435].

Die gängigen Prozessoren der Personal-Computer wie z.B. INTEL 80x86 und Pentium, aber auch die Prozessoren der Motorola 68xxx-er Familie sind CISC-Rechner.

2.1 Koprozessoren

Die MIPS R2000-CPU verfügt über die Möglichkeit mehrere Koprozessoren anzusteuern. Diese Koprozessoren sind für von Gleitkommaberechnungen oder Betriebssystemnahe Aufgaben zuständig. Mit diesen Möglichkeiten werden wir uns aber erst in Abschnitt 15 auf Seite 68 und Abschnitt 14 auf Seite 64 beschäftigen.

2.2 Register

Der MIPS verfügt über 32 Register. Jedes Register enthält ein 32-Bit-Wort und kann -theoretisch- für jeden Zweck verwendet werden. Man nennt solche Register auch *general purpose register*. Eine Ausnahme stellt das Register `$zero` dar, dessen Wert Null nicht verändert werden kann.

32 Register

general purpose register

Zusätzlich zu diesen 32 Registern verfügt der SPIM noch über zwei weitere Register mit den Namen `lo` für „low“ und `hi` für „high“, die bei der Multiplikation und der Division verwendet werden. Im eingangs erwähnten Koprozessor 0 finden sich vier weitere Spezialregister, die für uns erst einmal nicht von Bedeutung sind.

Die Adressierung eines Registers kann durch ein Dollarzeichen (\$) und die Registernummer 0-31 erfolgen. Die Register `lo` und `hi` sind nur mit Spezialbefehlen lesbar und schreibbar. Von der Adressierung über die Registernummern ist aber abzuraten, da dies zu einem sehr unübersichtlichen Code führt. Besser ist die Verwendung der Registernamen, die der Assembler (siehe Abschnitt 3.1 auf Seite 10) dann in die zugehörigen Registernummern umsetzt. Die Registernummern sind aber bei der Fehlersuche hilfreich, da SPIM bei geladenen Programmen die Register nur durch ihre Registernummern angibt.

Bezüglich der Verwendung der Register gibt es eine Konvention.² Sie ist unbedingt einzuhalten, damit Programmteile verschiedener Autoren austauschbar

²[Pat] verwendet leider nicht konventionsgemäße Registernummern

2. MASCHINENMODELL DES SPIM

Abbildung 1: MIPS-Register und ihre Verwendungszwecke nach Registernummern

Register- -name	-nr.	vereinbarte Nutzung	Bemerkungen
\$zero	0	Enthält den Wert 0	kann nicht verändert werden.
\$at	1	temporäres Assemblerregister	Nutzung ausschließlich durch den Assembler!
\$v0	2	Funktionsergebnisse 1 und 2	auch für Zwischenergebnisse
\$v1	3		
\$a0	4	Argumente 1 bis 4 für den Prozeduraufruf	
\$a1	5		
\$a2	6		
\$a3	7		
\$t0	8	temporäre Variablen 1-8	Können von aufgerufenen Prozeduren verändert werden.
\$t1	9		
\$t2	10		
\$t3	11		
\$t4	12		
\$t5	13		
\$t6	14		
\$t7	15		
\$s0	16	langlebige Variablen 1-8	Dürfen von aufgerufenen Prozeduren nicht verändert werden.
\$s1	17		
\$s2	18		
\$s3	19		
\$s4	20		
\$s5	21		
\$s6	22		
\$s7	23		
\$t8	24	temporäre Variablen 9 und 10	Können von aufgerufenen Prozeduren verändert werden.
\$t9	25		
\$k0	26	Kernel-Register 1 und 2	Reserviert für Betriebssystem, wird bei Unterbrechungen verwendet.
\$k1	27		
\$gp	28	Zeiger auf Datensegment	
\$sp	29	Stackpointer	Zeigt auf das erste <i>freie</i> Element des Stacks.
\$fp	30	Framepointer	Zeiger auf den Prozedurrahmen
\$ra	31	Return address	Enthält nach Aufruf u.a. des Befehls <i>jal</i> die Rücksprungadresse.

sind. Die Nichtbeachtung einzelner Konventionen kann zu nicht vorhersagbarem Laufzeitverhalten des Programmes führen.

\$zero: Enthält den Wert Null. Dieses Register ist auf Schaltungsebene realisiert und kann *nicht* verändert werden.

\$at, \$k0 und \$k1: Reserviert für Berechnungen des Assemblers. Die \$k- (Kernel)-Register werden bei der Unterbrechungsbehandlung verwendet. Obwohl sich das \$at-Register lesen und schreiben lässt, sollte es keinesfalls verwendet werden! Wir werden uns in Abschnitt 14 auf Seite 64 eingehender mit diesen Registern auseinandersetzen.

2.2 Register

- `$a0` bis `$a3`: Prozedurargumente, weitere Argumente müssen über den Stack übergeben werden, vergleiche Abschnitt 13 auf Seite 53.
- `$v0` und `$v1`: Funktionsergebnisse, gegebenenfalls können diese Register auch bei der Auswertung von Ausdrücken benutzt werden, vergleiche Abschnitt 13.
- `$t0` bis `$t9`: Diese Register sind für die Haltung kurzlebiger (temporärer) Variablen bestimmt. Sie können nach einem Prozeduraufruf von der aufgerufenen Prozedur verändert werden. Nach einem Prozeduraufruf kann also *nicht* davon ausgegangen werden, dass die Werte in den `$t`-Registern unverändert sind, vergleiche Abschnitt 13.
- `$s0` bis `$s7`: Die `$s`-Register dienen der Haltung langlebiger Variablen. Sollen sie in einem Unterprogramm geändert werden, so müssen sie zuvor gesichert werden, vergleiche Abschnitt 13.
- `$gp`: Globaler Zeiger auf die Mitte eines 64K großen Speicherblocks im statischen Datensegment, das für alle Dateien sichtbar ist.
- `$fp`: Framepointer, vergleiche Abschnitt 13.
- `$sp`: Stackpointer, vergleiche Abschnitt 12 auf Seite 48.
- `$ra`: Returnaddress, Rücksprungadresse nach dem Aufruf von `jal` und anderen Befehlen.

3 Grundprinzipien der Assemblerprogrammierung

3.1 Assembler, Assemblersprache und Maschinensprache

Maschinenbefehl	Prozessoren können lediglich numerische Instruktionen ausführen. Beispielsweise bedeutet die SPIM-Anweisung 0000 0010 0011 0010 0100 0000 0010 0000: „Addiere den Inhalt der Register 17 und 18 und lege das Ergebnis im Register 8 ab.“. Solche vom Prozessor ausführbaren Befehle nennt man <i>Maschinenbefehle</i> . Die Gesamtheit aller Maschinenbefehle die ein Rechner ausführen kann nennt man <i>Maschinensprache</i> .
Maschinensprache	Hinter der oben genannten scheinbar chaotischen Zahl steckt zwar ein System (die ersten sechs Nullen und die letzten sechs Ziffern stehen beispielsweise für „Addition“), aber schon die ersten Programmierer waren aus verständlichen Gründen doch sehr an einer Visualisierung dieser numerischen Befehle interessiert.
Assembler symbolische Bezeichner	Einfachste <i>Assembler</i> sind Programme, die aus Buchstaben bestehenden <i>symbolischen Bezeichnern</i> einen Maschinenbefehl zuordnen. Tatsächlich muss der Assembler auch noch einige weitere Entscheidungen treffen, etwa welche Register verwendet werden und wie diese zu codieren sind.
symbolische Adressen Marke	Aber die meisten Assembler können mehr als diese recht simple 1:1-Umsetzung von symbolischen Befehlen in Maschinenbefehle. Zur Grundausstattung eines Assemblers gehört z.B. die Verwaltung <i>symbolischer Adressen</i> . Das sind Wörter oder <i>Marken</i> , die der Programmierer im Assemblerprogramm definieren kann und die dann vom Assembler so verwaltet werden, dass von anderen Stellen im Programm her auf diese Marken Bezug genommen werden kann. Dies ist äußerst nützlich bei Sprunganweisungen, bei denen der Programmierer andernfalls erst ermitteln müsste, an welche Stelle (oder über wie viele Speicherworte) gesprungen werden soll. Auch Variablen werden durch solche Marken realisiert. Die meisten Assembler erlauben auch die Bezugnahme auf Marken, die gar nicht in der Datei selber definiert wurden. Dies ermöglicht die Nutzung bereits vorhandener Bibliotheken.
Assemblersprache	Den Wortschatz eines Assemblers nennt man die <i>Assemblersprache</i> , oft aber auch ebenfalls Assembler.

3.2 Befehlsarten

Im einfachsten Fall besteht eine Assemblersprache wie gesagt aus einer 1:1-Umsetzung von symbolischen Befehlen in Maschinensprache. Assembler verfügen jedoch auch über weitere Befehlsarten:

Pseudobefehl	(1) Ein <i>Pseudobefehl</i> ist eine Erweiterung der Maschinensprache. Pseudobefehle werden vom Assembler in eine Folge von Maschinenbefehlen umgesetzt. Sie werden vom Assembler zur Verfügung gestellt und können vom Benutzer nicht neu definiert werden. Auch bestimmte Adressierungsarten, die der Prozessor gar nicht beherrscht, fallen in diese Kategorie. Für den Assemblerprogrammierer sind Pseudobefehle (fast) wie Maschinenbefehle zu benutzen, sie unterscheiden sich rein äußerlich nicht von Maschinenbefehlen.
Assembleranweisung Direktive	(2) Eine <i>Assembleranweisung</i> oder <i>Direktive</i> ist eine Anweisung an den Assembler, etwas zur Assemblierzeit zu tun, z.B. Platz für Variablen zu lassen o.ä. Eine Direktive führt <i>nicht</i> zur Erzeugung von Maschinenbefehlen! Die Direktiven des SPIM beginnen alle mit einem Punkt, z.B. <code>.data</code> .

Es gibt auch Assembler, die benutzerdefinierte Pseudobefehle oder *Makros* erlauben. Der SPIM beherrscht diese Technik jedoch nicht.

3.3 Aufbau einer Assembler-Befehlszeile

Grundsätzlich kann in jede Zeile des Assembler-Programms nur ein einziger Assembler- oder Pseudobefehl geschrieben werden:

```
⟨Marke⟩: ⟨Befehl⟩ ⟨Arg 1⟩ ⟨Arg 2⟩ ⟨Arg 3⟩ #⟨Kommentar⟩
⟨Marke⟩: ⟨Befehl⟩ ⟨Arg 1⟩, ⟨Arg 2⟩, ⟨Arg 3⟩ #⟨Kommentar⟩
```

Die Argumente eines Befehls können nach Belieben durch Kommata oder Lücken (auch Tabulatoren) getrennt werden. *⟨Marke⟩* ist natürlich optional. Alle Zeichen vom Kommentarzeichen # an werden als Kommentar gewertet und überlesen.

Kommentar

Abgesehen von einigen Assembleranweisungen haben fast alle SPIM-Befehle zwischen einem und drei Argumenten.³ Zum Beispiel haben fast alle arithmetischen Befehle drei Argumente, zwei zur Angabe der Quelle und eines zur Angabe des Ziels. Die Befehle zur Übertragung von Daten zwischen Prozessor und Hauptspeicher haben zwei Argumente.

Die Argumente treten *immer* in folgender Reihenfolge auf:

- 1.) Register des Hauptprozessors, zuerst das Zielregister,
- 2.) Register des Koprozessors,
- 3.) Adressen, Werte oder Marken

Marken können übrigens auch in einer Zeile für sich definiert werden. Sie gelten dann für den folgenden Befehl. Dies ist bei längeren Markennamen, etwa bei Prozedurnamen, praktisch. In Markennamen sind auch Ziffern erlaubt, allerdings nicht als erstes Zeichen. Auch der Unterstrich () ist zugelassen. Marken dürfen nicht genauso heißen wie ein Befehl! Die Folge sind sehr schwer erklärliche Fehlermeldungen. In Abschnitt A.2 auf Seite 80 befindet sich eine alphabetische Übersicht der Befehlsnamen.

3.4 Trennung von Programm und Daten

Eines der Grundprinzipien des von-Neumann-Rechners ([Pat, S. 32], [Tan, S. 17-18], [Coy, S. 19-23]) ist ein gemeinsamer Speicher für Daten und Programme. Dieser gemeinsame Speicher führt aber auch zu Problemen. Der Programmierer muss verhindern, dass der Rechner versucht, die abgelegten Daten auszuführen oder Programmteile als Daten zu lesen oder zu überschreiben.

Ein Möglichkeit besteht beispielsweise darin, alle Daten an einem Ende des Programmes unterzubringen und die Ausführung des Programmes entweder vorher zu stoppen oder über den Datenbereich hinweg zu springen.

Eleganter ist das Konzept des SPIM. Der SPIM teilt den Hauptspeicher in mehrere sogenannte *Segmente*: *Daten-*, *Text-* und *Stacksegment* ein. So gibt es ein *Datensegment* für die Daten und ein *Textsegment* für das Programm. Ferner gibt es noch jeweils ein Text- und Datensegment für das Betriebssystem und ein *Stacksegment* für den Stack.

Segment

³Die einzigen Befehle ohne Argumente sind: `syscall`, `nop`, `bc(z)t`, `bc(z)f` und `rfe`

3. GRUNDPRINZIPIEN DER ASSEMBLERPROGRAMMIERUNG

Die Segmente haben jeweils eine feste Größe und einen bestimmten Verwendungszweck. So gibt es u.a. ein Segment für das Benutzerprogramm und eines für die Benutzerdaten. Der Simulator erlaubt die Festlegung der Größe der Segmente mit einem Parameter beim Aufruf des Simulators.

Mehr über das Hauptspeicherkonzept des SPIM R2000 werden wir in Abschnitt 12.1 auf Seite 48 und besonders in Abbildung 6 auf Seite 49 erfahren.

Mit den Direktiven `.text` und `.data` können wir festlegen, in welches Segment die folgenden Befehle geschrieben werden sollen.

`.data`-Direktive

Alle auf die `.data`-Direktive folgenden Befehle (fast immer Direktiven) werden in das Datensegment eingetragen. Theoretisch können hier auch Befehle stehen, die zwar übersetzt, aber nie ausgeführt werden.

`.text`-Direktive

Alle auf die `.text`-Direktive folgenden Befehle werden in das Textsegment eingetragen. Sie können ausgeführt werden.

Die `.data`-Direktive dient also zur Ablage von Daten im Datensegment, die `.text`-Direktive zur Ablage von Befehlen im Textsegment.

3.5 Tips zur übersichtlichen Gestaltung von Assemblerprogrammen

Assemblerprogramme werden noch schneller als Hochsprachenprogramme unübersichtlich. Daher sollten folgende Regeln unbedingt eingehalten werden:

(1) Zu jedem Assemblerprogramm gehört eine Beschreibung seines Verhaltens. Dies kann z.B. in Form eines Hochsprachenalgorithmus zu Beginn des Programmes erfolgen. An diesen Hochsprachenalgorithmus sollte man sich dann aber auch peinlich genau halten.

(2) Das Programm muss umfangreich kommentiert werden, andernfalls wird eine spätere Fehlersuche oder Veränderung des Programmes sehr zeitaufwendig. Es ist jedoch keinesfalls sinnvoll jeden Befehl in seiner Wirkung zu beschreiben, z.B.:

```
add    $t1, $t2, $t1      # $t1 := $t2 + $t1
```

Diese Information ist redundant, da sie unmittelbar aus der Programmzeile entnommen werden kann. Zweckmäßiger ist eine Information über den Sinn und den Zusammenhang der Zeile, z.B.:

```
add    $t1, $t2, $t1      # summe := messung + summe;
```

Hier bietet sich auch die Möglichkeit das eingangs erwähnte Hochsprachenprogramm zu kopieren und dieses dann Schritt für Schritt zu übersetzen.

(3) Assemblerprogramme werden etwas übersichtlicher, wenn sie in Spaltenform geschrieben werden. In der ersten Spalte (bis zum ersten Tabulatorstop) stehen alle Marken. Dadurch sind Sprungziele und Variablen schnell zu finden. In der zweiten Spalte (bis zum zweiten Tabulatorstop) stehen alle Befehle. Vom dritten Tabulatorstop an folgen die Argumente, die Kommentare stehen hinter dem fünften oder sechsten Tabulatorstop.

(4) Die Namen von Marken sollten für sich sprechen, z.B. `for:` am Anfang einer Zählschleife. Einfache Nummerierungen wie `a:`, `b:` sind sinnlos.

(5) Die Zuordnung der Variablen des Hochsprachenprogramms zu den Registern sollte kommentiert werden, da dies eine häufige Fehlerquelle ist. Die Konventionen zur Verwendung der Register (siehe Abschnitt 2 auf Seite 7) sollten unbedingt eingehalten werden.

3.6 Aufbau eines Assemblerprogramms

Den prinzipiellen Aufbau eines Assemblerprogramms können wir dem folgenden Beispiel entnehmen:

Beispiel 1: Ein erstes Beispielprogramm ohne besonderen praktischen Wert

```

# FIRST.S
2 # berechnet den Umfang des Dreiecks mit den Kanten x, y, z
.data
4 x:      .word   12
  y:      .word   14
6 z:      .word   5
  U:      .word   0
8 .text
main:    lw      $t0, x          # $t0 := x
10       lw      $t1, y          # $t1 := y
        lw      $t2, z          # $t2 := z
12       add     $t0, $t0, $t1   # $t0 := x+y
        add     $t0, $t0, $t2   # $t0 := x+y+z
14       sw      $t0, U          # U := x+y+z
        li      $v0, 10        # EXIT
16       syscall

```

Zu beachten ist, dass die Marke `main:` immer vorhanden sein muss, da diese vom SPIM-Betriebssystem angesprungen wird. Es ist empfehlenswert, jedes Programm wie in dem Beispiel angegeben zu beenden. Den Befehl `syscall` werden wir in Abschnitt 7 auf Seite 27 kennen lernen.

Anzumerken ist noch, dass es egal ist, ob man zunächst das Datensegment und dann das Textsegment verwendet oder umgekehrt. Die Einträge in die beiden Segmente können auch gemischt erfolgen. Wesentlich ist nur, dass die Reihenfolge der Einträge ins Textsegment korrekt ist.

3.7 Notation der Befehle

Wenn wir neue Befehle kennen lernen, so führen wir sie mit einer Tabelle wie der folgenden ein:

Befehl	Argumente	Wirkung	Erläuterung
<code>div</code>	<code>Rd, Rs1, Rs2</code>	<code>Rd=INT(Rs1/Rs2)</code>	divide
<code>li</code> [Ⓟ]	<code>Rd, Imm</code>	<code>Rd=Imm</code>	load immediate
		<code>:= ori Rd, \$zero, Imm</code>	

In der Spalte „Befehl“ steht der Name (der symbolische Bezeichner) des Befehls. Pseudo-Befehle sind mit dem Zeichen [Ⓟ] markiert. In der nachfolgenden Zeile steht dann die Befehlsfolge, zu der der Pseudobefehl aufgelöst wird.

In der Spalte „Argumente“ stehen die Argumente, die der Assembler erwartet.

In der Spalte „Wirkung“ wird die Wirkung des Befehls, meistens algebraisch, beschrieben. Die Spalte „Erläuterung“ enthält den ausgeschriebenen Namen des Befehls auf englisch.

Die angegebenen Argumente geben auch Aufschluss über die Art der Argumente, `Rd` und `Rs` stehen beispielsweise für Register. Eine komplette Übersicht liefert Abbildung 2 auf der folgenden Seite.

Die Tabelle wird leichter verständlich, wenn wir die Adressierungsarten kennen gelernt haben. Die angehängten Kleinbuchstaben `s` oder `d` stehen für „source“

3. GRUNDPRINZIPIEN DER ASSEMBLERPROGRAMMIERUNG

Abbildung 2: Argumentarten

Notation	Erläuterung	Beispiel
Rd, Rs1, Rs2, Rs	Registeroperanden, d: destination (Ziel), s: source (Quelle)	\$t1, \$s3
Imm	Unmittelbar (Immediate)	25, 0x23
label, Adr	Hauptspeicheradressen	main, 4+feld, feld(\$t1), 3+feld(\$t1)

(Quelle) bzw. „destination“ (Ziel). Viele Befehle haben zwei Quellargumente, diese werden durch eine nachstehende 1 bzw. 2 gekennzeichnet.

Auf den Seiten 74 bis 82 befinden sich hilfreiche Übersichten zum Nachschlagen.

4 Datenhaltung I: ganze Zahlen und Zeichenketten

In längeren Programmen gibt es Variablen, die unterschiedlich lange und unterschiedlich oft benötigt werden. Einige werden nur ein einziges Mal benötigt um Zwischenergebnisse abzulegen, andere werden über lange Abschnitte immer wieder benötigt.

Wir unterscheiden daher zwischen *lang-* und *kurzlebigen* Variablen. Die langlebigen Variablen werden meist im Hauptspeicher gehalten, damit sie nicht wertvolle Register belegen. Die kurzlebigen werden während ihrer Lebensdauer meist in Registern gehalten.

lang- und kurzlebige Variablen

Unser SPIM kann Berechnungen ausschließlich auf Registern durchführen. Variablen, die im Speicher abgelegt sind, müssen also zunächst mit Ladebefehlen in Register geladen werden und nach erfolgten Berechnungen ggf. wieder zurückgespeichert werden. Eine solche Rechnerarchitektur nennt man *Load-Store-Architektur*. Im Gegensatz dazu gibt es Rechnerarchitekturen, die Berechnungen auch im Hauptspeicher zulassen, z.B. die Motorola-CPU's der 68er Serie und die Intel 80x86.

Load-Store-Architektur

Jedes Register des SPIM enthält ein 32-Bit-Wort. Man sagt auch es ist 32 Bit breit.

Registerbreite

4.1 Ganze Zahlen im Datensegment

4.1.1 Die `.word`-Direktive

Im Hauptspeicher legt die Direktive `.word` die nachfolgenden Zahlen in je ein Doppelwort. Sinnvollerweise wird diese Direktive im Datensegment (nach der `.data`-Direktive) verwendet. Die Anwendung im Textsegment ist zwar zulässig, aber äußerst schlechter Stil. Es besteht stets die Gefahr, dass durch ungeschickte Programmierung versucht wird diese Daten als Befehle auszuführen.

`.word`

Die Werte können dezimal oder hexadezimal angegeben werden. Hexadezimale Werte müssen mit „0x“ beginnen.

Um auf die im Hauptspeicher abgelegten Variable komfortabel zugreifen zu können, sollte zudem eine Marke vor die `.word`-Direktive gesetzt werden:

```
marke: .word 256 0x100
```

Diese Direktive würde also in die beiden nächsten noch unbenutzten 32-Bit-Wörter im Datensegment jeweils den Wert 256 (=0x100)ablegen.

4.1.2 Weitere Ganzzahltypen

Eng verwandt mit der `.word`-Direktive sind die Direktiven

`.half`, `.byte`

- `.byte` für 8-Bit-Zahlen und
- `.half` für 16-Bit-Zahlen.

Die Initialisierung und das Anlegen mehrerer Werte geschieht wie bei der `.word`-Direktive.

4. DATENHALTUNG I: GANZE ZAHLEN UND ZEICHENKETTEN

4.2 Zeichenketten im Datensegment: die `.asciiz`-Direktive

Neben Zahlen bilden Zeichen, also einzelne Buchstaben oder Ziffern, den zweiten grundlegenden Datentyp. Kein Speicher oder Prozessor kann Zeichen direkt speichern oder verarbeiten. Seit Anbeginn der Textverarbeitung im weitesten Sinne behilft man sich damit die Zeichen als Zahlen zu kodieren und dann diese Zahlen zu verarbeiten. Nur bei der Ein- und Ausgabe wird dann eine lesbare Form verwendet. Es gibt eine Reihe verschiedener solcher Zuordnungen (Codes), die sich jedoch im wesentlichen gleichen. Nur die Anordnung von Sonderzeichen (etwa der deutsche Umlaut „Ä“) ist höchst unterschiedlich. Standard ist der *ASCII-Code*⁴, der in seiner Erweiterung 256 Zeichen (1 Byte) umfaßt.

ASCII-Code

`.asciiz`

Texte werden mit der `.asciiz`-Direktive in den Speicher geschrieben:

```
marke: .asciiz "Hallo Welt"
```

Der Text wird in doppelte Anführungszeichen eingeschlossen. Während des Assemblierens werden die einzelnen Zeichen des Textes kodiert und je vier von ihnen in ein Doppelwort geschrieben. An das Ende des Textes wird das Zeichen mit dem Code 0 (Chr 0) gesetzt, damit bei der Textausgabe das Ende des Textes erkannt werden kann. Dieses Verfahren nennt man *Nullterminierung*. Ein anderes Verfahren, das von vielen Modula-2-Compilern verwendet wird, ist im ersten Byte (Zeichen) der Zeichenkette deren Länge abzulegen. Zeichenketten können bei diesem Verfahren maximal 255 Zeichen enthalten.

Nullterminierung

`.ascii`

Es gibt auch eine `.ascii`-Direktive, die das Chr 0 nicht anhängt. Sie ist beispielsweise hilfreich, wenn längere Texte im Assemblerprogramm auf mehrere Zeilen verteilt werden müssen. In diesem Fall teilen wir die Zeichenkette in mehrere Assemblerzeilen auf, wobei nur die letzte Zeile die Direktive `.asciiz` enthält und alle anderen Textteile mit `.ascii` abgelegt werden:

```
marke: .ascii "Der Mond ist aufgegangen, die goldnen Sternl"  
      .asciiz "ein prangen, am Himmel hell und klar."
```

Um besondere Zeichen im Speicher abzulegen, müssen die in Abbildung 3 auf dieser Seite aufgeführten Kombinationen verwendet werden.

Abbildung 3: Zeichenkombinationen

<code>\n</code> :	Neue Zeile
<code>\t</code> :	Sprung zum nächsten Tabulator
<code>\"</code> :	Das doppelte Anführungszeichen

Ebenso wie die `.word`-Direktive sollten die Direktiven `.asciiz` und `.ascii` nur im Datensegment verwendet werden.

Übungen

Übung 1

Setze das folgende Modula-2-Codefragment in MIPS-Code um:

```
VAR  
i, j:   CARDINAL;  
Summe: LONGCARD;  
s1:    ARRAY[0..20] OF CHAR;
```

⁴ASCII: American Standard Code for Information Interchange

4.3 Die Datenausrichtung im Datensegment

Verwende die gleichen Bezeichnernamen und Datentypen (CARDINAL: 16 Bit, LONGCARD: 32 Bit), initialisiere `i` und `j` mit 0, `Summe` mit 200 und `s1` mit dem Text „Alle meine Entchen“. Die Obergrenze der Modula-Zeichenkette ist für die Umsetzung hier irrelevant.

Übung 2

Durch eine Eingabemaske sollen Vor- und Nachname, Straße, Postleitzahl und Ort einer Person nacheinander erfaßt werden. Lege die dafür nötigen Texte jeweils einzeln im Datensegment ab.

4.3 Die Datenausrichtung im Datensegment

Die meisten Transferbefehle, die wir im folgenden Abschnitt kennen lernen werden, arbeiten nur auf *ausgerichteten Daten* (*aligned Data*). Ein Datum ist ausgerichtet, wenn seine Hauptspeicheradresse ein ganzzahliges Vielfaches seiner Größe ist.

aligned Data

Zum Beispiel ist ein Halbwort (2 Byte Größe) ausgerichtet, wenn es an einer geraden Hauptspeicheradresse liegt, ein Vollwort (4 Byte Größe), wenn es an einer durch vier teilbaren Hauptspeicheradresse liegt.

Man kann diese Datenausrichtung durch die Direktive `.align` beeinflussen. Für unsere Zwecke genügt es zu wissen, dass die Direktive `.word`, `.half` etc. die Daten stets richtig ablegen und wir uns daher nicht um deren Ausrichtung zu kümmern brauchen.

.align

5. TRANSFERBEFEHLE

5 Transferbefehle

Wie im vorangegangenen Abschnitt erwähnt hat der SPIM eine Load-Store-Architektur. Wollen wir also mit denen im Datensegment abgelegten Daten arbeiten, so müssen wir sie zunächst in ein Register im Prozessor laden. Hierfür gibt es eine Reihe von Befehlen, die wir jetzt kennen lernen werden.

5.1 Ladebefehle

Es gibt eine recht große Zahl von Ladebefehlen für verschiedene Datentypen. Der am häufigsten verwendete Befehl ist sicherlich der Befehl `lw`:

Befehl	Argumente	Wirkung	Erläuterung
<code>lw</code>	<code>Rd, Adr</code>	<code>Rd:=MEM[Adr]</code>	Load word

`lw` lädt den Wert, der an der durch `Adr` angegebenen Adresse im Hauptspeicher steht, in das Register `Rd`. Die Möglichkeiten, mit denen man diese Adresse angeben kann, lernen wir sofort kennen:

5.1.1 Adressierungsmodi I: Register Indirekte, direkte und indexierte Adressierung

Der **MIPS** kennt mehrere Adressierungsmodi, von denen uns im Zusammenhang mit Ladebefehlen nur drei interessieren:

Register-indirekt

- `(Rs)`: Der gesuchte Wert steht im Hauptspeicher an der Adresse, die im Register `Rs` angegeben ist (*Register-indirekt*).

direkt

- `label` oder `label+Konstante`: Der gesuchte Wert steht im Hauptspeicher an der Adresse der angegebenen Marke, die ggf. zuvor um die Konstante erhöht oder verringert wird (*direkt*).

indexiert

- `label(Rs)` oder `label+Konstante(Rs)`: Der gesuchte Wert steht im Hauptspeicher an der Adresse der angegebenen Marke plus dem Inhalt des Registers `Rs`, die Konstante wird ggf. zuvor addiert oder subtrahiert (*indexiert*).

Dies sind die drei Möglichkeiten, durch `Adr` angegebene Parameter zu notieren.

Die vorangegangenen drei Adressierungsarten werden vom Assembler alle auf die indexierte Adressierung zurückgeführt. So wird bei direkter Adressierung das Register `$zero` mit dem Wert 0, bei der Register indirekten Adressierung für den Wert der Marke der Wert 0 verwendet. Die Konstanten werden zur Assemblerzeit bzw. beim Binden des Programmes, also vor der Programmausführung, ausgewertet.

Beispiel 2: Beispiele zur direkten Adressierung

```
# ADRMODEL.S
2 .data
  var:      .word   20, 4, 22, 25, 7
4 #
  .text
6 main:    lw  $t1, var           # $t1 enthaelt "20" (direkte Adr.)
          lw  $t1, var+4       # $t1 enthaelt "4" (direkte Adr.)
8          lw  $t2, var($t1)   # $t2 enthaelt "4" (indexierte Adr.)
          lw  $t2, var+8($t1) # $t2 enthaelt "25" (indexierte Adr.)
```

Übungen

Übung 3

Lade den Wert 7 aus Beispiel 2 auf der vorangegangenen Seite in das Register
`$t4 lw $t4, 16+var`

Übung 4

Register-indirekte und direkte Operanden werden vom Assembler auf die indirekte Adressierung umgesetzt. Gebe einen Algorithmus hierfür an und überprüfe ihn anhand eines Beispielprogramms. $(Rs) \rightarrow 0(Rs)$

`label → label($zero)`

`label+(Konstante) → (label+(Konstante))($zero)`

Der hohe Wert, den die Marken zugewiesen bekommen, erklärt sich daraus, dass das Datensegment nicht bei der Speicheradresse 0 beginnt.

5.1.2 Weitere Ladebefehle

Die folgenden Ladebefehle arbeiten analog zu `lw`:

Befehl	Argumente	Wirkung	Erläuterung
<code>lb</code>	<code>Rd, Adr</code>	<code>Rd:=MEM[Adr]</code>	Load byte
<code>lbu</code>	<code>Rd, Adr</code>	<code>Rd:=MEM[Adr]</code>	Load unsigned byte
<code>lh</code>	<code>Rd, Adr</code>	<code>Rd:=MEM[Adr]</code>	Load halfword
<code>lhu</code>	<code>Rd, Adr</code>	<code>Rd:=MEM[Adr]</code>	Load unsigned halfword
<code>ld[Ⓟ]</code>	<code>Rd, Adr</code>	Lädt das Doppelwort an der Stelle <code>Adr</code> in die Register <code>Rd</code> und <code>Rd+1</code> <code>:= lw Rd, Adr</code> <code>lw Rd+1, Adr+4</code>	Load double-word

Die Ladebefehle für Halbwörter (16 Bit) und Bytes (8 Bit) gibt es auch in einer *unsigned*-Version. Die normalen Befehle (`lb`, `lh`) wandeln automatisch negative 8- bzw. 16-Bit-Zahlen in Doppelwörter um: aus dem Byte `0xF0` (= -16) wird dann `0xFFFF FFF0`. Diese Umwandlung ist erwünscht, wenn es sich tatsächlich um vorzeichenbehaftete Zahlen handelt, aber unerwünscht, wenn die gespeicherten Zahlen vorzeichenlos sind, wenn also `0xF0` für die Zahl 240 steht. In diesem Fall soll im Register nach dem Laden eben `0x0000 00F0` stehen. Hierfür stehen die *unsigned-Befehle* zur Verfügung, die Bytes und Halbwörter unverändert laden.

unsigned

unsigned-Befehle

Die oben genannten Ladebefehle gehen davon aus, dass die Daten ausgerichtet (*aligned*) sind. Die Adressen von Wörtern müssen also immer durch vier teilbar sein, andernfalls wird das Programm zur Laufzeit abgebrochen. Die Adressen von Halbwörtern müssen dementsprechend gerade sein. Gelegentlich sollen aber auch Wörter oder Halbwörter über ihre Grenzen hinweg geladen werden. Hierfür stehen dem fortgeschrittenen Assemblerprogrammierer die folgenden Befehle zur Verfügung:

aligned

Befehl	Argumente	Wirkung	Erläuterung
<code>ulw</code>	<code>Rd, Adr</code>	<code>Rd:=MEM[Adr]</code>	unaligned Load word
<code>ulh</code>	<code>Rd, Adr</code>	<code>Rd:=MEM[Adr]</code>	unaligned Load halfword
<code>ulhu</code>	<code>Rd, Adr</code>	<code>Rd:=MEM[Adr]</code>	unaligned Load unsigned halfword

5. TRANSFERBEFEHLE

Befehl	Argumente	Wirkung	Erläuterung
lwr	Rd, Adr	Rd:=MEM[Adr] DIV 2^{16}	Load word right
lwl	Rd, Adr	Rd:=MEM[Adr] MOD 2^{16}	Load word left

Übungen

Übung 5

Lege im Datensegment die Werte 0xFFFF 0000 und 0x0000 FFFF ab und lade sie mit den Befehlen lb, lbu, lh und lhu. Erkläre die Ergebnisse!

5.2 Speicherbefehle

Nachdem wir Werte in den Hauptspeicher geladen haben und mit ihnen Berechnungen angestellt haben, wollen wir sie häufig wieder in den Hauptspeicher zurückschreiben um später nochmals auf sie zugreifen zu können.

Analog zu den Ladebefehlen gibt es für jeden Datentyp ein oder mehrere Speicherbefehle. Der wichtigste ist der Befehl sw:

Befehl	Argumente	Wirkung	Erläuterung
sw	Rs, Adr	MEM[Adr]:=Rs	store word

Er speichert den Wert des Registers Rs an der angegebenen Stelle im Hauptspeicher. Beachte, dass der Zieloperand diesmal an *zweiter* Stelle steht und nicht wie bisher an *erster*.

Adressierungsmodi

Als *Adressierungsmodi* stehen wieder die unter Abschnitt 5.1.1 auf Seite 18 genannten Modi zur Verfügung.

Weitere Speicherbefehle sind:

Befehl	Argumente	Wirkung	Erläuterung
sb	Rs, Adr	MEM[Adr]:=Rs MOD 256	store byte
sh	Rs, Adr	MEM[Adr]:=Rs MOD 2^{16}	store halfword
sd [Ⓣ]	Rs, Adr	MEM[Adr]:=Rs + 2^{16} Rs+1 := sw Rd, Adr sw Rd+1, Adr+4	store double-word

Wie bei der ersten Gruppe der Ladebefehle müssen auch bei diesen Speicherbefehlen die Adressen ausgerichtet sein. Aber auch unter den Speicherbefehlen gibt es Befehle, die nicht ausgerichtete Adressen verarbeiten können:

Befehl	Argumente	Wirkung	Erläuterung
swl	Rs, Adr	MEM[Adr]:=Rs MOD 2^{16}	store word left
swr	Rs, Adr	MEM[Adr]:=Rs MOD 2^{16}	store word right

Befehl	Argumente	Wirkung	Erläuterung
ush	Rs, Adr	MEM[Adr]:=Rs MOD 2^{16}	unaligned store halfword
usw	Rs, Adr	MEM[Adr]:=Rs	unaligned store word

5.3 Register-Transfer-Befehle

Außer dem Laden von Daten aus dem Hauptspeicher in Register (Abschnitt 5.1) und dem Speichern von Registerinhalten in den Hauptspeicher (Abschnitt 5.2) wollen wir häufig Daten in einem Register manipulieren, ohne auf den Hauptspeicher zuzugreifen. Dies ermöglichen uns die folgenden Befehle:

Befehl	Argumente	Wirkung	Erläuterung
move ^⑤	Rd, Rs	Rd:=Rs := addu Rd \$zero Rs	move
li ^⑤	Rd, Imm	Rd:=Imm := ori Rd, \$zero, Imm oder: lui \$at, Imm DIV 2^{16} ori Rd, \$at, Imm MOD 2^{16}	load immediate
lui	Rs, Imm	Rd:=Imm 2^{16}	load upper immediate

Die in den Befehlen li und lui bereits auftretende *unmittelbare (immediate)* Adressierung lernen wir in Abschnitt 6.1 auf der folgenden Seite kennen.

Der Vollständigkeit halber seien an dieser Stelle noch vier Befehle für die beiden bisher noch nicht erwähnten Register hi und lo genannt. Die genannten Register werden für Multiplikation und Division benutzt.

Befehl	Argumente	Wirkung	Erläuterung
mfhi	Rd	Rd:=hi	move from hi
mflo ⁵	Rd	Rd:=lo	move from lo
mthi	Rs	hi:=Rs	move to hi
mtlo	Rs	lo:=Rs	move to lo

Übungen

Übung 6

Lege im Speicher je ein Wort, ein Halbwort und ein Byte an. Speichere mit Hilfe des Befehls li und den Speicherbefehlen den Wert 20 nacheinander mit den passenden Speicherbefehlen.

Übung 7

Was passiert wenn du einen zu großen Wert, z.B. 256 mit sb speicherst? Und was passiert, wenn du 256 mit sw auf ein mit .byte angelegte Speicherstelle schreibst? Was ändert sich daran, wenn du .byte als erste Direktive ins Daten-segment schreibst?

⁵[Pat, A-65] nennt den falschen Befehl mfloi und eine falsche Erläuterung

6. ARITHMETISCHE BEFEHLE (GANZE ZAHLEN)

6 Arithmetische Befehle (ganze Zahlen)

In den beiden vorangegangenen Abschnitten haben wir gelernt, wie man Daten im Hauptspeicher anlegt und sie zwischen dort und den Registern der CPU hin und her bewegt. Wir wenden uns nun der arithmetischen Verarbeitung der Daten zu.

6.1 Adressierungsmodi II: Register direkte und unmittelbare Adressierung

Wie bereits verschiedentlich erwähnt, kann der **MIPS** Operationen ausschließlich auf Registern durchführen. Kein arithmetischer Befehl hat als Parameter Hauptspeicheradressen! Die aus Abschnitt 5.1.1 auf Seite 18 bekannten Adressierungsmodi reichen uns nun also nicht mehr aus.

Register-direkte
Adressierung

Zunächst benötigen wir einen Modus, der es uns erlaubt auf den Inhalt eines Registers direkt zuzugreifen. Wir nennen ihn *Register-direkte Adressierung* (*register direct*). Register-direkte Operanden werden durch Angabe des Registernamens oder der Registernummer notiert. Derartige Operanden sind in den Befehlstabellen mit Rd , Rs , $Rs1$ und $Rs2$ bezeichnet.

unmittelbare
Adressierung

Außerdem erscheint es zweckmäßig Konstanten direkt verwenden zu können ohne sie erst aufwendig in ein Register schreiben zu müssen. Diese Adressierungsart nennen wir *unmittelbare Adressierung* oder *immediate*. Diese Operanden sind in den Befehlstabellen mit Imm bezeichnet. Unmittelbare Adressierung ist nur im letzten Operanden möglich.

6.2 Addition und Subtraktion

Die folgenden drei Befehle bedürfen keiner weiteren Erklärung:

Befehl	Argumente	Wirkung	Erläuterung
add	$Rd, Rs1, Rs2$	$Rd := Rs1 + Rs2$	addition (with overflow)
addi	Rd, Rs, Imm	$Rd := Rs + Imm$	addition immediate (with overflow)
sub	$Rd, Rs1, Rs2$	$Rd := Rs1 - Rs2$	subtract (with overflow)

Obwohl es einen eigenen Befehl `addi` gibt, können wir auch beim Befehl `add` unmittelbare Operanden verwenden. Der Assembler verwendet dann automatisch den korrekten Befehl `addi`. Auch den Befehl `sub` können wir mit unmittelbaren Operanden verwenden. Der Assembler lädt in diesem Fall den Wert zunächst in sein `$at`-Register und führt dann die Subtraktion aus. Diese praktische Umwandlung geschieht auch bei den anderen arithmetischen Befehlen automatisch.

Ausnahmebehandlung

Wenn bei den zuvor genannten Befehlen ein Überlauf auftritt (Summe zweier positiver Zahlen wird negativ o.ä.), wird eine *Ausnahmebehandlung* (exception handler) aufgerufen, die das Programm abbricht. Unter Ausnahmen versteht man Fehler, die zur Laufzeit des Programmes auftreten und die vom Programm selbst verursacht werden. Das klassische Beispiel für eine Ausnahme ist die Division durch Null, eine Aufgabe, die der Prozessor nicht lösen kann. Von der Ausnahme abzugrenzen ist die *Unterbrechung*, die extern erfolgt, beispielsweise der Warmstart des Rechners durch eine bestimmte Tastenkombination. Mehr

Unterbrechung

über Unterbrechungen und Ausnahmen und deren Behandlung lernen wir in Abschnitt 14 auf Seite 64.

Zurück zu unseren Additionsbefehlen und den dabei eventuell auftretenden Überläufen. Handelt es sich bei den Zahlen um vorzeichenlose Zahlen oder soll der Überlauf anders bearbeitet werden, so verwenden wir einen der folgenden Befehle:

Befehl	Argumente	Wirkung	Erläuterung
addu	Rd, Rs1, Rs2	Rd := Rs1 + Rs2	addition (without overflow)
addiu	Rd, Rs, Imm	Rd := Rs + Imm	addition immediate (without overflow)
subu	Rd, Rs1, Rs2	Rd := Rs1 - Rs2	subtract (without overflow)

Der Buchstabe u scheint zwar auf 'unsigned' hinzuweisen, aber die Beschreibung 'overflow' ist treffender, da auch bei vorzeichenlosen Zahlen Überläufe auftreten können. Durch die Vernachlässigung der Überläufe werden diese Befehle aber eben auch für vorzeichenlose Arithmetik brauchbar. Deutlich wird dies, wenn man versucht, zur größten vorzeichenbehafteten Zahl 1 hinzuzugaddieren. Als Beispiel betrachten wir den übersichtlicheren Fall von Halbbytes:

$$\begin{array}{r}
 a = \quad 0111 \\
 b = \quad 0001 \quad + \\
 \hline
 \text{carry} = 01110 \\
 s = \quad 1000 \\
 \hline
 \hline
 \end{array}$$

Das Ergebnis +8 kann als vorzeichenbehaftete Zahl nicht dargestellt werden, es wird als -8 interpretiert und ist somit falsch. Haben wir jedoch mit vorzeichenlosen Zahlen gerechnet, so ist das Ergebnis richtig.

Übungen

Übung 8

Als erstes komplettes Programm sollst du ein Programm schreiben, das den Term $(w + x + 3) - (y + z - 5)$ ohne diesen zuvor zu optimieren berechnet. Halte die Operanden w, x, y, z als vorzeichenbehaftete von dir zur Assemblerzeit initialisierte ganze Zahlen im Hauptspeicher und lege dein Ergebnis in einem fünften Speicherplatz ab.

```

1 # FIRST.S
2 # berechnet den Umfang des Dreiecks mit den Kanten x, y, z
3
4 .data
5 x:      .word   12
6 y:      .word   14
7 z:      .word   5
8 U:      .word   0
9
10 .text
11 main:   lw      $t0, x           # $t0 := x
12         lw      $t1, y           # $t1 := y
13         lw      $t2, z           # $t2 := z
14         add     $t0, $t0, $t1     # $t0 := x+y
15         add     $t0, $t0, $t2     # $t0 := x+y+z
16         sw      $t0, U           # U := x+y+z
17         li      $v0, 10          # EXIT
18         syscall

```

Übung 9

Ändere das Programm aus Übung 8 so, dass es mit 8-Bit-Zahlen arbeitet

6. ARITHMETISCHE BEFEHLE (GANZE ZAHLEN)

Übung 10

Initialisiere dein Programm aus Übung 9 mit $w = y = 128$ und $x = z = 127$. Wie lautet das korrekte Ergebnis und was passiert bei der Programmausführung?

Übung 11

Schreibe dein Programm aus Übung 9 so um, dass es für den Fall $w \approx y, x \approx z, w > 100, z > 100$ etwas sicherer läuft.

6.3 Multiplikation und Division

Der **MIPS** verfügt lediglich über die folgenden vier Maschinenbefehle, mit denen alle anderen Divisions- und Multiplikationsbefehle realisiert werden:

Befehl	Argumente	Wirkung	Erläuterung
div	Rd, Rs	hi:=Rd MOD Rs; lo:=Rd DIV Rs	Divide (with overflow)
divu	Rd, Rs	hi:=Rd MOD Rs; lo:=Rd DIV Rs	Divide (without overflow)
mult	Rd, Rs	hi:=Rd \times Rs DIV 2^{16} ; lo:=Rd \times Rs MOD 2^{16}	multiply
multu	Rd, Rs	hi:=Rd \times Rs DIV 2^{16} ; lo:=Rd \times Rs MOD 2^{16}	Unsigned multiply

Alle Befehle verwenden zwei Register, die wir bisher noch nicht benutzt haben: hi(gh) und lo(w). Sie gehören nicht zu den 32 Registern, die wir mit dem \$-Zeichen adressieren können. Es gibt aber zwei Befehle (mfhi und mflo), mit denen wir ihren Inhalt in ein normales Register kopieren können. Wir haben sie bereits in Abschnitt 5.3 auf Seite 21 kennen gelernt.

Der **MIPS**-Assembler stellt einige Pseudoinstruktionen zur Verfügung, die das Kopieren der Ergebnisse aus den Registern lo und hi übernehmen:

Befehl	Argumente	Wirkung	Erläuterung
mul ^(P)	Rd, Rs1, Rs2	Rd:=Rs1 \times Rs2 := mult Rs1, Rs2 mflo Rd	Multiply (without overflow)
multo ^(P)	Rd, Rs1, Rs2	Rd:=Rs1 \times Rs2 := mult Rs1, Rs2 mfhi \$at beq \$at, \$zero 4 break \$zero mflo Rd	Multiply (with overflow)
mulou ^(P)	Rd, Rs1, Rs2	Rd:=Rs1 \times Rs2 := multu Rs1, Rs2 mfhi \$at beq \$at, \$zero 4 break \$zero mflo Rd	Unsigned multiply (with overflow)

Und für die Division:

Befehl	Argumente	Wirkung	Erläuterung
div ^(P)	Rd, Rs1, Rs2	Rd:=Rs1 DIV Rs2	Divide (with overflow)

6.4 Sonstige arithmetische Befehle

Befehl	Argumente	Wirkung	Erläuterung
		:= bne Rs2, \$zero, 4 break \$zero div Rs1, Rs2 mfl Rd	
divu ^(P)	Rd, Rs1, Rs2	Rd:=Rs1 DIV Rs2 := bne Rs2, \$zero, 4 break \$zero divu Rs1, Rs2 mfl Rd	Divide (without overflow)

Der Befehl `div` kann also zwei oder drei Parameter haben, je nachdem ob der Pseudobefehl oder der Maschinenbefehl vom Anfang dieses Abschnitts gemeint ist.

Werfen wir einen Blick auf die Übersetzung dieser Befehle. Das Grundprinzip ist bei allen diesen Pseudobefehlen dasselbe: Das Ergebnis wird zunächst mit einem der Maschinenbefehle berechnet. Danach befinden sich in den Registern `lo` und `hi` die Teilergebnisse. In Abhängigkeit von diesen Werten wird dann eine Ausnahmebehandlung ausgelöst oder nicht.

Hierfür werden die Befehle `bne` (branch if not equal) und `beq` (branch if equal) verwendet, die wir in Abschnitt 9.1 auf Seite 35 kennen lernen werden. Hier genügt erst einmal deren Effekt: Wenn der Inhalt der beiden Register gleich bzw. ungleich ist, werden die nächsten `n` Befehle des Programms übersprungen, wobei `n` der dritte Parameter der Befehle ist und durch vier teilbar sein muss.⁶ Die erste Zeile der Übersetzung des Befehls `div` überspringt also den folgenden Befehl, falls der Teiler Null ist. Womit wir einen weiteren neuen Befehl kennen lernen: `break`, er wird in Abschnitt 14 auf Seite 64 näher erläutert. Er verursacht den Aufruf einer Ausnahmebehandlungsroutine, die die Programmausführung nach der Ausgabe einer entsprechenden Meldung stoppt.

Übungen

Übung 12

Schreibe ein Programm, das den Umfang und den Flächeninhalt eines Rechteckes berechnet. Die Seiten des Rechteckes sollen im Hauptspeicher gehalten werden, die Ergebnisse dort abgelegt werden.

6.4 Sonstige arithmetische Befehle

Befehl	Argumente	Wirkung	Erläuterung
abs ^(P)	Rd, Rs	Rd:=ABS (Rs) := add Rd, \$zero, Rs bgez Rd 4 sub Rd, \$zero, Rs	Absolute value
neg ^(P)	Rd, Rs	Rd:=-Rs := sub Rd, \$zero, Rs	Negate value (with overflow)
negu ^(P)	Rd, Rs	Rd:=-Rs := subu Rd, \$zero, Rs	Negate value (without overflow)

⁶Diese Erklärung ist nicht ganz korrekt, auf Seite 37 werden wir mehr darüber erfahren!

6. ARITHMETISCHE BEFEHLE (GANZE ZAHLEN)

Befehl	Argumente	Wirkung	Erläuterung
<code>rem</code> [Ⓟ]	<code>Rd, Rs1, Rs2</code>	<code>Rd:=Rs1 MOD Rs2</code> <pre>:= bne Rs2, \$zero, 4 break \$zero div Rs1, Rs2 mfhi Rd</pre>	Remainder
<code>remu</code> [Ⓟ]	<code>Rd, Rs1, Rs2</code>	<code>Rd:=Rs1 MOD Rs2</code> <pre>:= bne Rs2, \$zero, 4 break \$zero divu Rs1, Rs2 mfhi Rd</pre>	Unsigned remainder

Warum gibt es zwei Negationsbefehle? In der Tat gibt es eine Zahl, die man nicht negieren kann: die kleinste darstellbare Zahl, beim SPIM also `0x8000 0000`. Dies liegt daran, dass jede Kombination von Bits eine gerade Anzahl von Kombinationsmöglichkeiten hat. Eine dieser Möglichkeiten wird aber für den Wert 0 benötigt, so dass für den negativen und den positiven Zahlraum unterschiedlich viele Zahlen übrig bleiben. Im 2er-Komplement führt dies dazu, dass der Betrag der kleinsten darstellbaren Zahl um eins größer als die größte darstellbare Zahl ist. Aus diesem Grunde kann auch bei der Verwendung des Befehls `abs` ein Überlauf auftreten.

Die Befehle `rem` und `remu` berechnen den Rest bei der ganzzahligen Division.

7. DAS SPIM-BETRIEBSSYSTEM (EIN- UND AUSGABE)

Übungen

Übung 13

Ergänze deine Lösung aus Übung 11 so, dass die Werte vom Benutzer eingegeben werden können. Die Haltung der Werte w bis z im Hauptspeicher entfällt also. Gebe das Ergebnis mit einem Hinweistext aus.

Übung 14

Ergänze deine Lösung aus Übung 12 so, dass die Werte vom Benutzer eingegeben werden können. Gebe die Ergebnisse mit einem Hinweistext aus.

Abbildung 4: Betriebssystemfunktionen des SPIM (system calls)

Funktion	Code in \$v0	Argument(e)	Ergebnis
print_int	1	Wert in \$a0 wird dezimal ausgegeben	
print_float	2	Wert in \$f12/13 ¹ wird als 32-Bit-Gleitkommazahl ausgegeben	
print_double	3	Wert in \$f12 ¹ wird als 64-Bit-Gleitkommazahl ausgegeben	
print_string	4	Die mit Chr 0 terminierte Zeichenkette, die an der Stelle (\$a0) beginnt, wird ausgegeben	
read_int	5		Die auf der Konsole dezimal eingegebene ganze Zahl in \$v0
read_float	6		Die auf der Konsole dezimal eingegebene 32-Bit-Gleitkommazahl in \$f0 ²
read_double	7		Die auf der Konsole dezimal eingegebene 64-Bit-Gleitkommazahl in \$f0/1 ²
read_string	8	Adresse, ab der die Zeichenkette abgelegt werden soll in \$a0, maximale Länge der Zeichenkette in \$a1	Speicher von (\$a0) bis (\$a0)+\$a1 wird mit der eingelesenen Zeichenkette belegt.
sbrk	9	Größe des Speicherbereichs in Bytes in \$a0	Anfangsadresse eines freien Blocks der geforderten Größe in \$v0
exit	10		

¹: \$f12 und \$f13 sind Register des Koprozessors 1, siehe 15)
²: \$f0 und \$f1 sind Register des Koprozessors 1, siehe 15)

8. LOGISCHE BEFEHLE

8 Logische Befehle

In den vorangegangenen Abschnitten haben wir gelernt, Daten zwischen Hauptspeicher und Prozessor hin und her zu bewegen und mit ihnen Arithmetik zu betreiben. Wir werden uns nun den logischen Operationen zuwenden.

bitweise Bearbeitung

Der aus Hochsprachen bekannte Datentyp `boolean` ist auf Assembler-Ebene nicht vorhanden. Die kleinste Zuordnungseinheit im Hauptspeicher ist 1 Byte, auf Prozessorebene gar ein ganzes Wort mit 4 Bytes. Die logischen Befehle arbeiten wie die arithmetischen alle auf Registern. Die Register werden *bitweise* bearbeitet:

$$\text{OP R1, R2} := (\text{R1}_0 \text{ OP R2}_0)(\text{R1}_1 \text{ OP R2}_1) \dots (\text{R1}_{31} \text{ OP R2}_{31})$$

z.B.:

$$0101_2 \wedge 1100_2 = 0100_2$$

8.1 Elementare logische Befehle

Die folgenden Operationen implementieren die Basisoperationen, deren Funktionalität bekannt sein sollte:

Befehl	Argumente	Wirkung	Erläuterung
<code>and</code>	<code>Rd, Rs1, Rs2</code>	$\text{Rd} := \text{Rs1} \wedge \text{Rs2}$	<code>and</code>
<code>andi</code>	<code>Rd, Rs, Imm</code>	$\text{Rd} := \text{Rs} \wedge \text{Imm}$	<code>and immediate</code>
<code>nor</code>	<code>Rd, Rs1, Rs2</code>	$\text{Rd} := \overline{\text{Rs1}} \vee \overline{\text{Rs2}}$	<code>nor</code>
<code>or</code>	<code>Rd, Rs1, Rs2</code>	$\text{Rd} := \text{Rs1} \vee \text{Rs2}$	<code>or</code>
<code>ori</code>	<code>Rd, Rs, Imm</code>	$\text{Rd} := \text{Rs} \vee \text{Imm}$	<code>or immediate</code>
<code>xor</code>	<code>Rd, Rs1, Rs2</code>	$\text{Rd} := \text{Rs1} \oplus \text{Rs2}$	<code>exclusive or</code>
<code>xori</code>	<code>Rd, Rs, Imm</code>	$\text{Rd} := \text{Rs} \oplus \text{Imm}$	<code>exclusive or immediate</code>
<code>not</code> [Ⓣ]	<code>Rd, Rs</code>	$\text{Rd} := \overline{\text{Rs}}$:= <code>xori Rd, Rs, -1</code>	<code>not</code>

Übungen

Übung 15

Der Pseudobefehl `move` wird mit Hilfe des Befehls `addu` übersetzt. Welche Übersetzungen wären noch denkbar?

Übung 16


Wie könnte man einen Pseudobefehl `swap` definieren, der den Inhalt zweier Register austauscht? Verzichte auf das `$at`-Register und beweise die Korrektheit deines Befehls! Tip: Verwende nur den Befehl `xor`.

8.2 Rotations- und Schiebefehle

Rotations- und Schiebefehle

Vielfach wird bei der Assemblerprogrammierung auf sogenannte *Rotations-* und *Schiebefehle* zurückgegriffen. Sie manipulieren die Reihenfolge der Bits in einem Register. Beim Rotieren werden die Bits um einen gewählten Betrag nach links oder rechts verschoben, die überzähligen Bits am anderen Ende jedoch wieder angehängt. Bei Schiebefehlen werden die überzähligen Bits ignoriert

Abbildung 5: Rotations- und Schiebepfehle



und die freien Bits mit Nullen oder Einsen aufgefüllt. Das Prinzip wird in der ersten Spalte der Abbildung 5 auf dieser Seite dargestellt.

Genutzt werden können die Rotations- und Schiebepfehle beispielsweise bei der Auswertung von Schnittstellen.

Werfen wir zunächst einen Blick auf die uns zur Verfügung stehenden Rotationsbefehle:

Befehl	Argumente	Wirkung	Erläuterung
rol ^(P)	Rd, Rs1, Rs2	Rd := Rs1 um Rs2 Stellen nach links rotiert	rotate left
		:= subu \$at, \$zero, Rs2 srlv \$at, Rs1, \$at sllv Rd, Rs2, Rs1 or Rd, Rd, \$at	
ror ^(P)	Rd, Rs1, Rs2	Rd := Rs1 um Rs2 Stellen nach rechts rotiert	rotate right
		:= subu \$at, \$zero, Rs2 srlv \$at, \$at, Rs1 srlv Rd, Rs1, Rs2 or Rd, Rd, \$at	

Bei den Schiebepfehlen müssen wir zwischen *logischen* und *arithmetischen* Befehlen differenzieren. Der Unterschied liegt in der Behandlung der hochwertigen Bits. Bei vorzeichenbehafteten Zahlen repräsentiert das höchstwertige Bit das Vorzeichen. Die logischen Schiebepfehle füllen die freigewordenen Bits mit Nullen auf, während der arithmetische Schiebepfehl das höchstwertige Bit der Quelle übernimmt. In vielen Fällen soll das Vorzeichen jedoch erhalten bleiben, wir müssen dann einen der arithmetischen Schiebepfehle wählen. Es gibt nur einen arithmetischen Schiebepfehl (*sra*), da beim Schieben nach links das Vorzeichen nicht beachtet werden muss. (Selber ausprobieren mit 0100, 1000 und 1101!)

logische und
arithmetische
Schiebepfehle

8. LOGISCHE BEFEHLE

Alle Schiebepfehle liegen zudem in einer Form für fixe Verschiebungen, und in einer für variable durch ein Register angegebene Verschiebungen vor. Die variablen Befehle sollten nur verwendet werden, wenn der Betrag der Verschiebung tatsächlich erst zur Laufzeit bestimmt wird. Bei den fixen Befehlen sind für den Verschiebungsbetrag 5 Bit, also die Werte 0 bis 32, vorgesehen. Andere Werte werden beim Assemblieren umgerechnet.

Befehl	Argumente	Wirkung	Erläuterung
sll	Rd, Rs, Imm	$Rd := Rs \times 2^{Imm}$	Shift left logical
sllv	Rd, Rs1, Rs2	$Rd := Rs1 \times 2^{Rs2}$	Shift left logical variable
srl	Rd, Rs, Imm	$Rd := Rs \text{ DIV } 2^{Imm}$	Shift right logical
srlv	Rd, Rs1, Rs2	$Rd := Rs1 \text{ DIV } 2^{Rs2}$	Shift right logical variable
sra	Rd, Rs, Imm	$Rd := Rs \times 2^{Imm}$	Shift right arithmetic
srav	Rd, Rs1, Rs2	$Rd := Rs1 \times 2^{Rs2}$	Shift right arithmetic variable

8.3 Vergleichsbefehle

Dieser Abschnitt soll kein Vorgriff auf Abschnitt 9 auf Seite 35 sein, obwohl die Vergleichsbefehle auch in Kontrollstrukturen Anwendung finden. Für kleinste Entscheidungen sind die folgenden Befehle jedoch oft auch ohne größere Kontrollstrukturen nützlich.

Der **MIPS** verfügt nur über Vergleichsbefehle, die die Relation „<“ realisieren. Im nächsten Abschnitt werden wir noch Maschinenbefehle kennen lernen, die auch die Relation „=“ sowie Vergleiche mit 0 ermöglichen. Diese spartanische Ausstattung wird durch einige unterschiedlich aufwendige Pseudobefehle so erweitert, dass uns alle Relationen zur Verfügung stehen.

Die Vergleichsbefehle setzen alle das Zielregister auf 1, falls ihre Bedingung erfüllt ist, sonst auf 0. Sie sind zwar alle nur mit register-direkter Adressierung definiert, aber der SPIM-Assembler übersetzt unmittelbare Operanden, indem er sie zuvor in das \$at-Register lädt. Die in der folgenden Tabelle genannten Übersetzungen der Pseudobefehle beziehen sich jedoch nur auf die rein register-direkten Adressierungen.

Tests auf Gleichheit:

Befehl	Argumente	Wirkung	Erläuterung
seq ^(P)	Rd, Rs1, Rs2	Rd := 1, falls Rs1=Rs2, 0 sonst	(=) set equal
		:= beq Rs2, Rs1, 8 ori Rd, \$zero, 0 beq \$zero, \$zero, 4 ori Rd, \$zero, 1	
sne ^(P)	Rd, Rs1, Rs2	Rd := 1, falls Rs1≠Rs2, 0 sonst	(≠) set not equal
		:= beq Rs2, Rs1, 8 ori Rd, \$zero, 1 beq \$zero, \$zero, 4 ori Rd, \$zero, 0	

Tests auf größer:

8.3 Vergleichsbefehle

Befehl	Argumente	Wirkung	Erläuterung
<code>sge^(P)</code>	<code>Rd, Rs1, Rs2</code>	<code>Rd := 1, falls Rs1 ≥ Rs2, 0 sonst</code> <code>:= bne Rs2, Rs1, 8 ori Rd, \$zero, 1 beq \$zero, \$zero, 4 slt Rd, Rs2, Rs1</code>	(\geq) set greater than equal
<code>sgeu^(P)</code>	<code>Rd, Rs1, Rs2</code>	<code>Rd := 1, falls Rs1 ≥ Rs2, 0 sonst</code> <code>:= bne Rs2, Rs1, 8 ori Rd, \$zero, 1 beq \$zero, \$zero, 4 sltu Rd, Rs2, Rs1</code>	(\geq) set greater than equal unsigned
<code>sgt^(P)</code>	<code>Rd, Rs1, Rs2</code>	<code>Rd := 1, falls Rs1 > Rs2, 0 sonst</code> <code>:= slt Rd, Rs2, Rs1</code>	$(>)$ set greater than
<code>sgtu^(P)</code>	<code>Rd, Rs1, Rs2</code>	<code>Rd := 1, falls Rs1 ≥ Rs2, 0 sonst</code> <code>:= sltu Rd, Rs2, Rs1</code>	$(>)$ set greater than unsigned

Tests auf kleiner:

Befehl	Argumente	Wirkung	Erläuterung
<code>sle^(P)</code>	<code>Rd, Rs1, Rs2</code>	<code>Rd := 1, falls Rs1 ≤ Rs2, 0 sonst</code> <code>:= bne Rs2, Rs1, 8 ori Rd, \$zero, 1 beq \$zero, \$zero, 4 slt Rd, Rs1, Rs2</code>	(\leq) set less than equal
<code>sleu^(P)</code>	<code>Rd, Rs1, Rs2</code>	<code>Rd := 1, falls Rs1 ≤ Rs2, 0 sonst</code> <code>:= bne Rs2, Rs1, 8 ori Rd, \$zero, 1 beq \$zero, \$zero, 4 sltu Rd, Rs1, Rs2</code>	(\leq) set less than equal unsigned
<code>slt</code>	<code>Rd, Rs1, Rs2</code>	<code>Rd := 1, falls Rs1 < Rs2, 0 sonst</code>	$(<)$ set less than
<code>sltu</code>	<code>Rd, Rs1, Rs2</code>	<code>Rd := 1, falls Rs1 < Rs2, 0 sonst</code>	$(<)$ set less than unsigned
<code>slti</code>	<code>Rd, Rs, Imm</code>	<code>Rd := 1, falls Rs < Imm, 0 sonst</code>	$(<)$ set less than immediate
<code>sltui</code>	<code>Rd, Rs, Imm</code>	<code>Rd := 1, falls Rs < Imm, 0 sonst</code>	$(<)$ set less than unsigned immediate

Bei der Umsetzung der Pseudo-Befehle in Maschinenbefehle finden wieder die Befehle `beq` und `bne` Anwendung. Wir lernen sie erst im nächsten Abschnitt kennen, haben aber schon im Abschnitt 6.3 auf Seite 25 von ihnen gehört.

Zur Erinnerung: der Maschinenbefehl `ori Rd, $zero, 0` ist die Übersetzung des Befehls `li` (load immediate), vgl. Abschnitt 5.3 auf Seite 21.

Übungen

Übung 17

Berechne mit Hilfe der dir jetzt bekannten Entscheidungsbefehle das Entgelt für einen Brief. Gehe vereinfachend davon aus, dass das Porto wie folgt berechnet

8. LOGISCHE BEFEHLE

wird:

<i>Gewicht bis</i>	<i>20g</i>	<i>50g</i>	<i>250g</i>	<i>500g</i>
<i>Entgelt</i>	<i>110</i>	<i>220</i>	<i>300</i>	<i>440</i>

9 Kontrollstrukturen

Grundlage vieler Algorithmen sind *Entscheidungen*:

Entscheidungen

Wenn der Betrag größer als 1.000,- DM ist, dann gewähre 3% Rabatt. Sonst beträgt der Rabatt 2%.

In MODULA würden wir dafür schreiben:

```
IF Betrag > 1000
  THEN Rabatt := 3
  ELSE Rabatt := 2
END; (*IF*)
```

Manche Anweisungen sollen mehrmals hintereinander ausgeführt werden. Wir sprechen dann von *Schleifen*. Z.B.:

Schleifen

Addiere solange die Folge der natürlichen Zahlen bis die Summe 100 überschreitet.

In MODULA:

```
summe := 0;
i := 0;
WHILE summe <= 100
  i := i + 1;
  summe := summe + i
END; (*WHILE*)
```

Hochsprachen stellen für solche Entscheidungen und Schleifen eine Fülle verschiedener ausgefeilter *Kontrollstrukturen* zur Verfügung. Der Assemblerprogrammierer muss sich diese Strukturen jedoch aus einer ganzen Reihe von Befehlen zusammensetzen. Die häufigsten Kontrollstrukturen lernen wir in diesem Abschnitt in die Befehle des **MIPS** zu übersetzen.

Kontrollstrukturen

9.1 Programmverzweigungen (einfache Fallunterscheidungen) oder: Sprungbefehle

In dem Rabatt-Beispiel zu Beginn dieses Abschnitts wurde ein Teil des Programms nur bedingt ausgeführt, nämlich dann, wenn die Variable *Betrag* größer als 1 000 ist. Diese Bedingung ist eine sehr einfache. Hochsprachen erlauben die Verkettung mehrerer Bedingungen, sogar komplexe Funktionsaufrufe sind erlaubt. Der **MIPS** erlaubt nur eine ganz geringe Zahl von Bedingungen: Lediglich Größenvergleiche zwischen den Werten zweier Register oder einem Register und einem unmittelbaren Wert sind erlaubt.

Es gibt zwei unterschiedliche Arten in einem Programm Sprünge auszuführen:

- Springe um n Befehle nach vorne oder nach hinten (*branching*) branching
- Springe an die Stelle x (*jumping*). jumping

9. KONTROLLSTRUKTUREN

Der SPIM verfügt über vier unbedingte Jump-Befehle, 21 bedingte und einen unbedingten Branch-Befehl, von denen wir die meisten jetzt kennen lernen werden. Beim Programmieren muss man sich um den Unterschied der beiden Sprungarten nicht kümmern, da es der Assembler bei allen Sprungbefehlen erlaubt, das Sprungziel durch eine Marke (label) anzugeben.

Die folgenden Tabellen liefern eine Übersicht über die möglichen Sprunganweisungen.

Unbedingte Sprunganweisungen:

Befehl	Argumente	Wirkung	Erläuterung
b ^(P)	label	unbedingter Sprung nach label	branch
		:= bgez \$zero label-PC	
j	label	unbedingter Sprung nach label	jump

Sprung bei Gleichheit (oder Ungleichheit):

Befehl	Argumente	Wirkung	Erläuterung
beq	Rs1, Rs2, label	Sprung nach label, falls Rs1=Rs2	(=) branch on equal
beqz ^(P)	Rs, label	Sprung nach label, falls Rs=0	(= 0) branch on equal zero
		:= beq \$zero, Rs, label-PC	
bne	Rs1, Rs2, label	Sprung nach label, falls Rs≠0	(≠) branch on not equal zero
bnez ^(P)	Rs, label	Sprung nach label, falls Rs≠0	(≠ 0) branch on not equal zero
		:= bne \$zero, Rs, label-PC	

Sprung bei größer oder gleich:

Befehl	Argumente	Wirkung	Erläuterung
bge ^(P)	Rs1, Rs2, label	Sprung nach label, falls Rs1≥Rs2	(≥) branch on greater than equal
		:= slt \$at, Rs1, Rs2 beq \$at, \$zero, label-PC	
bgeu ^(P)	Rs1, Rs2, label	Sprung nach label, falls Rs1≥Rs2	(≥) branch on greater than equal unsigned
		:= sltu \$at, Rs1, Rs2 beq \$at, \$zero, label-PC	
bgez	Rs, label	Sprung nach label, falls Rs≥0	(≥ 0) branch on greater than equal zero

Sprung bei größer:

Befehl	Argumente	Wirkung	Erläuterung
bgt ^(P)	Rs1 Rs2 label	Sprung nach label, falls Rs1>Rs2	(>) branch on greater than
		:= slt \$at, Rs2, Rs1 bne \$at, \$zero, label-PC	
bgtu ^(P)	Rs1 Rs2 label	Sprung nach label, falls Rs1>Rs2	(>) branch on greater than unsigned

9.1 Programmverzweigungen (einfache Fallunterscheidungen) oder: Sprungbefehle

Befehl	Argumente	Wirkung	Erläuterung
		:= sltu \$at, Rs2, Rs1 bne \$at, \$zero, label-PC	
bgtz	Rs, label	Sprung nach label, falls $Rs > 0$	(> 0) branch on greater than zero

Sprung bei kleiner oder gleich:

Befehl	Argumente	Wirkung	Erläuterung
ble ^(P)	Rs1 Rs2 label	Sprung nach label, falls $Rs1 \leq Rs2$	(\leq) branch on less than equal
		:= slt \$at, Rs2, Rs1 beq \$at, \$zero, label-PC	
bleu ^(P)	Rs1 Rs2 label	Sprung nach label, falls $Rs1 \leq Rs2$	(\leq) branch on less than equal unsigned
		:= sltu \$at, Rs2, Rs1 beq \$at, \$zero, label-PC	
blez	Rs, label	Sprung nach label, falls $Rs \leq 0$	(≤ 0) branch on less than equal zero

Sprung bei kleiner:

Befehl	Argumente	Wirkung	Erläuterung
blt ^(P)	Rs1, Rs2, label	Sprung nach label, falls $Rs1 < Rs2$	(<) branch on less than
		:= slt \$at, Rs1, Rs2 bne \$at, \$zero, label-PC	
bltu ^(P)	Rs1 Rs2 label	Sprung nach label, falls $Rs1 < Rs2$	(<) branch on less than unsigned
		:= sltu \$at, Rs1, Rs2 bne \$at, \$zero, label-PC	
bltz	Rs, label	Sprung nach label, falls $Rs < 0$	(< 0) branch on less than zero

Die Branchbefehle erwarten eigentlich eine Distanz (Offset) zum aktuellen Programmzähler (PC). Man spricht auch von *PC-relativer Adressierung*. Da diese Adressierung für den Assemblerprogrammierer sehr aufwendig und extrem fehleranfällig ist, erlaubt der SPIM-Assembler Marken als Parameter. Leider entfällt damit die Möglichkeit zur PC-relativen Adressierung völlig, da der Assembler nicht mehr zwischen der Marke „4“ und dem Wert 4 unterscheiden kann.

PC-relative
Adressierung

Nicht ganz konsistent ist die Angabe der Distanz:

- In der *Maschinensprache*, die für den Assemblerprogrammierer lediglich beim manuellen disassemblieren oder assemblieren interessant ist, wird in den letzten 16 Bit des Befehls (also in der niedrigwertigen Hälfte) die Zahl der zu überspringenden *Befehle* angegeben.
- Der SPIM-Disassemblierer gibt jedoch im Text-Fenster die Zahl der zu überspringenden *Bytes* an, also jeweils das vierfache der rechten Befehls-hälfte. Aus Konsistenzgründen habe ich bei der Angabe der Übersetzung der Pseudobefehle in diesem Tutorial ebenfalls diese Notation gewählt.
- Auf der Assembler-Ebene ist lediglich die Verwendung von Marken zugelassen, die Berechnung der Distanz übernimmt der Assembler.

9. KONTROLLSTRUKTUREN

Bei der Implementation *einfacher Verzweigungen* sollte man sich zunächst die Struktur der Fallunterscheidung deutlich machen. Eine gute Möglichkeit, die auch für die Dokumentierung dieser Assembler-Passage gut geeignet ist, ist eine Formulierung in MODULA oder einer anderen Hochsprache.

einfache Verzweigung

Nehmen wir beispielsweise das Problem vom Anfang dieses Kapitels:

```
IF Betrag > 1000
  THEN Rabatt := 3
  ELSE Rabatt := 2
END; (*IF*)
```

Gehen wir einmal davon aus, dass die Variable `Betrag` im Register `$t0` und die Variable `Rabatt` im Register `$t1` abgelegt sind. Es bietet sich an, zunächst einmal die Bedingung für den 3%-Rabatt in Assembler umzusetzen:

```
    bgt    $t0, 1000, then    # IF Betrag > 1000
```

Die Marke `then` haben wir noch nicht definiert, sie soll am Anfang des „Dann“-Zweiges stehen, den wir uns nun vornehmen:

```
then: li    $t1, 3            # THEN Rabatt := 3
```

Als nächstes übersetzen wir den „Sonst“-Zweig:

```
    li    $t1, 2            # ELSE Rabatt := 2
```

Würden wir diese drei Zeilen einfach hintereinander schreiben, so würden sowohl der „Dann“- als auch der „Sonst“-Zweig *immer* ausgeführt werden, was nicht im Sinne unserer Aufgabe ist. Eine korrekte, aber ineffiziente Lösung wäre:

Beispiel 4: Einfache Verzweigung (Erster Versuch)

```
# BSP001.S
2 main:  bgtu   $t0, 1000, then # IF Betrag > 1000
        b      else
4 then:  li    $t1, 3          # THEN Rabatt := 3
        b      endif
6 else:  li    $t1, 2          # ELSE Rabatt := 2
        b      endif:       # END;(*IF*)
```

Betrachten wir das Programm genauer, so fällt uns auf, dass wir auf einen der unbedingten Sprünge verzichten können, wenn wir statt der Bedingung für den „Dann“-Zweig die Bedingung für den „Sonst“-Zweig verwenden. Unser Programm sieht nun so aus:

Beispiel 5: Einfache Verzweigung (Zweiter Versuch)

```
# BSP002.S
2 main:  ble   $t0, 1000, else # IF Betrag > 1000
        li    $t1, 3          # THEN Rabatt := 3
        b      endif
4 else:  li    $t1, 2          # ELSE Rabatt := 2
6 endif:  b      endif:       # END;(*IF*)
```

Wir sparen eine Zeile im „Sonst“-Zweig. Eine andere Vereinfachung wäre:

Beispiel 6: Einfache Verzweigung (Zweiter Versuch, Variante)

```
# BSP003.S
2 main:  bgt   $t0, 1000, then # IF Betrag > 1000
        else: li    $t1, 2          # ELSE Rabatt := 2
4        b      endif
        then: li    $t1, 3          # THEN Rabatt := 3
6 endif:  b      endif:       # END;(*IF*)
```

Welche der Varianten man in der Regel wählt, ist Geschmackssache. Sollte einmal einer der Zweige leer sein (in der Regel wohl der „Sonst“-Zweig), so wäre allerdings die erste Variante zu bevorzugen, da hier die Zeilen 3 und 4 völlig weggelassen werden könnten.

Häufig wird die Sprungbedingung nicht in einem einzigen Vergleich bestehen. In diesem Fall muss sie zunächst separat ausgewertet werden. Am Ende des Auswertungsabschnittes steht dann einer der zur Verfügung stehenden einfachen Verzweigungsbefehle.

Übungen

Übung 18

Schreibe ein Programm, das die Länge der Seiten eines Dreieckes in beliebiger Reihenfolge einliest und dann ausgibt ob es sich um ein allgemeines (alle Seiten verschieden), ein gleichschenkliges (zwei gleich lange Seiten) oder ein gleichseitiges Dreieck (drei gleiche Seiten) handelt.

Übung 19

Ein Betrieb verkauft sein Produkt zu einem Preis von 1 000 DM pro Stück. Bei der Bestellung von 100 Stück gewährt er einen Rabatt von 5%, bei der Bestellung von 500 Stück einen Rabatt von 10%. Es handelt sich um einen Gesamtrabatt, d.h. 99 Stück kosten 99 000 DM, 100 Stück 95 000 DM, und 500 Stück 450 000 DM.

Schreibe ein Programm, das nach Eingabe der Bestellmenge den Gesamtpreis ausgibt.

Übung 20

Der Betrieb aus 19 führt alternativ zu dem bestehenden Rabattsystem ein weiteres ein. Dabei handelt es sich um einen Stufenrabatt, d.h. der Rabatt bezieht sich jeweils nur auf die Menge, die über die Rabattgrenze hinaus gekauft wurde: Der Rabatt für das 20. bis 49. Stück beträgt 2%, der Rabatt vom 50. bis 100. Stück 5% und ab dem 101. Stück 10%.

19 Stück kosten also 19 000 DM, 20 Stück 19 980 DM.

Schreibe ein Programm, das die Preise nach beiden Rabattsystemen berechnet und ausgibt.

9.2 Schleifen

Neben Fallunterscheidungen stellen Schleifen eine wesentliche Grundlage von Algorithmen dar. Wir kennen drei Grundtypen von Schleifen:

- Die *Abweisende Schleife* (While-Schleife), die vor jedem Ausführen der Schleife prüft, ob die Schleifenbedingung noch wahr ist. Abweisende Schleifen werden gar nicht ausgeführt, wenn die Schleifenbedingung zu Beginn unwahr ist. Abweisende Schleife
- Bei der *Nachprüfenden Schleife* (Repeat-Until-Schleife) wird erst nach dem ersten und jedem weiteren Ausführen der Schleife geprüft, ob die Schleifen(abbruch)bedingung noch wahr ist. Sie wird also mindestens einmal ausgeführt. Nachprüfende Schleife

9. KONTROLLSTRUKTUREN

- Bei der *Zählschleife* (For-Schleife) wird vor Eintritt in die Zählschleife festgelegt, wie oft sie ausgeführt werden soll. Die Zahl der absolvierten Schleifendurchläufe wird in einer Zählvariable gezählt. Viele Hochsprachen erlauben auch das Zählen mit einer bestimmten, ggf. auch negativen Schrittweite.

Zählschleife

9.2.1 Abweisende Schleifen

Betrachten wir das zweite Beispiel vom Anfang dieses Kapitels:

```
summe := 0;
i := 0;
WHILE summe <= 100
    i := i + 1;
    summe := summe + i
END;
```

Die ersten beiden Zeilen dieses (ineffizienten) Programmfragments sind schnell übersetzt:

```
li    $t0, 0        # summe := 0;
li    $t1, 0        # i := 0;
```

Für die Schleife benötigen wir sicherlich eine Marke, schließlich wollen wir am Ende der Schleife stets wieder an ihren Anfang springen. Falls die Eintrittsbedingung nicht mehr erfüllt ist, müssen wir dann von dort wieder an das Ende der Schleife springen. Wir benötigen also auch dort eine Marke. Unser Assemblerprogramm könnte also so aussehen:

Beispiel 7: Eine Abweisende Schleife

```
# BSP004.S
2      li    $t0, 0        # summe := 0;
      li    $t1, 0        # i := 0;
4  while: bgt  $t0, 100, elihw # While summe <= 100
      addi  $t1, $t1, 1    # i := i + 1;
6      add  $t0, $t1, $t0  # summe := summe + i
      b    while          # END;
8  elihw:
```

Zu beachten ist die umgekehrte Sprungbedingung wie wir sie auch schon bei den Fallunterscheidungen kennen gelernt haben. Ungünstig wirkt sich der unbedingte Rücksprung am Schleifenende aus. Dieser Befehl kann eingespart werden, wenn man statt ihm einen bedingten Sprungbefehl mit der negierten Abweisungsbedingung einsetzt, der an den ersten Befehl in der Schleife springt (`addi`). Diese effizientere Version ist jedoch etwas unübersichtlicher und damit fehleranfällig.

9.2.2 Nachprüfende Schleifen

Die nachprüfenden Schleifen sind etwas einfacher zu programmieren als die abweisenden Schleifen, weil hier der unbedingte Rücksprung zur Prüfung entfällt. Da dieser Rücksprung bei jedem Schleifendurchlauf ausgeführt wird, empfiehlt es sich, wenn möglich, die nachprüfenden Schleifen den abweisenden vorzuziehen.

Beispiel 8: Eine nachprüfende Schleife


```

# BSP005.S
2      li      $t0, 0          # summe := 0;
      li      $t1, 0          # i := 0;
4      # REPEAT
repeat: addi   $t1, $t1, 1     # i := i + 1;
6      add    $t0, $t1, $t0    # summe := summe + i
      ble    $t0, 100, repeat # UNTIL summe > 100;

```

9.2.3 Zählschleifen

Es gibt Zählschleifen, bei denen bereits zum Entwicklungszeitpunkt bekannt ist, wie oft sie ausgeführt werden. Ist die Zahl der Ausführungen sehr gering und die Schleife sehr kurz, so kann es unter Umständen günstiger sein, den zu wiederholenden Code mehrfach hintereinander zu schreiben. Im Regelfall sollte aber auch bei bekannter Durchlaufanzahl auf eine Zählschleife zurückgegriffen werden.

Die Zählschleifen lassen sich auch als abweisende Schleifen formulieren, nur liegt dann die Bearbeitung der Zählvariable in der Hand des Programmierers (in Hochsprachen). Im Assemblerprogramm unterscheiden sich die beiden Schleifen kaum. Zu Dokumentierungszwecken sollten die beiden Schleifenarten jedoch auseinander gehalten werden.

Wollen wir beispielsweise die geraden Zahlen von 0 bis 100 aufsummieren, so würden wir in MODULA schreiben:

```

summe := 0;
FOR i := 0 TO 100 STEP 2 DO
    summe := summe + i
END;

```

Die Übersetzung der ersten Zeile bereitet uns keine Schwierigkeiten. Wir entscheiden uns, die Laufvariable *i* im Register *\$t1* zu halten. Den Rest der Schleife können wir übersetzen wie eine abweisende Schleife. Natürlich müssen wir auch die Verwaltung der Laufvariablen bedenken, die uns in MODULA der MODULA-Compiler abgenommen hätte. Die Laufvariable wird grundsätzlich als letztes in der Schleife um die Schrittweite erhöht. In Abschnitt 7 auf der vorangegangenen Seite hatten wir erfahren, dass die abweisenden Schleifen effizienter sind, wenn sie keine unbedingte Sprunganweisung enthalten. Im folgenden Beispiel sehen wir jetzt eine solche effiziente Version.

Beispiel 9: Eine effiziente Zählschleife

```

# BSP006.S
2      li      $t0, 0          # summe := 0;
      li      $t1, 0          # i := 0;
4      bgt    $t1, 100, rof    # FOR i := 0 TO 100 STEP 2 DO
for:   add    $t0, $t0, $t1    # summe := summe + i
6      addi   $t1, $t1, 2
      blt    $t1, 100, for    # END;
8 rof:

```

9.2.4 Schleifen verlassen

Wenn wir in Assembler programmieren, haben wir die Möglichkeit, an jeder Stelle einer Schleife die Schleifenbearbeitung abzubrechen und an das Ende der

9. KONTROLLSTRUKTUREN

Schleife zu springen. Dieses Ende ist in der Regel ja durch eine Marke gekennzeichnet. Diese Möglichkeit besteht in vielen Hochsprachen nicht (in MODULA gibt es eine wenig bekannte LOOP-Schleife, die mit dem Befehl EXIT an jeder beliebigen Stelle verlassen werden kann).

Sicherlich tragen solche Schleifenabbrüche zu einer günstigeren Laufzeit bei, aber sie machen das Programm unübersichtlich. Daher sollte darauf wenn irgendmöglich verzichtet werden. Wenn Schleifen außerhalb der Abbruchbedingungen verlassen werden müssen, dann nur

- durch Sprung an den auf die Schleife unmittelbar folgenden Befehl und
- unter Dokumentierung, am besten mit Hilfe eines Befehls ähnlich dem EXIT-Befehl von MODULA.

Sind mehrere Schleifen verschachtelt, so kann es auch reizvoll sein, nicht an das Ende der gerade aktiven Schleife zu springen, sondern darüber hinaus an das Ende einer die aktive Schleife einschachtelnden Schleife zu springen. Es sollte jedoch genau abgewogen werden, ob sich dies lohnt, denn die Wartung und Fehlersuche werden dadurch sehr erschwert.

Übungen

Übung 21

Gebe ein aus Sternchen bestehendes Dreieck der Höhe n aus. n soll vom Benutzer eingegeben werden und zwischen 1 und 20 liegen. Für $n = 3$ soll folgendes Dreieck ausgegeben werden:

```
*  
***  
*****
```

Übung 22

Implementiere die Fibonacci-Zahlen iterativ. Zur Erinnerung:

$$\text{fib}(n) := \text{fib}(n - 1) + \text{fib}(n - 2), \text{fib}(2) = \text{fib}(1) = 1$$

10 Datenhaltung II: Komplexe Datentypen

Wir haben nun gelernt, mit dem **MIPS** ganze Zahlen verschiedener Länge zu bearbeiten. Höhere Programmiersprachen bieten aber auch die Möglichkeit einfache Datentypen zusammenzufassen. Datentypen verschiedenen oder gleichen Typs können miteinander verbunden werden und mit einer einzigen (Hochsprachen-) Operation mit Werten belegt werden. Diese *Verbunde* (*Record*, *structure*) bieten die Möglichkeit, alle Daten, die zu einem Objekt gehören, in einer einzigen Variable vorliegen zu haben.

Verbunde

Häufig brauchen wir aber auch eine ganze Reihe Variabler gleichen Typs, etwa eine Liste mit Personen. Hierfür stellen fast alle Hochsprachen *Felder* (*Array*, *List*) zur Verfügung. Innerhalb des Feldes haben die Feldelemente eine feste Reihenfolge und sind über ihren Index zugreifbar.

Felder

Beide komplexen Datentypen können in Hochsprachen mehr oder weniger beliebig miteinander kombiniert werden. Änderungen an ihrer Struktur sind oft mit wenigen Handgriffen möglich. Auf Assemblerebene ist dies ganz anders, wie wir jetzt sehen werden.

10.1 Felder

10.1.1 Erstellung und Initialisierung von Feldern

Abgesehen von sehr kleinen Feldern müssen Felder grundsätzlich im Hauptspeicher abgelegt sein. Nur zum Lesen werden die Feldelemente in die Prozessorregister geladen. Aus Abschnitt 4.1 auf Seite 15 kennen wir bereits einige Direktiven, die im Datensegment Platz für Zahlen belegen. Zur Erinnerung:

```
.byte legt die nachfolgenden Zahlen als 8-Bit-Zahlen ab,  
.half legt die nachfolgenden Zahlen als 16-Bit-Zahlen ab,  
.word legt die nachfolgenden Zahlen als 32-Bit-Zahlen ab und  
.double legt die nachfolgenden Zahlen als 64-Bit-Zahlen.
```

Ein Feld von 13 Wörtern können wir also wie folgt erstellen:

```
        .data  
feld:   .word   0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0
```

Damit sind alle Feldelemente zugleich mit dem Wert 0 initialisiert. Leider gibt es keine Direktive, die dieses wiederholte Eintragen der Werte erleichtert.

Es gibt aber auch die Möglichkeit mit dem Befehl `.space` ganze Bereiche zu reservieren, ohne sie gleichzeitig zu initialisieren:

```
        .data  
feld:   .space  52
```

Der Parameter gibt die Zahl von Bytes an, die reserviert werden. Beide Direktiven legen also ein Feld mit 13 Wörtern an. Wesentlich ist die Feststellung, dass `.space` die Feldelemente *nicht* initialisiert.

Die Initialisierung durch `.word` u.ä. Direktiven kostet zwar keine Laufzeit, da sie beim Assemblieren durchgeführt wird, ist aber auch nicht zwingend notwendig, da der erste Zugriff auf eine Variable sowieso nicht lesend, sondern besser

10. DATENHALTUNG II: KOMPLEXE DATENTYPEN

schreibend erfolgen sollte. Schließlich kann man sonst oft nicht mit letzter Sicherheit bestimmen, mit welchem Wert die Variable belegt ist. Es gibt sicher gute Gründe von diesem Prinzip abzuweichen. Aber bei Feldern sind solche Gründe sehr selten.

Wenn man wie wir den Simulator SPIM benutzt, gibt es ein weiteres Argument gegen die einmalige Initialisierung: Vor dem nächsten Programmdurchlauf werden die Daten im Datensegment nicht gelöscht, wenn man nicht das gesamte Programm neu lädt.

10.1.2 Arbeiten mit Feldern

Von dem Feld ist uns nur die Anfangsadresse bekannt. In der Regel wird erst zur Laufzeit entschieden, auf welches Feldelement wir zugreifen wollen. Die Hauptspeicheradresse, auf die wir zugreifen wollen, ist also variabel.

Die Lade- und Speicherbefehle erlauben alle die indexierte Adressierung (Vgl. Abschnitt 5.1.1 auf Seite 18). Wollen wir beispielsweise alle Feldelemente unseres 13-elementigen Feldes mit ihrem Index belegen, so könnten wir dies so tun:

Beispiel 10: Indexieren eines Feldes

```
# BSP007.S
2      .data
feld:  .space 52          # VAR:
4                                #   feld: ARRAY [0..12] OF INTEGER;

      .text
6 main:  li    $t0, 0
for:    bgt   $t0, 48, rof  # FOR i := 0 TO 12 DO
8        sw   $t0, feld($t0) #   feld[i] := i;
        addi  $t0, $t0, 4
10       b    for          # END;(*FOR*)
rof:
```

Das Beispielprogramm zeigt das Prinzip und lässt sich leicht effizienter gestalten.

Feldzuweisung

Soll ein Feld einem anderen *zugewiesen* werden, so müssen mühselig die Daten des einen Feldes in den Speicher des anderen kopiert werden.

Bereichsgrenzen

Großes Augenmerk müssen wir den *Bereichsgrenzen* widmen. Keinesfalls darf außerhalb des Feldes gelesen oder geschrieben werden. Derartige Fehler sind sehr schwierig zu finden und können unberechenbare Folgen haben. Daher ist hier Selbstdisziplin unabdingbar. Viele Compiler fügen vor jedem Zugriff auf ein Feldelement, egal ob lesend oder schreibend, Code ein, der überprüft, ob der Index im zulässigen Bereich liegt.

Übungen

Übung 23

Lese eine Folge von Zahlen über die Tastatur ein. Die Eingabe wird beendet, wenn die 0 eingegeben wird oder wenn die 20. Zahl gelesen wurde. Sortiere diese Folge aufsteigend in einem beliebigen Verfahren und gebe sie aus.

Übung 24

Lege im Datensegment zwei null-terminierte Zeichenketten ab. Kopiere den Inhalt der zweiten Zeichenkette in die erste Zeichenkette. Was für ein Problem kann sich dabei ergeben?

10.2 Verbunde

Verbunde werden wir nur kurz behandeln, da sie im Gegensatz zu den Feldern zwar umständlich aber unproblematisch zu behandeln sind.

Es soll hier ausreichen sich folgende Gedanken zu machen:

- Bei der Haltung im Datensegment reicht in der Regel die Belegung mit der `.space`-Direktive aus.
- Die Zuordnung der einzelnen Komponenten eines Verbundes zu den Bytes muss dokumentiert werden.
- Ein Record sollte ausgerichtet, d.h. Wörter sollten keinesfalls über die Wortgrenzen hinaus abgelegt werden. Komponenten von nicht ganzzahliger Wortanzahl sollten am Ende des Verbundes abgelegt werden.
- Änderungen an der Verbundstruktur sind in Assemblersprache wesentlich aufwendiger als in Hochsprachen.
- Das Gleichsetzen von zwei Verbunden kann wortweise, also nicht Komponente für Komponente erfolgen.

11. MEHRFACHE FALLUNTERSCHIEDUNG (CASE-ANWEISUNG)

11 Mehrfache Fallunterscheidung (CASE-Anweisung)

mehrfache
Fallunterscheidung

In Abschnitt 9.1 auf Seite 35 haben wir die einfache Fallunterscheidung kennen gelernt. Häufig benötigen wir jedoch eine *mehrfache Fallunterscheidung*, in der in Abhängigkeit vom Wert eines Ausdrucks an mehr als zwei verschiedene Stellen im Programm gesprungen wird, etwa nach einer Menüauswahl.

Eine Lösung dieses Problems wäre ein Folge einfacher Fallunterscheidungen:

Beispiel 11: Mehrfache Fallunterscheidung mit einfachen Fallunterscheidungen

```
# CASE1.S - Programmfragment, nicht lauffaehig!
2 # Menueauswahl in $v0, 0 <= $v0 <= 4
      beq    $v0, 0, case0
4      beq    $v0, 1, case1
      beq    $v0, 2, case2
6      beq    $v0, 3, case3
      b     case4
```

Der Nachteil dieses Verfahrens ist offensichtlich: Bei vielen Alternativen müssen alle vor der gewünschten Alternative auftretenden Verzweigungsbefehle abgearbeitet werden. Wir können zwar etwas nachhelfen und die Alternativen in der Reihenfolge ihrer voraussichtlichen Häufigkeit sortieren oder wir können uns durch Intervallschachtelung sehr viel schneller der gewünschten Alternative nähern.

Sprungtabelle (jump
address table)

Wir sollten aber auch einen Blick auf die *Sprungtabelle (jump address table)* werfen, einem Verfahren, das eine sehr effiziente Sprungverzweigung mit nur einer einzigen Sprunganweisung ermöglicht.

Dafür benötigen wir einen weiteren Befehl:

Befehl	Argumente	Wirkung	Erläuterung
<code>jr</code>	<code>Rs</code>	unbedingter Sprung an die Adresse in <code>Rs</code>	jump register

`jr` ermöglicht uns den Sprung an eine erst zur Laufzeit ermittelte Stelle im Programm. Die Idee der Sprungtabelle ist nun, im Datensegment ein Feld anzulegen, das die Adressen der Sprungziele enthält. Da diese Adressen bereits zur Assemblierzeit feststehen, muss zur Laufzeit lediglich noch die richtige Adresse geladen werden.

Beispiel 12: Mehrfache Fallunterscheidung mit Sprungtabelle

```
# CASE2.S - Mehrfache Fallunterscheidung mit Sprungtabelle
2 .data
  jat:    .word case0, case1, case2, case3, case4
4        # Sprungtabelle wird zur Assemblierzeit belegt.

6 .text
main:    li     $v0, 5          # RdInt
8        syscall
        blt    $v0, 0, error  # Eingabefehler abfangen
10       bgt    $v0, 4, error
        mul    $v0, $v0, 4     # 4-Byte-Adressen
12       lw     $t0, jat($v0)  # $t0 enthält Sprungziel
        jr     $t0
14 case0: li     $a0, 0        # tu dies und das
        j     exit
16 case1: li     $a0, 1        # tu dies und das
```

```

18 case2: j      exit
        li      $a0, 2      # tu dies und das
        j      exit
20 case3: li      $a0, 3      # tu dies und das
        j      exit
22 case4: li      $a0, 4      # tu dies und das
        j      exit
24 error: li      $a0, 999    # tu dies und das
exit:   li      $v0, 1      # WrInt
26      syscall
        li      $v0, 10     # Exit
28      syscall

```

Zu beachten ist, dass die Auswahl in die korrekte Speicheradresse umgewandelt wird. Da die Adressen jeweils ein Wort breit sind, muss die Auswahl mit vier multipliziert werden. Größte Sogfalt müssen wir darauf verwenden, dass die berechnete Adresse tatsächlich innerhalb der Sprungtabelle liegt. Ein Laden eines benachbarten Wertes führt zu einer Fortsetzung des Programmes an einer falschen Stelle. Fehler dieser Art können extrem schwierig zu lokalisieren sein.

Die Sprungtabelle ist auch dann verwendbar, wenn für eine ganze Wertegruppe ein Sprung an dieselbe Stelle ausgeführt werden soll. In diesem Fall braucht lediglich die Zeile 3 in dem vorangegangenen Beispiel entsprechend verändert zu werden.

Übungen

Übung 25

Der folgenden Tabelle können die (vereinfachten) Entgelte für weltweite Luftpostbriefe entnommen werden:

Gewichtsstufe:	1	2	3	4	5	6	7	8	9
Entgelt in DM:	3	6	10	16	24	32	40	56	72

Nach Eingabe des Gewichts soll der Benutzer das Entgelt ausgegeben bekommen. Verwende bei der Lösung eine Sprungtabelle, obwohl sie hier nicht unbedingt notwendig wäre!

12. KELLERSPEICHER (STACK)

12 Kellerspeicher (Stack)

12.1 Das Grundprinzip

temporäre Variablen Häufig benötigen wir *temporäre Variablen* für Zwischenergebnisse bei der Auswertung von Ausdrücken o.ä. Meist stehen hierfür die Register $\$t0$ bis $\$t9$ zur Verfügung. Diese reichen aber gelegentlich nicht aus und für die Auswertung von Ausdrücken variabler Länge sind sie ungeeignet.

Stack (Kellerspeicher) Abhilfe schafft der *Stack (Kellerspeicher)*, ein Bereich im Hauptspeicher, der je nach Bedarf wächst oder schrumpft. Bei dem Stack handelt es sich um einen sogenannten *LIFO (Last-in-First-out)-Speicher*: Bevor auf das vorletzte Element zugegriffen werden kann, muss zunächst das letzte Element vom Stack entfernt werden.

LIFO-Speicher

Bei dem **MIPS** wie auch bei den meisten anderen Rechnersystemen wird der Stack am Ende des adressierbaren Speichers angeordnet und wächst von dort mit geringer werdenden Hauptspeicheradressen in Richtung auf die Adresse 0. Eine andere Möglichkeit wäre, den Stack an das Ende des Programms zu setzen und ihn von dort mit größer werdenden Adressen auf das Ende des Hauptspeichers hinwachsen zu lassen.

Die gewählte Alternative hat den Vorteil, dass die Startadresse des Stacks nicht abhängig von der Länge des geladenen Programms ist, was das dynamische Linken fast unmöglich machen würde. Aufgabe des Betriebssystems ist es zu überprüfen, ob der Stack in das Programm hineinwächst. Tritt dieser Fall ein, kann das Programm nur noch abgebrochen werden.

Heap

Noch komplizierter wird das Ganze, wenn die Programme auch noch zur Laufzeit weiteren (Daten-)Speicher anfordern dürfen. In der Regel wird diese Anforderung erfüllt, indem ein „umgedrehter“ Stack, ein *Heap* (Haufen) dem Stack entgegen wächst. In diesem Fall kann es jedoch passieren, dass dieser dynamische Datenbereich in den Stack hineinwächst. Dies ist leicht festzustellen, da beim Einkellern und bei der Speicheranforderung lediglich der Stackpointer, ein Register, das auf den Stack zeigt, und der Heappointer auf Gleichheit verglichen werden müssen. Dieses Kriterium ist dann vom Betriebssystem zu überprüfen.

Das Speicherlayout des **MIPS** ist Abbildung 6 auf der folgenden Seite zu entnehmen. In dem Bereich oberhalb von $0x7FFF\ FFFF$ befinden sich Daten und Programme des Betriebssystems (Kernel).

12.2 Die Stackprogrammierung: Einkellern

Stackpointer


Der *Stackpointer* ($\$sp$) zeigt stets auf das *erste freie* Wort auf dem Stack. Dies ist eine Konvention, die *unbedingt* eingehalten werden muss. Nach dem Laden eines Programmes zeigt der Stackpointer also auf $0x7FFF\ FFFC$, da dies das oberste Wort des Stacksegmentes ist.⁷ Die Programmierung mit dem Stack ist fehleranfällig und sollte besonders sorgfältig vorgenommen werden, da Fehler sehr schwer zu lokalisieren sind.

Einige CISC-Prozessoren, wie z.B. die Motorola 68000er-Serie, stellen für Stackoperationen eigene Befehle wie z.B. *PUSH* für das „Einkellern“ oder *POP* für das Entfernen von Stackelementen zur Verfügung. Der **MIPS** kennt derartige Befehle nicht. Wir müssen also mit den uns bereits zur Verfügung stehenden Befehlen auskommen.

Wollen wir ein Element auf den Stack legen, so könnten wir wie folgt vorgehen:

⁷Tatsächlich wird der Stackpointer mit $0x7FFF\ FEFC$ initialisiert. Der Grund hierfür ist mir nicht bekannt.

Abbildung 6: Das Speicherlayout des MIPS



```

sw      $t0, ($sp)      # $t0 liegt auf dem ersten
                        # freien Platz
addi    $sp, -4         # Stackpointer dekrementieren

```

In der ersten Zeile wird der Inhalt des Registers `$t0` an den Speicherplatz geschrieben, der im Stackpointer angegeben ist. In der zweiten Zeile wird der Stackpointer verringert, da der Stack in Richtung auf die Speicherstelle 0 wächst. Das Dekrementieren des Stackpointers ist sehr wichtig, da andernfalls das oberste Element überschrieben werden kann. Da der Stackpointer auf Speicherplätze zeigt, muss er jeweils um ganze Worte erhöht oder verringert werden, weshalb der Stackpointer in der zweiten Zeile um 4 verringert wird.

Die hier vorgestellte Version ist jedoch gefährlich, da theoretisch nach *jedem* Befehl eine Unterbrechung (z.B. Tastatur gedrückt o.ä.) auftreten kann. Dann wird automatisch ein Betriebssystemteil ausgeführt, das eventuell den Stack manipuliert. Tritt eine solche Unterbrechung nun unglücklicherweise nach dem `sw`-Befehl auf, so kann unser `$t0` durch das Betriebssystem überschrieben werden. Ein solcher Fehler ist praktisch nicht reproduzierbar und muss daher unbedingt verhindert werden.

Durch zwei geringfügige Änderungen lässt sich aus dem vorangegangenen unsicheren Code ein sicherer Code machen:

```

addi    $sp, -4         # Stackpointer dekrementieren
sw      $t0, 4($sp)     # $t0 liegt auf dem erstem
                        # freien Platz

```

Nun darf unser Programm auch zwischen den beiden betrachteten Zeilen unterbrochen werden, ohne dass dies zu Problemen führen kann. Ein Pseudobefehl `push` könnte auf diese Weise definiert werden.


Häufig müssen wir gleich mehrere Werte im Stack ablegen. Man könnte dann die obige Sequenz entsprechend häufig hintereinander schreiben. Es bietet sich aber an, statt dessen wie folgt zu programmieren:

12. KELLERSPEICHER (STACK)

```
addi    $sp, -12      # $sp dekrementieren
sw      $t0, 12($sp)  # $t0 liegt auf erstem freien Platz
sw      $t4, 8($sp)   # $t4 liegt auf zweitem freien Platz
sw      $t2, 4($sp)   # $t2 liegt auf drittem freien Platz
```

Man beachte, dass der erste freie Platz nun durch den größten Index (12) adressiert wird. Zur Verdeutlichung ist die Entwicklung des Stacks in der Abbildung 7 auf dieser Seite dargestellt.

Abbildung 7: Die Entwicklung des Stacks beim „Einkellern“



Gegenüber dem naiven Ansatz, jedes Register durch die beiden Zeilen unseres nicht vorhandenen `push`-Pseudobefehls zu ersetzen, sparen wir zwei `addi`-Befehle. Aber sparen wir auch Rechenzeit? Schließlich muss der Stackpointer in den Zeilen zwei bis vier ebenfalls inkrementiert werden, um die richtige Hauptspeicheradresse zu erhalten. Diese Inkrementierung wird aber auf Schaltungsebene realisiert und benötigt somit keine zusätzliche Rechenzeit.

12.3 Die Stackprogrammierung: Lesen vom Stack


Analog zum Schreiben auf den Stack funktioniert das Abräumen des Stacks:

```
lw      $t0, 12($sp)  # $t2 erhält den dritten
                        # Wert vom Stack
lw      $t4, 8($sp)   # $t4 erhält den zweiten
                        # Wert vom Stack
lw      $t2, 4($sp)   # $t0 erhält den ersten
                        # Wert vom Stack
addi    $sp, 12      # $sp inkrementieren
```

Die Entwicklung des Stacks ist der Abbildung 8 auf der folgenden Seite zu entnehmen.

Zu beachten ist, dass die Werte bis zum nächsten Schreibvorgang erhalten bleiben. Sie sind jedoch nicht mehr sicher, da nach dem Inkrementieren des Stackpointers auftretende Unterbrechungen zu ihrer Veränderung führen können!

Abbildung 8: Die Entwicklung des Stacks beim „Abräumen“



Natürlich kann auf die Stackelemente auch ohne sofortiges Inkrementieren des Stackpointers zugegriffen werden, etwa für die weitere Speicherung der temporären Variablen. Diese Zugriffe können sowohl lesend als auch schreibend vorgenommen werden, sie sollten aber gut dokumentiert werden.

12.4 Anwendungen und Beispiele

Eine der wesentlichen Anwendungen des Stacks ist die Realisierung beliebig häufiger Unterprogramm- oder Prozedur-Aufrufe. Diesem Thema werden wir im nächsten Abschnitt einige Aufmerksamkeit schenken.

Doch kann der Stack auch sonst hilfreich sein. Werden zum Beispiel in einem Programmabschnitt große Datenmengen benötigt, die nicht in Registern gehalten werden können und nach Verlassen dieses Programmabschnittes nicht mehr benötigt werden, so könnte man für diesen Zweck Platz im Datensegment reservieren. Dieser Speicherplatz geht aber für andere Daten verloren. Es ist zwar denkbar diesen Speicherplatz auch von anderen Stellen im Programm aus zu benutzen, aber dies stellt eine gefährliche Fehlerquelle dar. Der Stack schafft Abhilfe und stellt ausreichend Speicherplatz zur Verfügung.

Eine weitere Anwendung ist die Auswertung von langen Ausdrücken. Statt jedes Zwischenergebnis in einem Register zu halten, können diese Zwischenergebnisse auch auf dem Stack abgelegt werden. Dies gilt besonders, wenn der Ausdruck erst zur Laufzeit festgelegt wird, etwa durch Eingabe des Benutzers.

Übungen

Übung 26

Bei der Postfix-Schreibweise werden Operatoren nach den Operanden geschrieben, z.B.

$$234 + - \hat{=} (3 + 4) - 2$$

Schreibe ein Programm, das einen Postfix-Ausdruck als Zeichenkette übergeben bekommt und ihn auswertet. Die Zeichenkette enthält nur die Ziffern 0 bis 9 und die Zeichen für die vier Grundrechenarten +, -, * und /. Als Eingabewerte sind nur Ziffern zugelassen, also die Zahlen von 0 bis 9.

12.5 Vor- und Nachteile der CISC-Befehle PUSH und POP

In Abschnitt 12.2 auf Seite 48 haben wir die CISC-Befehle PUSH und POP angesprochen. Ihr Vorteil ist, dass der Programmierer den Stackpointer nicht mehr direkt manipulieren

12. KELLERSPEICHER (STACK)

muss und somit die Fehlergefahr abnimmt. Die Reihenfolge des Einkellerns und Abräumens des Stacks ist deutlicher und kann vom Leser schneller erfasst werden.

Nachteilig an diesen Befehlen ist der (für den Fall $n \rightarrow \infty$) doppelte Rechenaufwand durch das n -malige Ausführen der Additionsbefehle. Soll auf ein anderes als das letzte Element des Stacks zugegriffen werden, so muss auch bei der Verwendung der Befehle PUSH und POP die Adressierung über den Abstand zum Stackpointer vorgenommen werden.

13 Prozeduren

Einer der bereits frühzeitig formulierten Grundsätze der Programmierung lautet 'Share identical code!', frei übersetzt: „Schreibe nichts doppelt!“. Code, der an mehreren Stellen benötigt wird, soll nur an einer Stelle im Programm stehen. Daraus ergibt sich eine Verringerung

1. des Wartungsaufwands,
2. des Programmier- und Prüfungsaufwands und
3. des Platzbedarfs.

Bereits frühzeitig wurde daher das Konzept des *Unterprogrammes* entwickelt. Der Name Unterprogramm rührt von dem Umstand her, dass diese Unterprogramme in vielen Programmiersprachen (z.B. in der Sprache C) hinter, also *unter* dem Hauptprogramm standen. Bei MODULA-2 ist es bekanntlich umgekehrt. Im einfachsten Fall erlaubt ein Rechner bzw. eine Programmiersprache nur den Sprung an eine bestimmte Stelle im Programm und ermöglicht den *Rücksprung* an die auf den aufrufenden Befehl *folgende* Anweisung. Dies entspricht den BASIC-Befehlen GOSUB und RETURN.

Moderne Hochsprachen erlauben die Übergabe von Daten an Unterprogramme. Solche Unterprogramme nennt man im Allgemeinen *Prozeduren* oder –wenn sie einen Wert zurückliefern– *Funktionen*. Auf Assemblerebene sind beide Verfahren gleich kryptisch und fehleranfällig.

Bei Unterprogrammen unterscheidet man zwischen dem *caller*, dem Aufrufer und dem *callee*, dem Aufgerufenen. Ein callee kann selbst zum caller werden, indem er ein weiteres Unterprogramm aufruft.

Unterprogramm

Rücksprung

Prozeduren/Funktionen

caller/callee

13.1 Einfache Unterprogrammaufrufe

Häufig lohnt sich die Programmierung nach den strengen Formalien, die wir im nächsten Abschnitt kennen lernen werden, nicht. Dies ist z.B. der Fall bei sehr kleinen Assemblerprogrammen und nur wenigen Unterprogrammen, die mit sehr wenigen Parametern auskommen. Wir behandeln diese Technik an dieser Stelle nur, um die Möglichkeit aufzuzeigen und um einige Befehle einzuführen. Von der Benutzung ist eher abzuraten.

Folgende Befehle werden uns im nützlich sein:

Befehl	Argumente	Wirkung	Erläuterung
jal	label	unbedingter Sprung nach label, Adresse des nächsten Befehls in \$ra	jump and link
jalr	Rs, Rd	unbedingter Sprung an die Stelle Rs, Adresse des nächsten Befehls in Rd, Default: \$ra	jump and link register
bgezal	Rs, label	Sprung nach label wenn $R_s \geq 0$, Adresse des nächsten Befehls in \$ra	Branch on greater than equal zero and link
bltzal	Rs, label	Sprung nach label wenn $R_s < 0$, Adresse des nächsten Befehls in \$ra	Branch on less than and link


Diese Befehle erlauben uns ein Unterprogramm aufzurufen. Die Rücksprungadresse (Adresse des ersten nach Abarbeitung des Unterprogramms auszuführenden Befehls) wird bei Benutzung dieser Befehle im Register \$ra abgelegt. Eine eventuell dort bereits

13. PROZEDUREN

befindliche Adresse wird logischerweise dabei überschrieben. Dies muss natürlich bei verschachtelten Unterprogrammaufrufen verhindert werden.

Mit dem uns bereits bekannten Befehl `jr` kann nach Abarbeitung des Unterprogramms die Programmausführung fortgesetzt werden. Das Grundprinzip ist in Abbildung 9 auf dieser Seite dargestellt: das aufrufende Hauptprogramm (caller) ruft mittels `jal` das Unterprogramm (callee) auf (1). Nach Abarbeitung des Unterprogramms gibt dieses die Kontrolle wieder an das Hauptprogramm ab (2). Später wiederholt sich dieser Vorgang (3, 4).

Abbildung 9: Das Grundprinzip des Unterprogrammaufrufs



Werfen wir einen Blick auf den zugehörigen SPIM-Code:

Beispiel 13: Ein einfacher Unterprogrammaufruf

```
# UNTER1.S - Einfacher Unterprogrammaufruf - nicht lauffaehig!
2 #
   .text
4 main:  # ... tu dies ...
        jal sub          # Aufruf des Unterprogramms
6        # ... tu das ...
        jal sub          # Aufruf des Unterprogramms
8        #
        li $v0, 10       # Exit
10       syscall
        #
12 sub:  # ... tu dies und das ...
        jr $ra           # Zurück zum Aufrufer
```

Dieses Beispiel zeigt jedoch nur das Prinzip. Das Unterprogramm soll eine Befehlsfolge ausführen. Da der **MIPS** eine Load-and-Store-Architektur hat, verändern fast alle Befehlsfolgen mindestens ein Register, zumindest greifen sie lesend auf den Inhalt eines Registers zu. Arbeiten wir mit solchen einfachen Unterprogrammaufrufen, so müssen alle gelesenen und veränderten Register im Assemblertext dokumentiert werden! Das vorangegangene Beispiel könnte also wie folgt aussehen:

Beispiel 14: Ein einfacher kommentierter Unterprogrammaufruf

```
# UNTER2.S - Einfacher Unterprogrammaufruf, kommentiert - nicht lauffaehig!
2 #
```

13.2 Verschachtelte Unterprogrammaufrufe (Prozedur-Konvention)

```
.text
4 main:  # ... tu dies ...
        jal sub                # Aufruf des Unterprogramms
6       # ... tu das ...
        jal sub                # Aufruf des Unterprogramms
8
        li $v0, 10            # Exit
10      syscall
12
        #
        #
14 # Unterprogramm sub
    # Eingabe:      $a0: a > 0
16 #               $a1: b > 0
    # Ausgabe:     $v0: 0, wenn a und b teilerfremd, 1 sonst
18 #               $a0, $a1: unverändert
    #               $t0: verändert, Datenschrott
20 #               $t1: verändert, Datenschrott
    #               $t2: verändert, Datenschrott
22 #               alle anderen: unverändert
    sub:  # ... tu dies und das ...
24      jr $ra                # Zurück zum Aufrufer
```

In diesem Fall müsste also der Aufrufer dafür Sorge tragen, dass er relevante Daten in den Registern $\$t0$ bis $\$t2$ zuvor sichert (*caller saved*). Hierfür bietet sich der Stack an.

caller saved

Es wäre natürlich auch denkbar, dass das Unterprogramm alle Register wieder so hinterlässt wie es sie vorgefunden hat (*callee saved*). Beide Verfahren haben Vor- und Nachteile. Muss der callee alle von ihm benutzten Register sichern, so sichert er vielleicht auch Register, die vom caller nie benutzt wurden oder deren Inhalt jetzt nicht mehr benötigt wird. Lässt man alle Register vom caller sichern, so sichert dieser vielleicht Register, die der callee gar nicht verändert. Ein Mittelweg, auf dem Register teilweise vom caller und teilweise vom callee gesichert werden, scheint viel versprechend zu sein.

callee saved

Sicherlich könnten die Autoren der Unterprogramme entsprechend genaue Angaben über veränderte Register machen. Die Einhaltung dieser Spezifikationen durch die Programmierer der caller ist jedoch fehlerträchtig. Zudem wäre es wünschenswert, dass einige Register vom caller nie gesichert werden müssten, während andere stets gesichert werden müssten. Erstere Register würden der Haltung langlebiger Variabler dienen, letztgenannte der Haltung von Zwischenergebnissen.

Eine entsprechende Konvention wurde für den **MIPS** vereinbart. Wir werden sie in dem folgenden Abschnitt kennen lernen.

13.2 Verschachtelte Unterprogrammaufrufe (Prozedur-Konvention)

Die **MIPS**-Prozedur-Konvention (*Procedure Call Convention*) lässt sich in vier Teile gliedern:

Procedure Call Convention

1. den Prolog des Callers,
2. den Prolog des Callees,
3. den Epilog des Callees und
4. den Epilog des Callers.

13.2.1 Prolog des Callers


(a) Sichere alle caller-saved Register:

Wir müssen alle Register, die vom callee verändert werden dürfen, die wir aber nach Beendigung des callees noch benötigen, sichern. Dies sind die Register $\$a0$ - $\$a3$,

13. PROZEDUREN

\$t0-\$t9 sowie \$v0 und \$v1. Alle weiteren Register muss der callee im gleichen Zustand hinterlassen wie er sie vorgefunden hat.⁸


Unser Stack sieht nun so aus:



(b) Übergebe die Argumente: Die ersten vier Argumente übergeben wir in den Registern \$a0 bis \$a3. Sollen weitere Argumente übergeben werden, so werden diese in *umgekehrter* Reihenfolge (das fünfte als letztes) auf dem Stack übergeben.

Hierbei ist die Art der Parameterübergabe (call-by-value bzw. call-by-reference) zu beachten, siehe Ende dieses Abschnittes!

Der Stack sieht jetzt so aus:



(c) Starte die Prozedur (jal)

Unter Nr. 2 haben wir erfahren, wo und in welcher Reihenfolge Argumente an die Prozedur übergeben werden. Bei einfachen Datentypen ist dieses Verfahren auch zufrieden stellend. Was aber soll mit größeren Argumenten, etwa Feldern, geschehen? Aus Modula-2 kennen wir *Wert- und Variablenparameter*. Wie können diese auf Assemblerebene realisiert werden?

Wert- und Variablenparameter

call-by-value

Zunächst zum Fall des *call-by-value* oder Wertparameters: Hier wird einfach der *Wert* des Parameters übergeben.

call-by-reference

Der zweite Fall ist fast noch simpler, denn er erlaubt die Übergabe komplexer Strukturen durch die schlichte Übergabe der *Adresse*, an der sich das Objekt befindet. Dieses Verfahren nennt man *call-by-reference* oder Variablenparameter. Wir benötigen dafür noch einmal einen neuen Befehl:

Befehl	Argumente	Wirkung	Erläuterung
la [Ⓢ]	label	Lade die Adresse des Labels, nicht seinen Wert	load address
	:=	lui Rd, upper ori Rd, \$at, lower # upper = label DIV 2 ³² # lower = label MOD 2 ³²	

Wesentlicher Unterschied zwischen beiden Verfahren ist, dass beim Variablenparameter der Parameter selbst vom callee verändert werden kann, was beim Wertparameter nicht möglich ist.

⁸[Pat, S. A-25] gibt eine ungünstige Reihenfolge an. Richtig ist die Reihenfolge 2-1-3. \$v0 und \$v1 werden dort nicht gesichert, weil es unüblich ist, dort Daten zu halten. Tatsächlich dürfen diese aber verändert werden.

13.2.2 Prolog des Callees


(a) Schaffe Platz für den Stackframe:

Unsere Prozedur-Konvention sieht einen sogenannten *Stackframe* vor, einen Bereich auf dem Stack, der in (fast) jeder Prozedur aufgebaut wird und der ein „Fenster“ auf dem Stack bezeichnet, in dem Argumente und lokale Variablen des callees zu sehen sind. Für diesen Stackframe müssen wir zunächst Platz schaffen, indem wir den Stackpointer um die Größe des Stackframes verringern. Der Stackframe hat die Größe

$$(\text{Zahl der zu sichernden Register} + \text{Zahl der lokalen Variablen}) \times 4$$

Der Stackpointer ist übrigens das einzige Register, das der Callee nicht sichern muss, obwohl es in der Regel verändert wird. Dies liegt daran, dass der Stackpointer nach Abarbeitung des Callees wieder im Ursprungszustand sein sollte.

Der Stack sieht jetzt so aus:



(b) Sichere alle callee-saved Register, die in der Prozedur verändert werden:


Verändert der Callee eines der Register $\$fp$, $\$ra$ und $\$s0-\$s7$, so müssen diese zuvor gesichert werden. Dabei darf das Register $\$ra$ nicht vergessen werden, obwohl es nur indirekt durch den Befehl `jal` geändert wird!

Das Register $\$fp$ sollte als erstes gesichert werden, und zwar auf den Speicherplatz, auf den zu Prozedurebeginn der Stackpointer zeigte. Daran schließen sich die oben genannten Register in beliebiger Reihenfolge an.

Alle callee-saved Register, insbesondere $\$a0-\$a3$, dürfen verändert werden!

13. PROZEDUREN

Ein Blick auf die Lage auf dem Stack:




(c) Erstelle den Framepointer:

Framepointer

Die letzte Maßnahme innerhalb des callee-Prologs ist das Erstellen des *Framepointers*, einem besonderen Register ($\$fp$), das den Wert des Stackpointers zu Beginn der Prozedur erhält. Da dieser Wert bereits in Schritt Nr. 1 verändert wurde, müssen wir ihn erst wieder durch Addition der Größe des Stackframe zum Stackpointer ermitteln. Das Ergebnis legen wir in $\$fp$ ab.

Der Stack sieht jetzt so aus:



Besondere Beachtung verdient der Umstand, dass der aktuelle Framepointer auf genau die Stelle auf dem Stack zeigt, an der der *vorherige* Framepointer liegt. So können wir auf die Argumente und lokalen Variablen der vorherigen Prozedur zugreifen. Ein effizientes, aber etwas fehleranfälliges, Verfahren!

13.2.3 Der Callee

Der Callee hat nun durch positive Indizes (z.B. $8(\$fp)$) die Möglichkeit auf den Wert der Argumente 5 ff. zuzugreifen und durch negative Indizes (z.B. $-12(\$fp)$) auf seine lokalen Variablen. Wobei bei letzteren natürlich die Werte der gesicherten Register keinesfalls überschrieben werden dürfen.

13.2.4 Epilog des Callees


- (a) Rückgabe des Funktionswertes:

Handelt es sich bei dem Callee um eine Funktion, so müssen wir das Funktionsergebnis im Register `$v0` (bzw. in `$v0` und `$v1`) ablegen. Natürlich bietet es sich an diese Register dann bereits bei der Berechnung des Funktionsergebnisses zu verwenden.

- (b) Stelle die gesicherten Register wieder her:

Alle vom Callee gesicherten Register werden wieder hergestellt. Zweckmäßigerweise stellen wir den Framepointer als letztes Register wieder her.


Der Stack sieht nun wieder so aus:



- (c) Entferne den Stackframe:

Wir stellen den Stackpointer wieder her, indem wir die Größe des Stackframes zum Stackpointer addieren⁹.

Der Stack sieht nun so aus wie zu Beginn der Prozedur:



- (d) Springe zum Caller zurück:

`jr $ra`

13.2.5 Epilog des Callers

- (a) Stelle die gesicherten Register wieder her:

Alle vom Caller gesicherten Register werden wieder hergestellt. Achtung: Eventuell über den Stack übergebene Argumente bei der Berechnung des Abstandes zum Stackpointer beachten!

- (b) Stelle den ursprünglichen Stackpointer wieder her:

Multipliziere die Zahl der Argumente und gesicherten Register mit vier und addiere sie zum Stackpointer.

⁹[Pat, S. A-25]: '3. Pop the stack frame by subtracting[sic!] the frame size from `$sp`' ist falsch.

13. PROZEDUREN

Der Prozeduraufruf ist abgeschlossen.

13.2.6 Ein Beispiel zum Prozeduraufruf

Wir wollen uns nun einmal die Prozedur-Konvention an einem Beispiel verdeutlichen. Um uns auch eines kleineren Beispiels bedienen zu können, werden wir davon ausgehen, dass die Register `$a1` bis `$a3` nicht existent sind, da wir sonst die Parameterübergabe auf dem Stack nicht mit einem handlichen Beispiel demonstrieren könnten.

Ballot-Zahlen

Eines der Standard-Beispiele für Prozeduren –und ganz besonders für rekursive– sind die Fibonacci-Zahlen. Nur wenig komplizierter sind die *Ballot-Zahlen*, die gleich zwei Parameter haben und deshalb an dieser Stelle als Beispiel dienen sollen.

Die Ballot-Zahlen (engl. ‘ballot’, frz. ‘ballotage’: Stichwahl) geben die Zahl der Möglichkeiten an, nach einer schriftlichen Stichwahl zwischen zwei Kandidaten die Stimmen so auszuzählen, dass bei der Auszählung stets derselbe Kandidat vorne liegt. Als Parameter dienen die Zahl der Stimmen für jeden der Kandidaten.

Das Bildungsgesetz lautet $a_{n,m} = a_{n,m-1} + a_{n-1,m}$ wobei $a_{n,1} = -1, n > 1, a_{1,m} = 1, m > 1$ und $a_{1,1} = 0$.

Es ergibt sich dann folgende Wertetabelle:

		m					
		1	2	3	4	4	6
n	1	0	-1	-1	-1	-1	-1
	2	1	0	-1	-2	-3	-4
	3	1	1	0	-2	-5	-9
	4	1	2	2	0	-5	-14
	5	1	3	5	5	0	-14
	6	1	4	9	14	14	0

Beispiel 15: Beispiel zur Prozedur-Konvention: Ballot-Zahlen

```
# BALLOT.S (c) 04.97
2 # Beispiele fuer die SPIM-Prozedurkonvention
# Ballot-Zahlen
4 # Zur Verdeutlichung der Konvention seien die Register $a1 bis $a3
# nicht existent.
6 .data
input1: .asciiz "\nBallot_(n,m)\nn=0:_Ende_\n\tn=_\n"
8 input2: .asciiz "\tm=_\n"
output: .asciiz "\nBallot_(n,m)_=_\n"
10 .text
main:
12 # Testumgebung fuer die Prozedur ballot
# Ruft ballot mit den vom Benutzer eingegebenen Werten
14 # auf bis dieser mit n oder m = 0 beendet.
li $v0, 4 # Eingaben anfordern
16 la $a0, input1
syscall
18 li $v0, 5
syscall
20 beqz $v0, exit
move $t0, $v0 # $t0 := n
22 li $v0, 4
la $a0, input2
24 syscall
li $v0, 5
26 syscall
beqz $v0, exit
28 sub $sp, $sp, 4 # Prolog (1)
# 1(a) Register sichern (hier: keine)
30 sw $v0, 4($sp) # 1(b) Argumente uebergeben
move $a0, $t0 #
32 jal ballot # 1(c) Prozedurstart
```

13.2 Verschachtelte Unterprogrammaufrufe (Prozedur-Konvention)

```

34                                     # Epilog (4)
34                                     # 4(a) Register wiederherstellen (keins)
    add    $sp, $sp, 4                # 4(b) Stackpointer wiederherstellen
36    move  $t0, $v0                   # $t0:=ballot(n,m)
    li     $v0, 4                      # Ergebnis ausgeben
38    la    $a0, output
    syscall
40    li    $v0, 1
    move  $a0, $t0
42    syscall
    j     main
44 exit:  li    $v0, 10
    syscall

46                                     #
ballot:
48 # Berechnet die Ballot-Zahl ballot (n,m)
# Argument 1 ($a0) : n
50 # Argument 2 (Stack): m => 4($fp)
# Ergebnis in $v0
52 # Registerbelegung innerhalb der Prozedur:
# $a0: n
54 # $s0: m (besser waere $t0, hier nur zur Demonstration!)
# $t0: Zwischenergebnis

56                                     # Prolog (2)
    sub    $sp, $sp, 12                # 2(a) Stackframe erstellen
58    sw    $fp, 12($sp)                # 2(b) Register sichern
    sw    $ra, 8($sp)
60    sw    $s0, 4($sp)
    addi   $fp, $sp, 12                # 2(c) Framepointer erstellen
62    lw    $s0, 4($fp)                # m laden
# Ballot-Zahlen berechnen...
64    beq   $a0, 1, min1                # IF n = 1 THEN ballot = -1
    beq   $s0, 1, plus1                # ELSIF m = 1 THEN ballot = 1
66                                     # ELSE ballot (n,m) =
# ballot (n, m-1) +
68                                     # ballot (n-1, m) END;
# Aufruf ballot (n, m-1)
70                                     # Prolog (1)
    sub    $sp, $sp, 12                # 1(a) Register sichern
72    sw    $a0, 12($sp)                # $a0 sichern, $t0 noch nicht benutzt
    sw    $s0, 8($sp)                  # Rueckrechnung waere einfacher!
74    sub   $s0, $s0, 1                 # 1(b) Arg. ueberg., $a0 ist bereits ok,
    sw    $s0, 4($sp)                 # (m-1) muss berechnet werden
76    jal   ballot                      # 1(c) Prozedurstart
# Epilog (4)
78    lw    $a0, 12($sp)                # 4(a) Register wiederherstellen
    lw    $s0, 8($sp)
80    addi   $sp, $sp, 12                # 4(b) Stackpointer wiederherstellen
    move  $t0, $v0                      # Ergebnis sichern
82    sub   $a0, $a0, 1                 # Aufruf ballot (n-1, m)
# Prolog (1)
84    sub   $sp, $sp, 8                 # 1(a) Register sichern
    sw    $t0, 8($sp)                  # $t0 sichern, $a0 nicht mehr benutzt
86    sw    $s0, 4($sp)                 # 1(b) Arg. ueberg., $a0 ist bereits ok
    jal   ballot                      # 1(c) Prozedurstart
88                                     # Epilog (4)
    lw    $t0, 8($sp)                  # 4(a) Register wiederherstellen
90    addi   $sp, $sp, 8                 # 4(b) Stackpointer wiederherstellen
    add   $v0, $v0, $t0                # Funktionsergebnis berechnen
92 tollab:                             # Epilog (3)
# 3(a) Funktionswert (liegt in $v0)
94                                     # 3(b) Register wiederherstellen
    lw    $fp, 12($sp)
    lw    $ra, 8($sp)
96    lw    $s0, 4($sp)
    addi   $sp, $sp, 12                # 3(c) Stackframe entfernen
98    jr    $ra                          # 3(d) Rucksprung
min1:  li    $v0, -1                    # ballot = -1
100    j     tollab
plus1: li    $v0, 1                      # ballot = 1
102    j     tollab

```

13. PROZEDUREN

13.2.7 Zusammenfassung: Prozedur-Konvention beim SPIM

Am Ende dieses Abschnittes hier noch eine Zusammenfassung der Konvention:

1. Prolog des Callers:
 - (a) Sichere alle caller-saved Register:

Sichere den Inhalt der Register $\$a0$ - $\$a3$, $\$t0$ - $\$t9$ sowie $\$v0$ und $\$v1$. Der Callee darf ausschließlich diese Register verändern ohne ihren Inhalt wieder herzustellen.¹⁰
 - (b) Übergebe die Argumente:

Die ersten vier Argumente werden in den Register $\$a0$ bis $\$a3$ übergeben, alle weiteren werden auf dem Stack. Das fünfte Argument kommt *zuletzt* auf den Stack. Achtung: Parameterübergabe (Call-by-Value bzw. Call-by-Reference) beachten!
 - (c) Starte die Prozedur (`jal`)
2. Prolog des Callees:
 - (a) Schaffe Platz für den Stackframe:

Subtrahiere die Größe des Stackframes vom Stackpointer.
 - (b) Sichere alle callee-saved Register, die in der Prozedur verändert werden:

Der Callee muss die Register $\$fp$, $\$ra$ und $\$s0$ - $\$s7$ sichern, wenn sie innerhalb der Prozedur verändert werden. Achtung: das Register $\$ra$ wird durch den Befehl `jal` geändert!
 - (c) Erstelle den Framepointer:

Addiere die Größe des Stackframe zum Stackpointer und lege das Ergebnis in $\$fp$ ab.
3. Epilog des Callees:
 - (a) Rückgabe des Funktionswertes:

Handelt es sich bei dem Callee um eine Funktion, so muss das Funktionsergebnis in den Registern $\$v0$ und $\$v1$ abgelegt werden.
 - (b) Stelle die gesicherten Register wieder her:

Alle vom Callee gesicherten Register werden wieder hergestellt. Achtung: den Framepointer als letztes Register wieder herstellen!
 - (c) Entferne den Stackframe:

Addiere¹¹ die Größe des stack frames vom Stackpointer.
 - (d) Springe zum Caller zurück:

`jr $ra`
4. Epilog des Callers:
 - (a) Stelle die gesicherten Register wieder her:


Alle vom Caller gesicherten Register werden wieder hergestellt. Achtung: Evtl. über den Stack übergebene Argumente bei der Berechnung des Abstandes zum Stackpointer beachten!
 - (b) Stelle den ursprünglichen Stackpointer wieder her:

Multipliziere die Zahl der Argumente und gesicherten Register mit vier und addiere sie zum Stackpointer.

¹⁰[Pat, A-25] gibt eine ungünstige Reihenfolge an. Richtig: 2-1-3. $\$v0$ und $\$v1$ werden dort nicht gesichert, weil es unüblich ist, dort Daten zu halten. Tatsächlich dürfen diese aber verändert werden.

¹¹[Pat, S. A-25]: '3. Pop the stack frame by subtracting[sic!] the frame size from $\$sp$ ' ist falsch.

13.2 Verschachtelte Unterprogrammaufrufe (Prozedur-Konvention)



14 Unterbrechnungen und Ausnahmen

In Abschnitt 6.2 auf Seite 22 haben wir bereits von Unterbrechnungen und Ausnahmen gehört. Sie sollen nun noch Gegenstand eines ganzen Abschnittes in diesem Tutorial sein.

Es ist jedoch anzumerken, dass SPIM hier deutliche Schwächen aufweist. So werden etliche Befehle, die wir benötigen, nicht ganz korrekt simuliert. Um unnötigen Frust zu vermeiden, sollten wir uns daher auf eine theoretische Behandlung dieses interessanten Teils der Assemblerprogrammierung beschränken. Mir bekannte Fehler des Simulators sind in Fußnoten erläutert.

Dieser Abschnitt soll lediglich ein Leitfaden sein, keinesfalls ist das Thema Unterbrechnungen und Ausnahmen hier erschöpfend behandelt.

14.1 Zu den Begriffen „Unterbrechnung“ und „Ausnahme“

Interrupt, Trap, Ausnahme, exception usw., all dies sind Begriffe, die fast jede(r) Informatikstudent(in) schon mal gehört und zumindest eine ungefähre Vorstellung von ihrer Bedeutung hat. Bevor wir uns mit ihnen beschäftigen, wollen wir versuchen das Begriffswirrwarr etwas zu lichten und diese Begriffe im Sinne des MIPS zu definieren, denn nicht überall sind die Begriffe gleich definiert.

Interrupt, Unterbrechnung Ein *Interrupt* oder eine *Unterbrechnung* ist ein Ereignis, das *asynchron* zum Programmablauf eintritt, also nicht in direkter Abhängigkeit zu bestimmten Befehlen steht, und eine (sofortige) Behandlung durch das Betriebssystem notwendig macht. Typische Beispiele sind Ein- und Ausgabegeräte, z.B. die Tastatur. Unterbrechnungen sind nicht reproduzierbar, sie lauern immer und überall während der gesamten Programmausführung.

exception, Ausnahme, Trap Eine *exception* oder eine (*Ausnahme*), gelegentlich auch *Trap*, ist ein Ereignis, das *synchron* zum Programmablauf eintritt, also in Zusammenhang mit bestimmten Befehlen, und eine Behandlung durch das Betriebssystem notwendig macht. Typisches Beispiel sind arithmetische Fehler, wie z.B. die Division durch Null, Überläufe etc. Ebenfalls zu den Ausnahmen zählen ungültige Instruktionen, wie sie beim Sprung ins Datensegment auftreten können, oder der Aufruf des Betriebssystems durch das Benutzerprogramm, beim MIPS mit dem bereits bekannten Befehl `syscall`.¹² Bei gleichem Ausgangszustand des Rechnersystems sind Ausnahmen im Gegensatz zu Unterbrechnungen reproduzierbar, d.h. sie treten immer wieder an derselben Stelle auf.

Hardwarefehler können sowohl asynchrone Unterbrechnungen verursachen als auch synchrone Ausnahmen, je nachdem wo und wie sie auftreten.

14.2 Die Behandlung von Unterbrechnungen und Ausnahmen

Da die Stellen, an denen Ausnahmen auftreten können, bekannt sind, könnte man Ausnahmen vollständig durch das Benutzerprogramm abarbeiten lassen, etwa durch vorherige Tests der Operanden oder durch Überprüfung des Ergebnisses. Da der Benutzer aber fast immer korrekte Ergebnisse haben will, würde bei fast jedem arithmetischen Befehl Code zur Überprüfung des Ergebnisses stehen, eine ineffiziente Lösung, zumal sich gerade die arithmetischen Ausnahmen sehr gut auf Schaltungsebene erkennen lassen.

¹²Im SPIM werden die Betriebssystemaufrufe allerdings nicht simuliert, sondern vom Simulator direkt verarbeitet.

14.2 Die Behandlung von Unterbrechungen und Ausnahmen

Folglich behandelt man Unterbrechungen und Ausnahmen gemeinsam durch das Betriebssystem. Beim Auftreten eines solchen Ereignisses wird nach Abarbeitung des aktuellen Befehls ein Sprung an eine von der CPU-Hardware festgelegte Stelle erfolgen, beim SPIM nach 0x8000 0080. Dort liegt eine sogenannte *Unterbrechungsbehandlungsroutine* oder *Interrupthandler*, die die Unterbrechung bzw. die Ausnahme behandelt. Hierfür stehen im Koprozessor 0 einige Register zur Verfügung, in die die CPU Informationen über den Grund der Unterbrechung oder der Ausnahme schreibt. Von den zahlreichen Registern des Koprozessors 0 werden die vier in Abbildung 10 auf dieser Seite vom SPIM simuliert.

Unterbrechungs-
behandlungsroutine,
Interrupthandler

Abbildung 10: Register des Koprozessors 0

Register-name	Register-nummer	Inhalt
BadVAddr	8	Hauptspeicheradresse, auf welche zugegriffen werden sollte. Dieses Register ist zur Verwaltung des virtuellen Speichersystems notwendig.
Status	12	Unterbrechungsmaske und interrupt bits
Cause	13	Ausnahmetyp und wartende (noch nicht bearbeitete) Unterbrechungen
EPC	14	Adresse des Befehls, der die Ausnahme auslöste.

Auf den Inhalt dieser Register kann mit denen aus dem Abschnitt 15 bekannten Register-Transfer- und Ladebefehlen, diesmal mit der Koprozessornummer 0, zugegriffen werden: `lwc0`, `mfc0`, `mtc0` und `swc0`.

14.2.1 Das Statusregister

Im Statusregister ist in den Bits 15 bis 8 eine Interruptmaske abgelegt. Ist ein Bit 1, so erlaubt es auf dieser Ebene Unterbrechungen. Fünf Hardware- und drei Softwareunterbrechungen sind möglich.

Die Bits 5 bis 0 (siehe Abbildung 11 auf dieser Seite) enthalten Informationen über den Status des Systems. Die sechs Bits sind in drei Zweiergruppen eingeteilt, die für das aktuelle Programm, das vorangegangene und vorvorangegangene Programm enthalten. Das Interrupt-enable-Bit ist 1, wenn Unterbrechungen zugelassen sind, 0 sonst. Das Kernel/User-Bit ist 1, wenn das Programm unter Kernel-Berechtigung lief.

Abbildung 11: Das Statusregister

Bit	Bedeutung	
0	Interrupt enable	aktueller
1	Kernel/User	Prozess
2	Interrupt enable	vorheriger
3	Kernel/User	Prozess
4	Interrupt enable	vorvorheriger
5	Kernel/User	Prozess
6-7	—	
8-15	Interruptmaske	
16-31	—	

14. UNTERBRECHUNGEN UND AUSNAHMEN

14.2.2 Das Cause-Register

Pending Interrupts
exception Code

Vom Cause-Register werden vom SPIM die Bits 15 bis 8¹³, die die noch nicht bearbeiteten Unterbrechungen (*Pending Interrupts*), und die Bits 5 bis 2, die den Grund der Unterbrechung (*exception Code*) enthalten, simuliert. Die Ausnahmecodes sind der Abbildung 12 auf dieser Seite zu entnehmen.

Abbildung 12: Ausnahmecodes (exception codes) (Bit 5 bis 2 des Cause-Registers)

Code	Name	Erläuterung
0	INT	Externe Unterbrechung
4	ADDRL	Adressfehler (Ladebefehl, auch beim Laden des Befehls!)
5	ADDRS	Adressfehler (Speicherbefehl)
6	IBUS	Busfehler beim Laden des Befehls
7	DBUS	Busfehler beim Laden oder Speichern von Daten
8	SYSCALL	Betriebssystemaufruf durch den Befehl <code>syscall</code>
9	BKPT	Breakpoint
10	RI	Verwendung einer privilegierten Instruktion (reserved instruction)
12	OVF	arithmetischer Überlauf (Overflow)

Für die nicht bearbeiteten Unterbrechungen stehen fünf Hardware-Level und drei Software-Level zur Verfügung. Die Bits 15 bis 8 entsprechen den Bits 15 bis 8 des Status-Registers.

14.2.3 Die Unterbrechungsbehandlungsroutine

Die Unterbrechungsbehandlungsroutine soll hier nicht im Detail beschrieben, sondern nur in ihren Grundfunktionen erläutert werden.

Kernelsegment

`.ktext`-Direktive

Die Ausführung der Unterbrechungsbehandlungsroutine muss unter besonderen Betriebssystemrechten geschehen, deshalb wird sie im *Kernelsegment* abgelegt, einem Segment, in dem der Betriebssystemkern untergebracht wird. Um Befehle ins Kernelsegment zu schreiben steht uns die *.ktext-Direktive* zur Verfügung. Ihr Syntax entspricht dem der bereits bekannten *.text-Direktive*.

Da bei Unterbrechungen und Ausnahmen zur Adresse 0x8000 0080 gesprungen wird, muss dafür Sorge getragen werden, dass die Unterbrechungsbehandlungsroutine an dieser Stelle liegt. Wir müssen die *.ktext-Direktive* also mit Hilfe des optionalen Parameters die Stelle angeben, an der unsere Routine liegen soll: `.ktext 0x8000 0080`.

`.kdata`-Direktive

Analog zur *.data-Direktive* gibt es auch eine *.kdata-Direktive*, die Daten im Kernel-datensegment ablegt.

Im Kernelsegment stehen die für uns bisher verbotenen CPU-Register `$k0` und `$k1` zur Verfügung. Andere Register dürfen nur dann verwendet werden, wenn sie zuvor gesichert werden!

Aus dem Statusregister kann nun ermittelt werden, ob es sich um eine Unterbrechung handelt, dann ist es $> 0x44$ oder eine Ausnahme ($< 0x44$). Entsprechend kann nun durch Fallunterscheidungen ermittelt werden, welche Unterbrechung oder welche Ausnahme aufgetreten ist und diese dann bearbeitet werden. Schlimmstenfalls muss das System auch angehalten werden, bzw. der Prozess gelöscht werden.

Soll die Programmausführung nach der Unterbrechungsbehandlung fortgesetzt werden, so muss zunächst die korrekte Rücksprungadresse ermittelt werden. Da die Adresse des

¹³[Pat, S. A-32, Fig. A.13] gibt falsch Bits 15 bis 10 an!

14.2 Die Behandlung von Unterbrechungen und Ausnahmen

Befehls, an der die Ausnahme auftrat, im Register EPC steht, kann diese einfach um vier erhöht und dann mittels des Befehls `jr` angesprungen werden.

Zuvor muss jedoch noch das Statusregister wieder hergestellt werden, wofür es einen eigenen Befehl gibt:

Befehl	Argumente	Wirkung	Erläuterung
<code>rfe</code>		Stellt das Statusregister wieder her.	return from exception

Zu Testzwecken kann der folgende Befehl `break`¹⁴ verwendet werden, den wir bereits in einigen Pseudobefehlen kennen gelernt haben:

Befehl	Argumente	Wirkung	Erläuterung
<code>break</code>	<code>Imm</code>	Verursacht die Ausnahme <code>Imm</code>	<code>break</code>

¹⁴Der Befehl arbeitet leider fehlerhaft.

15. GLEITKOMMAZAHLEN

15 Gleitkommazahlen

Gleitkommazahlen

Wenn wir uns bisher mit Zahlen beschäftigt haben, dann handelte es sich stets um ganze Zahlen. In der realen Welt nehmen diese jedoch nur einen kleinen Bereich ein, so dass in vielen Anwendungen die Arbeit mit rationalen Zahlen notwendig ist. Computer arbeiten nur mit einer kleinen Teilmenge der rationalen Zahlen, ebenso wie sie nur mit einer winzigen Teilmenge der ganzen Zahlen arbeiten (beim **MIPS** mit denjenigen ganzen Zahlen zwischen -2^{31} und $+2^{31} - 1$). Das Pendant der rationalen Zahlen im Computerbereich heißt *Gleitkommazahlen oder Floating-Point-Zahlen*. Die Repräsentation von Gleitkommazahlen ist [Pat, S. 226f] zu entnehmen, der gleiche Abschnitt 4.8 beschäftigt sich auch mit Rechenverfahren für Gleitkommazahlen.

Koprozessor
floating point unit

Der **MIPS**-Prozessor kann selber keine Gleitkommazahlen verarbeiten. Er bedient sich dazu eines (*mathematischen*) *Koprozessors* für Gleitkommazahlen, er wird auch *FPU (floating point unit)* genannt. Der **MIPS**-Prozessor kann über mehrere Koprozessoren für verschiedene Aufgaben verfügen. Im **SPIM**-Simulator werden zwei von ihnen simuliert:

1. Koprozessor 0:

Er wird bei der Behandlung von Unterbrechungen und Ausnahmen verwendet. Im Original-Prozessor wird dieser Prozessor auch bei der Hauptspeicherverwaltung verwendet. Diese wird jedoch vom **SPIM** nicht simuliert. Er wird in Abschnitt 14 auf Seite 64 weiter behandelt.

2. Koprozessor 1:

Er ist der mathematische Koprozessor und zuständig für Berechnungen mit Gleitkommazahlen.

Der mathematische Koprozessor verfügt wie der Hauptprozessor über 32 Register von je 32 Bit Breite. Die Register sind von $\$0$ bis $\$31$ durchnummeriert und haben die Namen $\$f0$ bis $\$f31$. Es handelt sich um echte general purpose register, alle Register sind also beschreibbar und für jeden beliebigen Zweck einsetzbar. Mit zwei Einschränkungen: Rechnet man mit doppelt genauen Gleitkommazahlen, so werden hierfür 64 Bit benötigt, die wir durch Zusammenfassen je zweier Register erhalten. Zusammengefasst werden ein Register mit einer geraden Registernummer und das folgende Register, also $\$f0$ und $\$f1$, $\$f2$ und $\$f3$ usw.

Die zweite Einschränkung betrifft die Register $\$f0$ und $\$f1$ (im Folgenden werden wir für dies „Doppelregister“ $\$f0/1$ schreiben), in die das Betriebssystem die Benutzereingaben schreibt (Betriebssystemfunktionen 6 und 7) und die Register $\$f12/13$, aus denen die Ausgaben gelesen werden (Betriebssystemfunktionen 2 und 3). Vergleiche hierzu die Abbildung 4 auf Seite 29 bzw. Abschnitt A.5 auf Seite 82!

Leider funktioniert das Registerfenster nicht unter allen Windows-Versionen, es werden lediglich die Werte 0.000... bzw. -0.000... angezeigt. Dieses Problem betrifft jedoch nur die *Anzeige* der Werte!

15.1 Datenhaltung III: Gleitkommazahlen

Gleitkommazahlen werden wie alle anderen Daten auch im Datensegment gehalten. Für das Ablegen von Gleitkommazahlen stehen zwei Direktiven zur Verfügung:

`.float`

einfache Genauigkeit

- `.float`

Die Direktive `.float` legt die folgenden Werte als 32-Bit-Gleitkommazahlen ab. Man spricht auch von Gleitkommazahlen *einfacher Genauigkeit* oder *single precision*.

Die Werte können in nicht normalisierter Form in Exponentialschreibweise eingetragen werden: $123.45e67$, $0.0003e-9$. Das *e* muss klein geschrieben sein. Auf die Exponentialschreibweise kann auch verzichtet werden, dann muss allerdings ein Dezimalpunkt vorhanden sein: $1234567.$ und 0.09999 sind erlaubt, 1234567 jedoch nicht.

- `.double`

Die Direktive `.double` legt die folgenden Werte als 64-Bit-Gleitkommazahlen ab. Man spricht auch von Gleitkommazahlen *doppelter Genauigkeit* oder *double precision*.

15.2 Transferbefehle

Daten können zwischen einem der Koprozessoren und dem Hauptprozessor oder dem Speicher hin- und herbewegt werden. Hierfür dienen die folgenden Befehle.

Für die Notation der Koprozessorregister verwenden wir die Notation aus Abbildung 13 auf dieser Seite.

Abbildung 13: Neue Argumentarten

Notation	Erläuterung	Beispiel
Cd, Cs	Koprozessor-Registeroperanden, d: destination (Ziel), s: source (Quelle) Achtung: Registernamen wie z.B. $\$f10$ sind hier nicht erlaubt!	$\$1, \3
$Fd, Fs, Fs1, Fs2$	Registeroperanden des <i>mathematischen</i> Koprozessors	$\$f0, \$f13$

15.2.1 Ladebefehle

Für jeden Koprozessor gibt es einen eigenen Ladebefehl `lwc(z)`, wobei für $\langle z \rangle$ eine Zahl zwischen 0 und 3 zu wählen ist. Der mathematische Koprozessor hat die Nummer 1.

Befehl	Argumente	Wirkung	Erläuterung
<code>lwc(z)</code>	Cd, Adr	$Cd := Mem[Adr]; \langle z \rangle: [0..3]$	load word coprocessor
<code>l.d[Ⓟ]</code>	Fd, Adr	$Fd := Mem[Adr]; Fd+1 := Mem[Adr+4]$ <pre>:= lui \$at, Adr lwcl Fd, 0(\$at) lui \$at, Adr lwcl Fd+1, 4(\$at)</pre>	Load floating-point double
<code>l.s[Ⓟ]</code>	Fd, Adr	$Fd := Mem[Adr]$ <pre>:= lui \$at, Adr lwcl Fd, 0(\$at)</pre>	Load floating-point single

Befehle, die ausschließlich der Verarbeitung von Gleitkommazahlen dienen, sind an den Punkten im Befehlsnamen zu erkennen. Die meisten Befehle gibt es in einer Version für einfach genaue Zahlen und in einer für doppelt genaue Zahlen. Die beiden Versionen werden in der Regel durch `.s` (single) und `.d` (double) auseinander gehalten.

Wie eingangs erwähnt, werden doppelt genaue Zahlen in zwei aufeinanderfolgenden Registern abgelegt, beginnend mit einer geraden Registernummer. Die Befehle für doppelt genaue Zahlen arbeiten zwar auch mit ungeraden Registernummern, jedoch basieren die Anzeigen des SPIM auf der Annahme, dass die doppelt genauen Zahlen in geraden Registern beginnen.

15. GLEITKOMMAZAHLEN

15.2.2 Speicherbefehle

Die Speicherbefehle entsprechen weitgehend den bereits bekannten Ladebefehlen:

Befehl	Argumente	Wirkung	Erläuterung
<code>swc(z)</code>	<code>Cs, Adr</code>	<code>Mem[Adr] := Cs; (z): [0..3]</code>	store word coprocessor
<code>s.d[Ⓟ]</code>	<code>Fs, Adr</code>	<code>Mem[Adr] := Fs;</code> <code>Mem[Adr+4] := Fs+1</code> <code>:= lui \$at, Adr</code> <code>swc1 Fd, 0(\$at)</code> <code>lui \$at, Adr</code> <code>swc1 Fd+1, 4(\$at)</code>	Store floating-point double
<code>s.s[Ⓟ]</code>	<code>Fs, Adr</code>	<code>Mem[Adr] := Fs</code> <code>:= lui \$at, Adr</code> <code>swc1 Fd, 0(\$at)</code>	Store floating-point single

15.2.3 Register-Transfer-Befehle

Mit den folgenden Befehlen können Daten zwischen dem Haupt- und dem Koprozessor hin- und herbewegt werden:

Befehl	Argumente	Wirkung	Erläuterung
<code>mfc(z)</code>	<code>Rd, Cs</code>	<code>Rd := Cs; (z): [0..3]</code>	Move from coprocessor
<code>mfc1.d[Ⓟ]</code>	<code>Rd, Cs</code>	<code>Rd := Cs; Rd+1 :=</code> <code>Cs+1</code> <code>:= mfc1 Rd, Cs</code> <code>mfc1 Rd+1, Cs+1</code>	Move double from coprocessor 1
<code>mtc(z)</code>	<code>Rs, Cd</code>	<code>Cd := Rs; (z): [0..3]</code>	Move to coprocessor

Achtung: Der Befehl `mtc(z)` funktioniert in der Windows-Version nicht!
 Zum Laden eines unmittelbaren Wertes müssen wir diesen also zuvor im Speicher ablegen und dann von dort laden.

Innerhalb des mathematischen Koprozessors dienen die folgenden beiden Befehle zum Verschieben des Inhalts von Registern:

Befehl	Argumente	Wirkung	Erläuterung
<code>mov.d</code>	<code>Fd, Fs</code>	<code>Fd/Fd+1 := Fs/Fs+1</code>	Move floating-point double
<code>mov.s</code>	<code>Fd, Fs</code>	<code>Fd := Fs</code>	Move floating-point single

Häufig ist es notwendig ganze Zahlen in Gleitkommazahlen, doppelt genaue Zahlen in einfach genaue Zahlen o.ä., umzuwandeln. Hierfür stehen die folgenden Konvertierungsbefehle zur Verfügung:

Befehl	Argumente	Wirkung	Erläuterung
<code>cvt.d.s</code>	<code>Fd, Fs</code>	<code>Fd/Fd+1 := Fs</code>	Convert single to double
<code>cvt.d.w</code>	<code>Fd, Fs</code>	<code>Fd/Fd+1 := Fs</code>	Convert integer to double
<code>cvt.s.d</code>	<code>Fd, Fs</code>	<code>Fd := Fs/Fs+1</code>	Convert double to single
<code>cvt.s.w</code>	<code>Fd, Fs</code>	<code>Fd := Fs</code>	Convert integer to single
<code>cvt.w.d</code>	<code>Fd, Fs</code>	<code>Fd := Fs/Fs+1</code>	Convert double to integer
<code>cvt.w.s</code>	<code>Fd, Fs</code>	<code>Fd := Fs/Fs+1</code>	Convert single to integer

15.3 Arithmetische Befehle für Gleitkommazahlen

Die Additionsbefehle für Gleitkommazahlen entsprechen im Wesentlichen denen für Ganzzahlen, nur dass es keine Versionen ohne Überlaufbehandlung gibt. Dafür existieren sie in jeweils einer Variante für doppelt und einfach genaue Gleitkommazahlen.

Addition und Subtraktion:

Befehl	Argumente	Wirkung	Erläuterung
add.d	Fd, Fs1, Fs2	$Fd/Fd+1 := Fs1/Fs1+1 + Fs2/Fs2+1$	Floating-point addition double
add.s	Fd, Fs1, Fs2	$Fd := Fs1 + Fs2$	Floating-point addition single
sub.d	Fd, Fs1, Fs2	$Fd/Fd+1 := Fs1/Fs1+1 - Fs2/Fs2+1$	Floating-point subtract double
sub.s	Fd, Fs1, Fs2	$Fd := Fs1 - Fs2$	Floating-point subtract single

Multiplikation und Division:

Befehl	Argumente	Wirkung	Erläuterung
mul.d	Fd, Fs1, Fs2	$Fd/Fd+1 := Fs1/Fs1+1 * Fs2/Fs2+1$	Floating-point multiply double
mul.s	Fd, Fs1, Fs2	$Fd := Fs1 * Fs2$	Floating-point multiply single
div.d	Fd, Fs1, Fs2	$Fd/Fd+1 := Fs1/Fs1+1 / Fs2/Fs2+1$	Floating-point divide double
div.s	Fd, Fs1, Fs2	$Fd := Fs1 / Fs2$	Floating-point divide single

sonstige arithmetische Befehle:

Befehl	Argumente	Wirkung	Erläuterung
abs.d	Fd, Fs	$Fd/Fd+1 := ABS (Fs/Fs+1)$	Floating-point absolute value double
abs.s	Fd, Fs	$Fd := ABS (Fs)$	Floating-point absolute value single
neg.d	Fd, Fs	$Fd/Fd+1 := - Fs/Fs+1$	Floating-point negate double
neg.s	Fd, Fs	$Fd := - Fs$	Floating-point negate single

15.4 Vergleichsbefehle für Gleitkommazahlen

Wir haben sechs verschiedene Vergleichsbefehle zur Verfügung, die jeweils zwei Zahlen vergleichen und als Ergebnis das *Bedingungs-Bit* (*condition-flag*) des mathematischen Koprozessors auf 1 (=wahr) oder auf 0 (=falsch) setzen. Dieses condition-flag kann von der CPU als Sprungkriterium verwendet werden. Die beiden folgenden Befehle verzweigen die Programmausführung in Abhängigkeit von dem condition-flag des Koprozessors:

condition-flag

Befehl	Argumente	Wirkung	Erläuterung
bc(z)t		Sprung, wenn condition-flag des Koprozessors (z) wahr (1) ist	Branch coprocessor (z) true

15. GLEITKOMMAZAHLEN

Befehl	Argumente	Wirkung	Erläuterung
bc⟨z⟩f		Sprung, wenn condition-flag des Koprozessors ⟨z⟩ falsch (0) ist	Branch coprocessor ⟨z⟩ false

Die Vergleichsbefehle für Gleitkommazahlen sind:

Befehl	Argumente	Wirkung	Erläuterung
c.eq.d	Fs1, Fs2	Setze condition-flag auf wahr, wenn $Fs1/Fs1+1 = Fs2/Fs2+1$	Compare equal double
c.eq.s	Fs1, Fs2	Setze condition-flag auf wahr, wenn $Fs1 = Fs2$	Compare equal single
c.le.d	Fs1, Fs2	Setze condition-flag auf wahr, wenn $Fs1/Fs1+1 \leq Fs2/Fs2+1$	Compare less than equal double
c.le.s	Fs1, Fs2	Setze condition-flag auf wahr, wenn $Fs1 \leq Fs2$	Compare less than equal single
c.lt.d	Fs1, Fs2	Setze condition-flag auf wahr, wenn $Fs1/Fs1+1 < Fs2/Fs2+1$	Compare less than double
c.lt.s	Fs1, Fs2	Setze condition-flag auf wahr, wenn $Fs1 < Fs2$	Compare less than single

15.5 Ein Beispiel zu Gleitkommazahlen

Das Hornerschema dient u.a. zur Berechnung von Polynomen n-ter Ordnung. Durch fortgesetztes Ausklammern des x ergibt sich:

$$\begin{aligned}
 & a_n x^n + a_{n-1} x^{n-1} + \dots + a_1 x^1 + a_0 \\
 = & ((((((a_n x) + a_{n-1}) x + a_{n-2}) x + \dots) x + a_1) x + a_0
 \end{aligned}$$

Die rechte Seite des Terms lässt sich einfach iterativ berechnen.

Beispiel 16: Das Hornerschema mit doppelt genauen Gleitkommazahlen

```

# HORNER.S (c) 06/97
2 # Implementation des Hornerschemas fuer Gleitkommazahlen doppelter
  # Genauigkeit.
4 #
  .data
6 Koeffs: .double 1.,      5.,      -2.      # ein Testfeld
  last:   .double -24.
8 .text
  #
10 # Hauptprogramm main
  # Nach Eingabe des x-Wertes wird der Wert des Polynoms zwischen
12 # Koeffs und last an der Stelle x berechnet.
  #
14 __start:
  main:  la    $a0, Koeffs      # Initialisierung
16      la    $a1, last
        li    $v0, 7           # Read_double

```


15.5 Ein Beispiel zu Gleitkommazahlen

```

18      syscall
      jal    poly
20      li    $v0, 3          # Print_double
      syscall
22      li    $v0, 10        # Exit
      syscall
24 # Funktion poly
      # Berechnet den Wert eines Polynoms f an einer gegebenen Stelle x
26 # Eingabe:
      # $a0:  Adresse des hoechststrangigen Koeffizienten a_n des Polynoms.
28 #         Die Koeffizienten sind hintereinander mit sinkender Wertigkeit
      #         als doppelt genaue Gleitkommazahlen abgelegt.
30 # $a1:  Adresse des niedrigstranigen Koeffizienten a_0 (Konstante)
      # $f0/1:  Stelle x, an der das Polynom berechnet werden soll.
32 # $f12/13:Ergebnisakkumulation (akk)
      # zusaetzlich verwendetes Register:
34 # $f10/11:temporaer a_n (tmp); wird gesichert!
      .data
36 db10:  .double 0.
      .text
38 poly:  addiu  $sp, $sp, -8   # Platz fuer $f10/11
      s.d    $f10, 8($sp)    # $f10/11 sichern
40      l.d    $f12, db10     # Initialisierung, Windows-Version kann
      # kein mtc1!
42      # besser waere: Zeile 35/36 loeschen,
      # Zeile 40 ff:
44      # li    $t0, 0
      # mtc1  $t0, $12
46      # cvt.d.w $f12, $f12
      # oder einfach durch
48      # sub.d $f12, $f12, $f12
      bgt    $a0, $a1, exit  # Fehler in der Adressierung
50 repeat: l.d    $f10, ($a0)  # tmp := a_n
      mul.d  $f12, $f12, $f0  # akk := akk * x
52      add.d  $f12, $f12, $f10# akk := akk + a_n
      addiu  $a0, $a0, 8      # n := n + 1; 64-Bit-Zahlen!
54      ble    $a0, $a1, repeat# solange nicht am Ende der Koeffizienten
      exit:  l.d    $f10, 8($sp) # $f10/11 wiederherstellen
56      addiu  $sp, $sp, 8     # $sp wiederherstellen
      jr    $ra              # Bye-bye

```

A. ZUSAMMENFASSENDE TABELLEN

A Zusammenfassende Tabellen

A.1 Befehlsübersicht

Ladebefehle

Befehl	Argumente	Wirkung	Erläuterung	S.
lw	Rd, Adr	Rd:=MEM[Adr]	Load word	18
lb	Rd, Adr	Rd:=MEM[Adr]	Load byte	19
lbu	Rd, Adr	Rd:=MEM[Adr]	Load unsigned byte	19
lh	Rd, Adr	Rd:=MEM[Adr]	Load halfword	19
lhu	Rd, Adr	Rd:=MEM[Adr]	Load unsigned halfword	19
ld [Ⓟ]	Rd, Adr	Lädt das Doppelwort an der Stelle Adr in die Register Rd und Rd+1	Load double-word	19
ulw	Rd, Adr	Rd:=MEM[Adr]	unaligned Load word	19
ulh	Rd, Adr	Rd:=MEM[Adr]	unaligned Load halfword	19
ulhu	Rd, Adr	Rd:=MEM[Adr]	unaligned Load unsigned halfword	19
lwr	Rd, Adr	Rd:=MEM[Adr] DIV 2 ¹⁶	Load word right	19
lwl	Rd, Adr	Rd:=MEM[Adr] MOD 2 ¹⁶	Load word left	19

Speicherbefehle

Befehl	Argumente	Wirkung	Erläuterung	S.
sw	Rs, Adr	MEM[Adr]:=Rs	store word	20
sb	Rs, Adr	MEM[Adr]:=Rs MOD 256	store byte	20
sh	Rs, Adr	MEM[Adr]:=Rs MOD 2 ¹⁶	store halfword	20
sd [Ⓟ]	Rs, Adr	MEM[Adr]:=Rs + 2 ¹⁶ Rs+1	store double-word	20
swl	Rs, Adr	MEM[Adr]:=Rs MOD 2 ¹⁶	store word left	20
swr	Rs, Adr	MEM[Adr]:=Rs MOD 2 ¹⁶	store word right	20
ush	Rs, Adr	MEM[Adr]:=Rs MOD 2 ¹⁶	unaligned store halfword	20
usw	Rs, Adr	MEM[Adr]:=Rs	unaligned store word	20

Register-Transfer-Befehle

Befehl	Argumente	Wirkung	Erläuterung	S.
move [Ⓟ]	Rd, Rs	Rd:=Rs	move	20
li [Ⓟ]	Rd, Imm	Rd:=Imm	load immediate	20
lui	Rs, Imm	Rd:=Imm 2 ¹⁶	load upper immediate	20
la	label	Lade die Adresse des Labels, nicht seinen Wert	load address	56
mfhi	Rd	Rd:=hi	move from hi	21
mflo ¹⁵	Rd	Rd:=lo	move from lo	21
mthi	Rs	hi:=Rs	move to hi	21
mtlo	Rs	lo:=Rs	move to lo	21

Addition und Subtraktion

Befehl	Argumente	Wirkung	Erläuterung	S.
add	Rd, Rs1, Rs2	Rd := Rs1 + Rs2	addition (with overflow)	22
addi	Rd, Rs, Imm	Rd := Rs + Imm	addition immediate (with overflow)	22
sub	Rd, Rs1, Rs2	Rd := Rs1 - Rs2	subtract (with overflow)	22
addu	Rd, Rs1, Rs2	Rd := Rs1 + Rs2	addition (without overflow)	22
addiu	Rd, Rs, Imm	Rd := Rs + Imm	addition immediate (without overflow)	22
subu	Rd, Rs1, Rs2	Rd := Rs1 - Rs2	subtract (without overflow)	22

Multiplikation und Division

Befehl	Argumente	Wirkung	Erläuterung	S.
div	Rd, Rs	hi:=Rd MOD Rs; lo:=Rd DIV Rs	Divide (with overflow)	23
divu	Rd, Rs	hi:=Rd MOD Rs; lo:=Rd DIV Rs	Divide (without overflow)	23
mult	Rd, Rs	hi:=Rd × Rs DIV 2 ¹⁶ ; lo:=Rd × Rs MOD 2 ¹⁶	multiply	23
multu	Rd, Rs	hi:=Rd × Rs DIV 2 ¹⁶ ; lo:=Rd × Rs MOD 2 ¹⁶	Unsigned multiply	23
mul [Ⓟ]	Rd, Rs1, Rs2	Rd:=Rs1 × Rs2	Multiply (without overflow)	24
mulo [Ⓟ]	Rd, Rs1, Rs2	Rd:=Rs1 × Rs2	Multiply (with overflow)	24
mulou [Ⓟ]	Rd, Rs1, Rs2	Rd:=Rs1 × Rs2	Unsigned multiply (with overflow)	24
div [Ⓟ]	Rd, Rs1, Rs2	Rd:=Rs1 DIV Rs2	Divide (with overflow)	24
divu [Ⓟ]	Rd, Rs1, Rs2	Rd:=Rs1 DIV Rs2	Divide (without overflow)	24

sonstige arithmetische Befehle

Befehl	Argumente	Wirkung	Erläuterung	S.
abs [Ⓟ]	Rd, Rs	Rd:=ABS (Rs)	Absolute value	25
neg [Ⓟ]	Rd, Rs	Rd:=-Rs	Negate value (with overflow)	25
negu [Ⓟ]	Rd, Rs	Rd:=-Rs	Negate value (without overflow)	25
rem [Ⓟ]	Rd, Rs1, Rs2	Rd:=Rs1 MOD Rs2	Remainder	25
remu [Ⓟ]	Rd, Rs1, Rs2	Rd:=Rs1 MOD Rs2	Unsigned remainder	25

Betriebssystemfunktion

Befehl	Argumente	Wirkung	Erläuterung	S.
syscall		führt die Systemfunktion in \$v0 aus	vgl. Abbildung 4 auf Seite 29	26

A. ZUSAMMENFASSENDE TABELLEN

elementare logische Befehle

Befehl	Argumente	Wirkung	Erläuterung	S.
and	Rd, Rs1, Rs2	$Rd := Rs1 \wedge Rs2$	and	28
andi	Rd, Rs, Imm	$Rd := Rs \wedge Imm$	and immediate	28
nor	Rd, Rs1, Rs2	$Rd := \overline{Rs1} \vee \overline{Rs2}$	nor	28
or	Rd, Rs1, Rs2	$Rd := Rs1 \vee Rs2$	or	28
ori	Rd, Rs, Imm	$Rd := Rs \vee Imm$	or immediate	28
xor	Rd, Rs1, Rs2	$Rd := Rs1 \oplus Rs2$	exclusive or	28
xori	Rd, Rs, Imm	$Rd := Rs \oplus Imm$	exclusive or immediate	28
not [Ⓟ]	Rd, Rs	$Rd := \overline{Rs}$	not	28

Rotations- und Schiebepfehle

Befehl	Argumente	Wirkung	Erläuterung	S.
rol [Ⓟ]	Rd, Rs1, Rs2	$Rd := Rs1$ um Rs2 Stellen nach links rotiert	rotate left	29
ror [Ⓟ]	Rd, Rs1, Rs2	$Rd := Rs1$ um Rs2 Stellen nach rechts rotiert	rotate right	29
sll	Rd, Rs, Imm	$Rd := Rs \times 2^{Imm}$	Shift left logical	29
sllv	Rd, Rs1, Rs2	$Rd := Rs1 \times 2^{Rs2}$	Shift left logical variable	29
srl	Rd, Rs, Imm	$Rd := Rs \text{ DIV } 2^{Imm}$	Shift right logical	29
srlv	Rd, Rs1, Rs2	$Rd := Rs1 \text{ DIV } 2^{Rs2}$	Shift right logical variable	29
sra	Rd, Rs, Imm	$Rd := Rs \times 2^{Imm}$	Shift right arithmetic	29
srav	Rd, Rs1, Rs2	$Rd := Rs1 \times 2^{Rs2}$	Shift right arithmetic variable	29

Vergleichsbefehle

Befehl	Argumente	Wirkung	Erläuterung	S.
seq [Ⓟ]	Rd, Rs1, Rs2	$Rd := 1$, falls $Rs1 = Rs2$, 0 sonst	(=) set equal	30
sne [Ⓟ]	Rd, Rs1, Rs2	$Rd := 1$, falls $Rs1 \neq Rs2$, 0 sonst	(\neq) set not equal	30
sge [Ⓟ]	Rd, Rs1, Rs2	$Rd := 1$, falls $Rs1 \geq Rs2$, 0 sonst	(\geq) set greater than equal	30
sgeu [Ⓟ]	Rd, Rs1, Rs2	$Rd := 1$, falls $Rs1 \geq Rs2$, 0 sonst	(\geq) set greater than equal unsigned	30
sgt [Ⓟ]	Rd, Rs1, Rs2	$Rd := 1$, falls $Rs1 > Rs2$, 0 sonst	(>) set greater than	30
sgtu [Ⓟ]	Rd, Rs1, Rs2	$Rd := 1$, falls $Rs1 \geq Rs2$, 0 sonst	(>) set greater than unsigned	30
sle [Ⓟ]	Rd, Rs1, Rs2	$Rd := 1$, falls $Rs1 \leq Rs2$, 0 sonst	(\leq) set less than equal	31
sleu [Ⓟ]	Rd, Rs1, Rs2	$Rd := 1$, falls $Rs1 \leq Rs2$, 0 sonst	(\leq) set less than equal unsigned	31
slt	Rd, Rs1, Rs2	$Rd := 1$, falls $Rs1 < Rs2$, 0 sonst	(<) set less than	31
sltu	Rd, Rs1, Rs2	$Rd := 1$, falls $Rs1 < Rs2$, 0 sonst	(<) set less than unsigned	31
slti	Rd, Rs, Imm	$Rd := 1$, falls $Rs < Imm$, 0 sonst	(<) set less than immediate	31

Befehl	Argumente	Wirkung	Erläuterung	S.
sltui	Rd, Rs, Imm	Rd := 1, falls Rs < Imm, 0 sonst	(<) set less than unsigned immediate	31

unbedingte Sprünge

Befehl	Argumente	Wirkung	Erläuterung	S.
b [Ⓢ]	label	unbedingter Sprung nach label	branch	32
j	label	unbedingter Sprung nach label	jump	32

bedingte Sprünge

Befehl	Argumente	Wirkung	Erläuterung	S.
beq	Rs1, Rs2, label	Sprung nach label, falls Rs1=Rs2	(=) branch on equal	33
beqz [Ⓢ]	Rs, label	Sprung nach label, falls Rs=0	(= 0) branch on equal zero	33
bne	Rs1, Rs2, label	Sprung nach label, falls Rs≠0	(≠) branch on not equal zero	33
bnez [Ⓢ]	Rs, label	Sprung nach label, falls Rs≠0	(≠ 0) branch on not equal zero	33
bge [Ⓢ]	Rs1, Rs2, label	Sprung nach label, falls Rs1≥Rs2	(≥) branch on greater than equal	33
bgeu [Ⓢ]	Rs1, Rs2, label	Sprung nach label, falls Rs1≥Rs2	(≥) branch on greater than equal unsigned	33
bgez	Rs, label	Sprung nach label, falls Rs≥0	(≥ 0) branch on greater than equal zero	33
bgt [Ⓢ]	Rs1 Rs2 label	Sprung nach label, falls Rs1>Rs2	(>) branch on greater than	33
bgtu [Ⓢ]	Rs1 Rs2 label	Sprung nach label, falls Rs1>Rs2	(>) branch on greater than unsigned	33
bgtz	Rs, label	Sprung nach label, falls Rs>0	(> 0) branch on greater than zero	33
ble [Ⓢ]	Rs1 Rs2 label	Sprung nach label, falls Rs1<Rs2	(<) branch on less than equal	33
bleu [Ⓢ]	Rs1 Rs2 label	Sprung nach label, falls Rs1<Rs2	(<) branch on less than equal unsigned	33
blez	Rs, label	Sprung nach label, falls Rs≤0	(≤ 0) branch on less than equal zero	33
blt [Ⓢ]	Rs1, Rs2, label	Sprung nach label, falls Rs1<Rs2	(<) branch on less than	34
bltu [Ⓢ]	Rs1 Rs2 label	Sprung nach label, falls Rs1<Rs2	(<) branch on less than unsigned	34
bltz	Rs, label	Sprung nach label, falls Rs<0	(< 0) branch on less than zero	34

Prozedursprunganweisungen

A. ZUSAMMENFASSENDE TABELLEN

Befehl	Argumente	Wirkung	Erläuterung	S.
jr	Rs	unbedingter Sprung an die Adresse in Rs	jump register	43
jal	label	unbedingter Sprung nach label, Adresse des nächsten Befehls in \$ra	jump and link	49
jalr	Rs, Rd	unbedingter Sprung an die Stelle Rs, Adresse des nächsten Befehls in Rd, Default: \$ra	jump and link register	49
bgezal	Rs, label	Sprung nach label wenn $R_s \geq 0$, Adresse des nächsten Befehls in \$ra	Branch on greater than equal zero and link	49
bltzal	Rs, label	Sprung nach label wenn $R_s < 0$, Adresse des nächsten Befehls in \$ra	Branch on less than and link	49

Gleitkommazahlen: Lade- und Speicherbefehle

Befehl	Argumente	Wirkung	Erläuterung	S.
lwc(z)	Cd, Adr	Cd := Mem[Adr]; (z): [0..3]	load word coprocessor	61
l.d ^(P)	Fd, Adr	Fd := Mem[Adr]; Fd+1 := Mem[Adr+4]	Load floating-point double	61
l.s ^(P)	Fd, Adr	Fd := Mem[Adr]	Load floating-point single	61
swc(z)	Cs, Adr	Mem[Adr] := Cs; (z): [0..3]	store word coprocessor	62
s.d ^(P)	Fs, Adr	Mem[Adr] := Fs; Mem[Adr+4] := Fs+1	Store floating-point double	62
s.s ^(P)	Fs, Adr	Mem[Adr] := Fs	Store floating-point single	62

Gleitkommazahlen: Register-Transfer- und Konvertierungs-Befehle

Befehl	Argumente	Wirkung	Erläuterung	S.
mfc(z)	Rd, Cs	Rd := Cs; (z): [0..3]	Move from coprocessor	62
mfc1.d ^(P)	Rd, Cs	Rd := Cs; Rd+1 := Cs+1	Move double from coprocessor 1	62
mtc(z)	Rs, Cd	Cd := Rs; (z): [0..3]	Move to coprocessor	62
mov.d	Fd, Fs	Fd/Fd+1 := Fs/Fs+1	Move floating-point double	62
mov.s	Fd, Fs	Fd := Fs	Move floating-point single	62
cvt.d.s	Fd, Fs	Fd/Fd+1 := Fs	Convert single to double	63
cvt.d.w	Fd, Fs	Fd/Fd+1 := Fs	Convert integer to double	63
cvt.s.d	Fd, Fs	Fd := Fs/Fs+1	Convert double to single	63
cvt.s.w	Fd, Fs	Fd := Fs	Convert integer to single	63
cvt.w.d	Fd, Fs	Fd := Fs/Fs+1	Convert double to integer	63

A.1 Befehlsübersicht

Befehl	Argumente	Wirkung	Erläuterung	S.
cvt.w.s	Fd, Fs	Fd := Fs/Fs+1	Convert single to integer	63

Gleitkommazahlen: Addition und Subtraktion

Befehl	Argumente	Wirkung	Erläuterung	S.
add.d	Fd, Fs1, Fs2	Fd/Fd+1 := Fs1/Fs1+1 + Fs2/Fs2+1	Floating-point addition double	63
add.s	Fd, Fs1, Fs2	Fd := Fs1 + Fs2	Floating-point addition single	63
sub.d	Fd, Fs1, Fs2	Fd/Fd+1 := Fs1/Fs1+1 - Fs2/Fs2+1	Floating-point subtract double	63
sub.s	Fd, Fs1, Fs2	Fd := Fs1 - Fs2	Floating-point subtract single	63

Gleitkommazahlen: Multiplikation und Division

Befehl	Argumente	Wirkung	Erläuterung	S.
mul.d	Fd, Fs1, Fs2	Fd/Fd+1 := Fs1/Fs1+1 * Fs2/Fs2+1	Floating-point multiply double	63
mul.s	Fd, Fs1, Fs2	Fd := Fs1 * Fs2	Floating-point multiply single	63
div.d	Fd, Fs1, Fs2	Fd/Fd+1 := Fs1/Fs1+1 / Fs2/Fs2+1	Floating-point divide double	63
div.s	Fd, Fs1, Fs2	Fd := Fs1 / Fs2	Floating-point divide single	63

Gleitkommazahlen: sonstige arithmetische Befehle

Befehl	Argumente	Wirkung	Erläuterung	S.
abs.d	Fd, Fs	Fd/Fd+1 := ABS (Fs/Fs+1)	Floating-point absolute value double	63
abs.s	Fd, Fs	Fd := ABS (Fs)	Floating-point absolute value single	63
neg.d	Fd, Fs	Fd/Fd+1 := - Fs/Fs+1	Floating-point negate double	63
neg.s	Fd, Fs	Fd := - Fs	Floating-point negate single	63

Gleitkommazahlen: Vergleichsbefehle

Befehl	Argumente	Wirkung	Erläuterung	S.
bc⟨z⟩t		Sprung, wenn condition-flag des Koprozessors ⟨z⟩ wahr (1) ist	Branch coprocessor ⟨z⟩ true	64
bc⟨z⟩f		Sprung, wenn condition-flag des Koprozessors ⟨z⟩ falsch (0) ist	Branch coprocessor ⟨z⟩ false	64
c.eq.d	Fs1, Fs2	Setze condition-flag auf wahr, wenn Fs1/Fs1+1 = Fs2/Fs2+1	Compare equal double	64

A. ZUSAMMENFASSENDE TABELLEN

Befehl	Argumente	Wirkung	Erläuterung	S.
c.eq.s	Fs1, Fs2	Setze condition-flag auf wahr, wenn Fs1 = Fs2	Compare equal single	64
c.le.d	Fs1, Fs2	Setze condition-flag auf wahr, wenn $Fs1/Fs1+1 \leq Fs2/Fs2+1$	Compare less than equal double	64
c.le.s	Fs1, Fs2	Setze condition-flag auf wahr, wenn $Fs1 \leq Fs2$	Compare less than equal single	64
c.lt.d	Fs1, Fs2	Setze condition-flag auf wahr, wenn $Fs1/Fs1+1 < Fs2/Fs2+1$	Compare less than double	64
c.lt.s	Fs1, Fs2	Setze condition-flag auf wahr, wenn $Fs1 < Fs2$	Compare less than single	64

Unterbrechungen und Ausnahmen

Befehl	Argumente	Wirkung	Erläuterung	S.
rfe		Stellt das Statusregister wieder her.	return from exception	69
break	Imm	Verursacht die Ausnahme Imm	break	69

A.2 Alphabetische Befehlsübersicht

abs.s	blez	cvt.w.d	li	mulou	sge	sub.d
add.s	blt	cvt.w.s	lui	mult	sgeu	sub.s
addi	bltu	div	lw	multu	sgt	subu
addiu	bltz	div.d	lwc(z)	neg	sgtu	sw
addu	bltzal	div.s	lwl	neg.s	sh	swc(z)
andi	bne	divu	lwr	negu	sle	swl
bc(z)f	bnez	divu	mfc(z)	nop	sleu	swr
bc(z)t	break	j	mfc1.d	nor	sll	ulh
beqz	c.eq.d	jal	mfhi	not	sllv	ulhu
bge	c.eq.s	jalr	mflo	or	slt	ulw
bgeu	c.le.d	l.d	mov.d	ori	slti	ush
bgez	c.le.s	l.s	mov.s	rem	sltui	usw
bgezal	c.lt.d	la	mtc(z)	remu	sne	xor
bgt	c.lt.s	lb	mthi	ror	sra	xori
bgtu	cvt.d.s	lbu	mtlo	s.d	srav	
bgtz	cvt.d.w	ld	mul	s.s	srl	
ble	cvt.s.d	lh	mul.s	sb	srlv	
bleu	cvt.s.w	lhu	mulo	sd	sub	

A.3 Direktiven-Übersicht

Direktive	Argumente	Erläuterung
.text	[(Adr)]	Legt die folgenden Befehle bzw. Direktivenresultate im Textsegment ab. Mit dem optionalen Argument (Adr) kann eine Startadresse angegeben werden. Defaultwert ist die nächste freie Adresse.
.ktext	[(Adr)]	Wie .text, jedoch erfolgen die Eintragungen im Textsegment des Kernels.

A.3 Direktiven-Übersicht

Direktive	Argumente	Erläuterung
.data	[(<i>Adr</i>)]	Legt die folgenden Befehle bzw. Direktivenresultate im Datensegment ab. Mit dem optionalen Argument (<i>Adr</i>) kann eine Startadresse angegeben werden. Defaultwert ist die nächste freie Adresse.
.kdata	[(<i>Adr</i>)]	Wie .data, jedoch erfolgen die Eintragungen im Datensegment des Kerns, welches aber vom SPIM nicht vom Datensegment unterschieden wird.
.sdata	[(<i>Adr</i>)]	Wie .data, jedoch erfolgen die Eintragungen im Datensegment des Stacks, welches aber vom SPIM nicht vom Datensegment unterschieden wird.
.align	(<i>n</i>)	Richtet die folgenden Eintragungen an 2 ^{<i>n</i>} -Byte-Grenzen aus. (<i>n</i>)=0 schaltet die automatische Ausrichtung bis zur nächsten .(<i>k</i>)data-Direktive aus.
.asciiiz	"(<i>Str</i>)"	Legt die Zeichenkette (<i>Str</i>) ab und hängt Chr(0) an.
.ascii	"(<i>Str</i>)"	Legt die Zeichenkette (<i>Str</i>) ab, hängt Chr(0) aber <i>nicht</i> an.
.word	(<i>w</i> ₁) [, (<i>w</i> ₂) ...]	Legt die folgenden Werte als 32-Bit-Worte ab.
.half	(<i>h</i> ₁) [, (<i>h</i> ₂) ...]	Legt die folgenden Werte als 16-Bit-Worte ab.
.byte	(<i>b</i> ₁) [, (<i>b</i> ₂) ...]	Legt die folgenden Werte als 8-Bit-Worte ab.
.float	(<i>f</i> ₁) [, (<i>f</i> ₂) ...]	Legt die folgenden Werte als 32-Bit-Gleitkommazahlen ab.
.double	(<i>d</i> ₁) [, (<i>d</i> ₂) ...]	Legt die folgenden Werte als 64-Bit-Gleitkommazahlen ab.
.space	(<i>n</i>)	Lässt die nächsten (<i>n</i>) Byte frei.
.set	noat	Schaltet die SPIM-Warnung vor dem Benutzen von \$at <i>aus</i> .
.set	at	Schaltet die SPIM-Warnung vor dem Benutzen von \$at <i>ein</i> .
.globl	(<i>label</i>)	Exportiert die Marke (<i>label</i>), so dass sie von anderen Dateien aus benutzt werden kann.
.extern	(<i>label</i>) (<i>Größe</i>)	Legt das mit (<i>label</i>) bezeichnete Objekt mit der Größe (<i>Größe</i>) Bytes in den durch \$gp bezeichneten Bereich.

LITERATUR

A.4 Registerübersicht

Register- -name	-nr.	vereinbarte Nutzung	beschreibbar	caller- saved	callee- saved
\$zero	0	Enthält den Wert 0	ja	-	-
\$at	1	temporäres Assemblerregister	nein	-	-
\$v0-\$v1	1-2	Funktionsergebnisse und 2	ja	×	-
\$a0-\$a3	4-7	Argumente 1 bis 4 für den Prozeduraufruf	ja	×	-
\$t0-\$t9	8-15, 24-25	temporäre Variablen 1-9	ja	×	-
\$s0-\$s7	16-23	langlebige Variablen 1-8	ja	-	×
\$k0-\$k1	26-27	Kernel-Register 1 und 2	nein	-	-
\$gp	28	Zeiger auf Datensegment	nein	-	-
\$sp	29	Stackpointer	ja	-	-
\$fp	30	Framepointer	ja	-	×
\$ra	31	Return address	ja	-	×

Register der Koprozessoren: Siehe Seite 68 (mathematischer Koprozessor) und Abbildung 10 auf Seite 65 (Koprozessor 0).

A.5 Betriebssystemfunktionen

Funktion	Code in \$v0	Argument(e)	Ergebnis
print_int	1	\$a0	
print_float	2	\$f12	
print_double	3	\$f12/13	
print_string	4	\$a0 enthält die Anfangsadresse der mit Chr(0) terminierten Zeichenkette	
read_int	5		\$v0
read_float	6		\$f0
read_double	7		\$f0/1
read_string	8	\$a0 enthält die Adresse ab der die Zeichenkette abgelegt wird, \$a1 ihre maximale Länge	Zeichenkette ab (\$a0)
sbrk	9	\$a0 benötigte Größe	\$v0 Anfangsadresse des Speicherbereichs
exit	10		

B Literaturverzeichnis

Literatur

- [Pat] Patterson, David A. und John L. Hennessy: Computer Organization and Design. The Hardware/Software Interface. San Francisco 1994.
Das Buch enthält [Lar] als Anhang A.
- [Lar] Larus, James R.: SPIM S20: A MIPS R2000 Simulator. Madison 1993.
http://www.cs.wisc.edu/~larus/SPIM_manual/spim-manual.html
Das Dokument ist als Anhang A in [Pat] enthalten.

- [Tan] Tanenbaum, Andrew S.: Structured Computer Organisation. Third Edition (1990). Amsterdam 1976.
- [Coy] Coy, Wolfgang: Aufbau und Wirkungsweise von Rechenanlagen. 2. Auflage. Braunschweig/Wiesbaden 1992.

C Verzeichnisse

Abbildungsverzeichnis

1	MIPS-Register und ihre Verwendungszwecke nach Registernummern	8
2	Argumentarten	14
3	Zeichenkombinationen	16
4	Betriebssystemfunktionen des SPIM (system calls)	29
5	Rotations- und Schiebebefehle	31
6	Das Speicherlayout des MIPS	49
7	Die Entwicklung des Stacks beim „Einkellern“	50
8	Die Entwicklung des Stacks beim „Abräumen“	51
9	Das Grundprinzip des Unterprogrammaufrufs	54
10	Register des Koprozessors 0	65
11	Das Statusregister	65
12	Ausnahmecodes (exception codes) (Bit 5 bis 2 des Cause-Registers)	66
13	Neue Argumentarten	69

Beispielverzeichnis

1	Ein erstes Beispielprogramm ohne besonderen praktischen Wert	13
2	Beispiele zur direkten Adressierung	18
3	Ein- und Ausgabe mit Betriebssystemfunktionen	27
4	Einfache Verzweigung (Erster Versuch)	38
5	Einfache Verzweigung (Zweiter Versuch)	38
6	Einfache Verzweigung (Zweiter Versuch, Variante)	38
7	Eine Abweisende Schleife	40
8	Eine nachprüfende Schleife	40
9	Eine effiziente Zählschleife	41
10	Indexieren eines Feldes	44
11	Mehrfache Fallunterscheidung mit einfachen Fallunterscheidungen	46
12	Mehrfache Fallunterscheidung mit Sprungtabelle	46
13	Ein einfacher Unterprogrammaufruf	54
14	Ein einfacher kommentierter Unterprogrammaufruf	54
15	Beispiel zur Prozedur-Konvention: Ballot-Zahlen	60
16	Das Hornerschema mit doppelt genauen Gleitkommazahlen	72

Index

- abweisende Schleife, [39](#)
- Adresse>symbolische, [10](#)
- Adressierungsmodus>direkt, [18](#)
- Adressierungsmodus>immediate, [22](#)
- Adressierungsmodus>indexiert, [18](#)
- Adressierungsmodus>PC-relativer, [37](#)
- Adressierungsmodus>Register-direkt, [22](#)
- Adressierungsmodus>Register-indirekt, [18](#)
- Adressierungsmodus>unmittelbar, [22](#)
- align-Direktive= .align-Direktive, [17](#)
- aligned data, [17](#)
- Argumente, [56](#)
- arithmetische Befehle, [22–26](#), [71](#)
- Array, *siehe* Felder
- ASCII-Code, [16](#)
- ascii-Direktive= .ascii-Direktive, [16](#)
- asciiz-Direktive= .asciiz-Direktive, [16](#)
- Assembler, [10](#)
- Assembler>Gestaltung von Programmen, [12](#)
- Assembler>Programmaufbau, [13](#)
- Assemblieranweisung, *siehe* Direktive
- Assemblersimulator, [5](#)
- Assemblersprache, [10](#)
- Aufgerufener, *siehe* callee
- Aufrufer, *siehe* caller
- Ausgabe, [27](#)
- Ausnahme, [22](#), [64](#)
- Ausnahmebehandlung, [22](#), [25](#), [64–67](#)

- Ballot-Zahlen, [60](#)
- Befehle, [74](#), [74–80](#)
- Befehle>arithmetische, [22–26](#), [71](#)
- Befehle>Betriebssystem-, [27](#)
- Befehle>Lade-, [18–20](#), [69](#)
- Befehle>logische, [30](#)
- Befehle>Pseudo-, [10](#)
- Befehle>Register-Transfer-, [21](#), [70](#)
- Befehle>Rotations-, [30–31](#)
- Befehle>Schiebe-, [30–32](#)
- Befehle>Speicher-, [20](#), [70](#)
- Befehle>Sprung-, [36–37](#), [46](#), [53](#)
- Befehle>Transfer-, [18–21](#), [69–71](#)
- Befehle>Vergleichs-, [32](#), [71–72](#)
- Betriebssystem, [27](#), [64](#)
- Betriebssystem>-funktion, [27](#), [82](#)
- Bezeichner, symbolischer, [10](#), [13](#)
- branching, [35](#)
- byte-Direktive= .byte-Direktive, [15](#)

- call-by-reference, [56](#)
- call-by-value, [56](#)
- callee, [53](#)
- callee>-saved Register, [55](#), [57](#)
- caller, [53](#)

- caller>-saved Register, [55](#), [55](#)
- Case-Anweisung, [46–47](#)
- Cause-Register, [66](#)
- CISC, [7](#), [48](#), [51](#)
- Complex Instruction Set Computer, [7](#), [48](#), [51](#)
- condition-flag, [71](#)
- Coprocessor, *siehe* Koprozessor

- data-Direktive= .data-Direktive, [12](#)
- Datenausrichtung, [17](#)
- Datenhaltung>Felder, [43–44](#)
- Datenhaltung>ganze Zahlen, [15](#)
- Datenhaltung>Gleitkommazahlen, [68–69](#)
- Datenhaltung>Verbunde, [45](#)
- Datenhaltung>Zeichenketten, [15](#), [16](#)
- Datensegment, [11](#)
- direkte Adressierung, [18](#)
- Direktive, [10](#), [80](#)
- Direktive>.space= .space, [43](#)
- Direktive>align= .align, [17](#)
- Direktive>ascii= .ascii, [16](#)
- Direktive>asciiz= .asciiz, [16](#)
- Direktive>byte= .byte, [15](#)
- Direktive>data= .data, [12](#)
- Direktive>double= .double, [69](#)
- Direktive>extern= .extern, [81](#)
- Direktive>float= .float, [68](#)
- Direktive>globl= .globl, [81](#)
- Direktive>half= .half, [15](#)
- Direktive>kdata= .kdata, [66](#)
- Direktive>ktext= .ktext, [66](#)
- Direktive>set= .set, [81](#)
- Direktive>text= .text, [12](#)
- Direktive>word= .word, [15](#)
- double precision, [69](#)
- double-Direktive= .double-Direktive, [69](#)

- Ein- und Ausgabe, [27](#)
- einfache Fallunterscheidung, [38](#), [35–39](#)
- exception, *siehe* Ausnahme
- exception handler, *siehe* Ausnahmebehandlung

- Fallunterscheidung>einfache, [38](#), [35–39](#)
- Fallunterscheidung>mehrfache, [46–47](#)
- Felder, [43](#), [43–44](#)
- float-Direktive= .float-Direktive, [68](#)
- Floating Point Unit, [68](#)
- Floating-Point-Zahlen, *siehe* Gleitkommazahlen
- For-Schleife, [40](#), [41](#)
- FPU, [68](#)
- Framepointer, [58](#)
- Funktionen, *siehe* Unterprogramme

- ganze Zahlen, Datenhaltung, 15
 general purpose register, 7, 68
 Gleitkommazahlen, 68, 68–73
 Gleitkommazahlen>Datenhaltung, 68–69
 Gleitkommazahlen>doppelte Genauigkeit, 69
 Gleitkommazahlen>einfache Genauigkeit, 68
 GOSUB=GOSUB, 53
- half-Direktive= .half-Direktive, 15
 Heap, 48
- IF-Anweisung, 38, 35–39
 indizierte Adressierung, 18
 Instruktion, 10
 INTEL, 7
 Internet, 5
 Interrupt, *siehe* Unterbrechung
 Interrupthandler, *siehe* Unterbrechungsbehandlungsroutine
- jump address table, 46
 jumping, 35
- kdata-Direktive= .kdata-Direktive, 66
 Kellerspeicher, *siehe* Stack
 Kernelsegment, 66
 Kommentar, 12
 Kommentar>-zeichen, 11
 Kontrollstrukturen, 35–42
 Kontrollstrukturen>IF-Anweisung, 38, 35–39
 Kontrollstrukturen>mehrfache Fallunterscheidung, 46–47
 Kontrollstrukturen>Schleifen, 39–42
 Kontrollstrukturen>Schleifen>abweisende, 39
 Kontrollstrukturen>Schleifen>For-, 40, 41
 Kontrollstrukturen>Schleifen>nachprüfende, 39, 40
 Kontrollstrukturen>Schleifen>Repeat-Until-, 39, 40
 Kontrollstrukturen>Schleifen>verlassen, 41
 Kontrollstrukturen>Schleifen>While-, 39
 Kontrollstrukturen>Schleifen>Zähl-, 40, 41
 Koprozessor, 7, 68
 Koprozessor>0 (Betriebssystem), 65
 Koprozessor>1 (mathematischer), 68
 ktext-Direktive= .ktext-Direktive, 66
- label, 10
 Ladebefehle, 18–20, 69
 LIFO, 48
 Load-Store-Architektur, 15
 logische Befehle, 30
- Makro, 11
 Marke, 10
 Maschinenbefehl, 10
 Maschinensprache, 10
 mathematischer Koprozessor, 68
 mehrfache Fallunterscheidung, 46–47
 MIPS R2000=**MIPS** R2000, 5, 7
 Motorola, 7, 48
- nachprüfende Schleife, 39, 40
- Parameterübergabe, 56
 PC-relative Adressierung, 37
 POP=POP, 48, 51
 Procedure Call Convention, *siehe* Unterprogramme/Prozedur-Konvention
 Programmverzweigungen, 38, 35–39
 Prozedur-Konvention, *siehe* Unterprogramme/Prozedur-Konvention
 Prozeduren, *siehe* Unterprogramme
 Pseudobefehle, 10
 PUSH=PUSH, 48, 51
- Rücksprung, 53, 53
 Records, *siehe* Verbunde
 Reduced Instruction Set Computer, 7
 Register, 7, **82**
 Register-direkte Adressierung, 22
 Register-indirekte Adressierung, 18
 Register-Transfer-Befehle, 21, 70
 Register>callee-saved Register, 55, 57, **82**
 Register>caller-saved Register, 55, 55
 Register>Cause-, 66
 Register>floating point, 68
 Register>Framepointer, 58
 Register>general purpose register, 7, 68
 Register>lo und hi=Register>lo und hi, 7
 Register>lo und hi=lo und hi, 21, 24
 Register>Status-, 65
 Register>Verwendungszweck, 7, **8**, **82**
 Repeat-Until-Schleife, 39, 40
 RETURN=RETURN, 53
 RISC, 7
 Rotationsbefehle, 30–31
- Schiebebefehle, 30–32
 Schleifen, *siehe* Kontrollstrukturen/Schleifen
 Segment, 11
 single precision, 68
 Speicherbefehle, 20, 70
 Speicherlayout, 49
 SPIM=SPIM, 5
 Sprungbefehle, 36–37, 46, 53
 Sprungtabelle, 46
 Stack, 48–52, 56, 57
 Stack>-pointer, 48

INDEX

Stack>abräumen, 50, **51**
Stack>einkellern, 49, **50**
Stack>frame=-frame, **57**
Stack>pointer=-pointer, 57
Stack>segment=-segment, 11
Statusregister, 65
Sun, 7
symbolische Adresse, **10**
symbolischer Bezeichner, **10**, 13
system call, 27, **82**

text-Direktive= .text-Direktive, 12
Textsegment, 11
Transferbefehle, 18–21, 69–71
Trap, *siehe* Ausnahme
Traphandler, *siehe* Unterbrechungsbe-
handlungsroutine

unmittelbare Adressierung, 22
unsigned, 19
Unterbrechung, 22, 49, **64**
Unterbrechungsbehandlung, 64–67
Unterbrechungsbehandlungsroutine,
65
Unterprogramme, **53**, **54**, 53–63
Unterprogramme>Argumente, 56
Unterprogramme>callee-saved Register,
57
Unterprogramme>caller-saved Register,
55
Unterprogramme>Parameterübergabe,
56
Unterprogramme>Prozedur-
Konvention, **55**, **62**, 55–63
Unterprogramme>Rücksprung, 53, **53**
Unterprogramme>Stackframe, **57**
Unterprogramme>Variablenparameter,
56
Unterprogramme>Wertparameter, 56

Variablen, 15
Variablen>temporäre, 48
Variablenparameter, 56
Verbunde, **43**, 45
Vergleichsbefehle, 32, 71–72
von-Neumann-Rechner, 11
vorzeichenlose Arithmetik, 19

Wertparameter, 56
While-Schleife, **39**
word-Direktive= .word-Direktive, 15

Zählschleife, **40**, 41
Zeichenketten, 15, **16**